

Free Pascal :
Reference guide.

Reference guide for Free Pascal, version 1.9.2
Document version 1.9
January 2004

Michaël Van Canneyt

Contents

I	The Pascal language	13
1	Pascal Tokens	14
1.1	Symbols	14
1.2	Comments	14
1.3	Reserved words	15
	Turbo Pascal reserved words	15
	Delphi reserved words	16
	Free Pascal reserved words	16
	Modifiers	16
1.4	Identifiers	16
1.5	Numbers	17
1.6	Labels	18
1.7	Character strings	18
2	Constants	19
2.1	Ordinary constants	19
2.2	Typed constants	20
2.3	Resource strings	21
3	Types	22
3.1	Base types	22
	Ordinal types	23
	Integers	23
	Boolean types	24
	Enumeration types	25
	Subrange types	26
	Real types	26
3.2	Character types	27
	Char	27
	Strings	27
	Short strings	27

Ansistrings	28
WideStrings	29
Constant strings	29
PChar - Null terminated strings	30
3.3 Structured Types	31
Arrays	31
Static arrays	32
Dynamic arrays	33
Record types	35
Set types	38
File types	39
3.4 Pointers	39
3.5 Forward type declarations	41
3.6 Procedural types	42
3.7 Variant types	43
Definition	43
Variants in assignments and expressions	44
Variants and interfaces	45
4 Objects	46
4.1 Declaration	46
4.2 Fields	47
4.3 Constructors and destructors	48
4.4 Methods	49
4.5 Method invocation	49
Static methods	50
Virtual methods	50
Abstract methods	51
4.6 Visibility	52
5 Classes	53
5.1 Class definitions	53
5.2 Class instantiation	55
5.3 Methods	55
invocation	55
Virtual methods	55
Message methods	56
5.4 Properties	57
6 Interfaces	61
6.1 Definition	61

6.2	Identification: A GUID	62
6.3	Interfaces and COM	62
7	Expressions	64
7.1	Expression syntax	65
7.2	Function calls	66
7.3	Set constructors	67
7.4	Value typecasts	68
7.5	The @ operator	69
7.6	Operators	69
	Arithmetic operators	69
	Logical operators	70
	Boolean operators	71
	String operators	71
	Set operators	71
	Relational operators	71
8	Statements	73
8.1	Simple statements	73
	Assignments	73
	Procedure statements	74
	Goto statements	75
8.2	Structured statements	75
	Compound statements	76
	The Case statement	76
	The If . . then . . else statement	77
	The For . . to/downto . . do statement	78
	The Repeat . . until statement	79
	The While . . do statement	80
	The With statement	80
	Exception Statements	82
8.3	Assembler statements	82
9	Using functions and procedures	84
9.1	Procedure declaration	84
9.2	Function declaration	85
9.3	Parameter lists	85
	Value parameters	86
	Variable parameters	86
	Out parameters	87
	Constant parameters	87

Open array parameters	88
Array of const	88
9.4 Function overloading	90
9.5 Forward defined functions	91
9.6 External functions	92
9.7 Assembler functions	93
9.8 Modifiers	93
alias	94
cdecl	94
export	95
inline	95
interrupt	95
pascal	95
popstack	95
public	96
register	96
saveregisters	96
safecall	96
stdcall	96
9.9 Unsupported Turbo Pascal modifiers	97
10 Operator overloading	98
10.1 Introduction	98
10.2 Operator declarations	98
10.3 Assignment operators	99
10.4 Arithmetic operators	101
10.5 Comparision operator	102
11 Programs, units, blocks	104
11.1 Programs	104
11.2 Units	105
11.3 Blocks	106
11.4 Scope	107
Block scope	107
Record scope	108
Class scope	108
Unit scope	108
11.5 Libraries	109
12 Exceptions	110
12.1 The raise statement	110

12.2 The try...except statement	111
12.3 The try...finally statement	112
12.4 Exception handling nesting	113
12.5 Exception classes	113
13 Using assembler	114
13.1 Assembler statements	114
13.2 Assembler procedures and functions	114
II Reference : The System unit	115
14 The system unit	116
14.1 Types, Constants and Variables	116
Types	116
Constants	119
Variables	122
14.2 Function list by category	123
File handling	123
Memory management	124
Mathematical routines	125
String handling	125
Operating System functions	126
Miscellaneous functions	126
14.3 Functions and Procedures	127
Abs	127
Addr	127
Append	128
Arctan	128
Assert	129
Assign	129
Assigned	130
BinStr	130
Blockread	131
Blockwrite	131
Break	132
Chdir	132
Chr	133
Close	133
CompareByte	134
CompareChar	135

CompareDWord	136
CompareWord	137
Concat	138
Continue	139
Copy	139
Cos	140
CSeg	140
Dec	141
Delete	141
Dispose	142
DSeg	143
Eof	143
Eoln	144
Erase	144
Exclude	145
Exit	146
Exp	147
Filepos	148
Filesize	148
FillByte	149
Fillchar	150
FillDWord	150
Fillword	151
Flush	151
Frac	152
Freemem	152
Getdir	153
Getmem	153
GetMemoryManager	154
Halt	154
HexStr	154
Hi	155
High	155
Inc	156
Include	157
IndexByte	157
IndexChar	158
IndexDWord	159
IndexWord	160
Insert	160

IsMemoryManagerSet	161
Int	161
IOresult	161
Length	163
Ln	163
Lo	164
LongJump	164
Low	164
Lowercase	165
Mark	165
Maxavail	166
Memavail	166
Mkdir	167
Move	167
MoveChar0	168
New	168
Odd	168
OctStr	169
Ofs	169
Ord	170
Paramcount	170
Paramstr	171
Pi	171
Pos	172
Power	172
Pred	172
Ptr	173
Random	173
Randomize	174
Read	174
Readln	175
Real2Double	175
Release	176
Rename	176
Reset	177
Rewrite	177
Rmdir	178
Round	179
Runerror	179
Seek	179

SeekEof	180
SeekEoln	181
Seg	181
SetMemoryManager	182
SetJmp	182
SetLength	182
SetString	183
SetTextBuf	183
Sin	184
SizeOf	185
Sptr	185
Sqr	185
Sqrt	186
SSeg	186
Str	187
StringOfChar	187
Succ	188
Swap	188
Trunc	188
Truncate	189
Ucase	189
Val	190
Write	191
WriteLn	191
15 The OBJPAS unit	193
15.1 Types	193
15.2 Functions and Procedures	193
AssignFile	193
CloseFile	194
Freemem	194
Getmem	195
GetResourceStringCurrentValue	195
GetResourceStringDefaultValue	196
GetResourceStringHash	196
GetResourceStringName	197
Hash	197
Paramstr	198
ReAllocMem	198
ResetResourceTables	199

ResourceStringCount	199
ResourceStringTableCount	199
SetResourceStrings	200
SetResourceStringValue	200

List of Tables

3.1	Predefined integer types	23
3.2	Predefined integer types	24
3.3	Boolean types	24
3.4	Supported Real types	27
3.5	PChar pointer arithmetic	31
3.6	Set Manipulation operators	39
7.1	Precedence of operators	64
7.2	Binary arithmetic operators	70
7.3	Unary arithmetic operators	70
7.4	Logical operators	70
7.5	Boolean operators	71
7.6	Set operators	72
7.7	Relational operators	72
8.1	Allowed C constructs in Free Pascal	74
9.1	Unsupported modifiers	97

About this guide

This document describes all constants, types, variables, functions and procedures as they are declared in the system unit. Furthermore, it describes all pascal constructs supported by Free Pascal, and lists all supported data types. It does not, however, give a detailed explanation of the pascal language. The aim is to list which Pascal constructs are supported, and to show where the Free Pascal implementation differs from the Turbo Pascal implementation.

Notations

Throughout this document, we will refer to functions, types and variables with `typewriter` font. Functions and procedures have their own subsections, and for each function or procedure we have the following topics:

Declaration The exact declaration of the function.

Description What does the procedure exactly do ?

Errors What errors can occur.

See Also Cross references to other related functions/commands.

The cross-references come in two flavours:

- References to other functions in this manual. In the printed copy, a number will appear after this reference. It refers to the page where this function is explained. In the on-line help pages, this is a hyperlink, which can be clicked to jump to the declaration.
- References to Unix manual pages. (For linux and unix related things only) they are printed in `typewriter` font, and the number after it is the Unix manual section.

Syntax diagrams

All elements of the pascal language are explained in syntax diagrams. Syntax diagrams are like flow charts. Reading a syntax diagram means getting from the left side to the right side, following the arrows. When the right side of a syntax diagram is reached, and it ends with a single arrow, this means the syntax diagram is continued on the next line. If the line ends on 2 arrows pointing to each other, then the diagram is ended.

Syntactical elements are written like this

→ syntactical elements are like this →

Keywords which must be typed exactly as in the diagram:

→ **keywords are like this** →

When something can be repeated, there is an arrow around it:

→ this can be repeated →

When there are different possibilities, they are listed in columns:

→ First possibility
Second possibility →

Note, that one of the possibilities can be empty:



This means that both the first or second possibility are optional. Of course, all these elements can be combined and nested.

Part I

The Pascal language


Chapter 1

Pascal Tokens

In this chapter we describe all the pascal reserved words, as well as the various ways to denote strings, numbers, identifiers etc.

1.1 Symbols

Free Pascal allows all characters, digits and some special ASCII symbols in a Pascal source file.



The following characters have a special meaning:

+ - * / = < > [] . , () : ^ @ { } \$ #

and the following character pairs too:

<= >= := += -= *= /= (* *) (. .) //

When used in a range specifier, the character pair (. is equivalent to the left square bracket [. Likewise, the character pair .) is equivalent to the right square bracket]. When used for comment delimiters, the character pair (* is equivalent to the left brace { and the character pair *) is equivalent to the right brace }. These character pairs retain their normal meaning in string expressions.

1.2 Comments

Free Pascal supports the use of nested comments. The following constructs are valid comments:

```
(* This is an old style comment *)
{ This is a Turbo Pascal comment }
// This is a Delphi comment. All is ignored till the end of the line.
```

The following are valid ways of nesting comments:

```
{ Comment 1 (* comment 2 *) }
(* Comment 1 { comment 2 } *)
{ comment 1 // Comment 2 }
(* comment 1 // Comment 2 *)
// comment 1 (* comment 2 *)
// comment 1 { comment 2 }
```

The last two comments *must* be on one line. The following two will give errors:

```
// Valid comment { No longer valid comment !!
}
```

and

```
// Valid comment (* No longer valid comment !!
*)
```

The compiler will react with a 'invalid character' error when it encounters such constructs, regardless of the -So switch.

1.3 Reserved words

Reserved words are part of the Pascal language, and cannot be redefined. They will be denoted as **this** throughout the syntax diagrams. Reserved words can be typed regardless of case, i.e. Pascal is case insensitive. We make a distinction between Turbo Pascal and Delphi reserved words, since with the -So switch, only the Turbo Pascal reserved words are recognised, and the Delphi ones can be redefined. By default, Free Pascal recognises the Delphi reserved words.

Turbo Pascal reserved words

The following keywords exist in Turbo Pascal mode

absolute	else	nil	shl
and	end	not	shr
array	file	object	string
asm	for	of	then
begin	function	on	to
break	goto	operator	type
case	if	or	unit
const	implementation	packed	until
constructor	in	procedure	uses
continue	inherited	program	var
destructor	inline	record	while
div	interface	repeat	with
do	label	self	xor
downto	mod	set	

Delphi reserved words

The Delphi (II) reserved words are the same as the pascal ones, plus the following ones:

as	finalization	library	threadvar
class	finally	on	try
except	initialization	property	
exports	is	raise	

Free Pascal reserved words

On top of the Turbo Pascal and Delphi reserved words, Free Pascal also considers the following as reserved words:

dispose	false	true
exit	new	

Modifiers


The following is a list of all modifiers. Contrary to Delphi, Free Pascal doesn't allow the programmer to redefine these modifiers.

absolute	external	pascal	register
abstract	far	popstack	saveregisters
alias	forward	private	stdcall
assembler	index	protected	virtual
cdecl	name	public	write
default	near	published	
export	override	read	

Remark: Predefined types such as `Byte`, `Boolean` and constants such as `maxint` are *not* reserved words. They are identifiers, declared in the system unit. This means that these types can be redefined in other units. The programmer is, however, not encouraged to do this, as it will cause a lot of confusion.

1.4 Identifiers

Identifiers denote constants, types, variables, procedures and functions, units, and programs. All names of things that are defined are identifiers. An identifier consists of 255 significant characters (letters, digits and the underscore character), from which the first must be an alphanumeric character, or an underscore (`_`). The following diagram gives the basic syntax for identifiers.




1.5 Numbers

Numbers are by default denoted in decimal notation. Real (or decimal) numbers are written using engineering or scientific notation (e.g. `0.314E1`).

For integer type constants, Free Pascal supports 4 formats:

1. Normal, decimal format (base 10). This is the standard format.
2. Hexadecimal format (base 16), in the same way as Turbo Pascal does. To specify a constant value in hexadecimal format, prepend it with a dollar sign (\$). Thus, the hexadecimal `$FF` equals 255 decimal. Note that case is insignificant when using hexadecimal constants.
3. As of version 1.0.7, Octal format (base 8) is also supported. To specify a constant in octal format, prepend it with an ampersand (&). For instance 15 is specified in octal notation as `&17`.
4. Binary notation (base 2). A binary number can be specified by preceding it with a percent sign (%). Thus, 255 can be specified in binary notation as `%11111111`.

The following diagrams show the syntax for numbers.



Remark: It is to note that all decimal constants which do not fit within the `-2147483648..2147483647` range, are silently and automatically parsed as real-type constants.


Chapter 2

Constants

Just as in Turbo Pascal, Free Pascal supports both normal and typed constants.

2.1 Ordinary constants

Ordinary constants declarations are not different from the Turbo Pascal or Delphi implementation.



The compiler must be able to evaluate the expression in a constant declaration at compile time. This means that most of the functions in the Run-Time library cannot be used in a constant declaration. Operators such as `+`, `-`, `*`, `/`, `not`, `and`, `or`, `div`, `mod`, `ord`, `chr`, `sizeof`, `pi`, `int`, `trunc`, `round`, `frac`, `odd` can be used, however. For more information on expressions, see chapter 7, page 64. Only constants of the following types can be declared: Ordinal types, Real types, Char, and String. The following are all valid constant declarations:

```
Const
  e = 2.7182818; { Real type constant. }
  a = 2;         { Ordinal (Integer) type constant. }
  c = '4';       { Character type constant. }
  s = 'This is a constant string'; {String type constant.}
  ls = chr(32)
  ls = SizeOf(Longint);
```


Assigning a value to an ordinary constant is not permitted. Thus, given the previous declaration, the following will result in a compiler error:

```
s := 'some other string';
```

Prior to version 1.9, Free Pascal did not correctly support 64-bit constants. As of version 1.9, 64-bits constants can be specified.

2.2 Typed constants

Typed constants serve to provide a program with initialised variables. Contrary to ordinary constants, they may be assigned to at run-time. The difference with normal variables is that their value is initialised when the program starts, whereas normal variables must be initialised explicitly.



Given the declaration:

```
Const
  S : String = 'This is a typed constant string';
```

The following is a valid assignment:

```
S := 'Result : '+Func;
```

Where `Func` is a function that returns a `String`. Typed constants are often used to initialize arrays and records. For arrays, the initial elements must be specified, surrounded by round brackets, and separated by commas. The number of elements must be exactly the same as the number of elements in the declaration of the type. As an example:

```
Const
  tt : array [1..3] of string[20] = ('ikke', 'gij', 'hij');
  ti : array [1..3] of Longint = (1,2,3);
```

For constant records, each element of the record should be specified, in the form `Field : Value`, separated by commas, and surrounded by round brackets. As an example:

```
Type
  Point = record
    X,Y : Real
  end;
Const
  Origin : Point = (X:0.0; Y:0.0);
```

The order of the fields in a constant record needs to be the same as in the type declaration, otherwise a compile-time error will occur.

Remark: It should be stressed that typed constants are initialized at program start. This is also true for *local* typed constants. Local typed constants are also initialized at program start. If their value was changed during previous invocations of the function, they will retain their changed value, i.e. they are not initialized each time the function is invoked.

2.3 Resource strings

A special kind of constant declaration part is the `Resourestring` part. This part is like a `Const` section, but it only allows to declare constant of type string. This part is only available in the `Delphi` or `objfpc` mode.

The following is an example of a `resourcestring` definition:

```
Resourcestring
```

```
FileMenu = '&File...';  
EditMenu = '&Edit...';
```

All string constants defined in the `resourcestring` section are stored in special tables, allowing to manipulate the values of the strings at runtime with some special mechanisms.

Semantically, the strings are like constants; Values can not be assigned to them, except through the special mechanisms in the `objpas` unit. However, they can be used in assignments or expressions as normal constants. The main use of the `resourcestring` section is to provide an easy means of internationalization.

More on the subject of `resourcestrings` can be found in the [Programmers guide](#), and in the chapter on the `objpas` later in this manual.

Chapter 3

Types

All variables have a type. Free Pascal supports the same basic types as Turbo Pascal, with some extra types from Delphi. The programmer can declare his own types, which is in essence defining an identifier that can be used to denote this custom type when declaring variables further in the source code.

Type declaration

→ type declaration – identifier – = – type – ; →

There are 7 major type classes :

Types




The last class, `type identifier`, is just a means to give another name to a type. This presents a way to make types platform independent, by only using these types, and then defining these types for each platform individually. The programmer that uses these units doesn't have to worry about type size and so on. It also allows to use shortcut names for fully qualified type names. e.g. define `system.longint` as `Olongint` and then redefine `longint`.

3.1 Base types

The base or simple types of Free Pascal are the Delphi types. We will discuss each separate.

Simple types



Ordinal types

With the exception of `int64`, `qword` and `Real` types, all base types are ordinal types. Ordinal types have the following characteristics:

1. Ordinal types are countable and ordered, i.e. it is, in principle, possible to start counting them one by one, in a specified order. This property allows the operation of functions as `Inc` (156), `Ord` (170), `Dec` (141) on ordinal types to be defined.
2. Ordinal values have a smallest possible value. Trying to apply the `Pred` (172) function on the smallest possible value will generate a range check error if range checking is enabled.
3. Ordinal values have a largest possible value. Trying to apply the `Succ` (188) function on the largest possible value will generate a range check error if range checking is enabled.

Integers

A list of pre-defined integer types is presented in table (3.1) The integer types, and their ranges and

Table 3.1: Predefined integer types

Name
Integer
Shortint
SmallInt
Longint
Longword
Int64
Byte
Word
Cardinal
QWord
Boolean
ByteBool
LongBool
Char

sizes, that are predefined in Free Pascal are listed in table (3.2). It is to note that the `qword` and `int64` types are not true ordinals, so some pascal constructs will not work with these two integer types.

The `integer` type maps to the `smallint` type in the default Free Pascal mode. It maps to either a `longint` or `int64` in either Delphi or ObjFPC mode. The `cardinal` type is currently always mapped to the `longword` type. The definition of the `cardinal` and `integer` types may change from one architecture to another and from one compiler mode to another. They usually have the same size as the underlying target architecture.

Table 3.2: Predefined integer types

Type	Range	Size in bytes
Byte	0 .. 255	1
Shortint	-128 .. 127	1
Smallint	-32768 .. 32767	2
Word	0 .. 65535	2
Integer	either smallint, longint or int64	size 2,4 or 8
Cardinal	either word, longword or qword	size 2,4 or 8
Longint	-2147483648 .. 2147483647	4
Longword	0..4294967295	4
Int64	-9223372036854775808 .. 9223372036854775807	8
QWord	0 .. 18446744073709551615	8

Free Pascal does automatic type conversion in expressions where different kinds of integer types are used.

Boolean types

Free Pascal supports the `Boolean` type, with its two pre-defined possible values `True` and `False`. It also supports the `ByteBool`, `WordBool` and `LongBool` types. These are the only two values that can be assigned to a `Boolean` type. Of course, any expression that resolves to a boolean value, can also be assigned to a boolean type. Assuming `B` to be of type `Boolean`, the following

Table 3.3: Boolean types

Name	Size	Ord(True)
<code>Boolean</code>	1	1
<code>ByteBool</code>	1	Any nonzero value
<code>WordBool</code>	2	Any nonzero value
<code>LongBool</code>	4	Any nonzero value

are valid assignments:

```
B := True;
B := False;
B := 1<>2; { Results in B := True }
```

Boolean expressions are also used in conditions.


Remark: In Free Pascal, boolean expressions are always evaluated in such a way that when the result is known, the rest of the expression will no longer be evaluated (Called short-cut evaluation). In the following example, the function `Func` will never be called, which may have strange side-effects.

```
...
B := False;
A := B and Func;
```

Here `Func` is a function which returns a `Boolean` type.

Enumeration types

Enumeration types are supported in Free Pascal. On top of the Turbo Pascal implementation, Free Pascal allows also a C-style extension of the enumeration type, where a value is assigned to a particular element of the enumeration list.



(see chapter 7, page 64 for how to use expressions) When using assigned enumerated types, the assigned elements must be in ascending numerical order in the list, or the compiler will complain. The expressions used in assigned enumerated elements must be known at compile time. So the following is a correct enumerated type declaration:

```
Type
  Direction = ( North, East, South, West );
```

The C style enumeration type looks as follows:

```
Type
  EnumType = (one, two, three, forty := 40, fortyone);
```

As a result, the ordinal number of `forty` is 40, and not 3, as it would be when the `' := 40 '` wasn't present. The ordinal value of `fortyone` is then 41, and not 4, as it would be when the assignment wasn't present. After an assignment in an enumerated definition the compiler adds 1 to the assigned value to assign to the next enumerated value. When specifying such an enumeration type, it is important to keep in mind that the enumerated elements should be kept in ascending order. The following will produce a compiler error:

```
Type
  EnumType = (one, two, three, forty := 40, thirty := 30);
```

It is necessary to keep `forty` and `thirty` in the correct order. When using enumeration types it is important to keep the following points in mind:

1. The `Pred` and `Succ` functions cannot be used on this kind of enumeration types. Trying to do this anyhow will result in a compiler error.
2. Enumeration types stored using a default size. This behaviour can be changed with the `{ $PACKENUM n }` compiler directive, which tells the compiler the minimal number of bytes to be used for enumeration types. For instance

```

Type
{$PACKENUM 4}
  LargeEnum = ( BigOne, BigTwo, BigThree );
{$PACKENUM 1}
  SmallEnum = ( one, two, three );
Var S : SmallEnum;
    L : LargeEnum;
begin
  WriteLn ( 'Small enum : ', SizeOf(S));
  WriteLn ( 'Large enum : ', SizeOf(L));
end.

```

will, when run, print the following:

```


Small enum : 1
Large enum : 4

```

More information can be found in the [Programmers guide](#), in the compiler directives section.

Subrange types

A subrange type is a range of values from an ordinal type (the *host* type). To define a subrange type, one must specify its limiting values: the highest and lowest value of the type.



Some of the predefined integer types are defined as subrange types:

```

Type
  Longint  = $80000000..$7fffffff;
  Integer  = -32768..32767;
  shortint = -128..127;
  byte     = 0..255;
  Word     = 0..65535;

```

Subrange types of enumeration types can also be defined:

```

Type
  Days = (monday, tuesday, wednesday, thursday, friday,
          saturday, sunday);
  WorkDays = monday .. friday;
  WeekEnd = Saturday .. Sunday;

```

Real types

Free Pascal uses the math coprocessor (or emulation) for all its floating-point calculations. The Real native type is processor dependant, but it is either Single or Double. Only the IEEE floating point types are supported, and these depend on the target processor and emulation options. The true Turbo Pascal compatible types are listed in table (3.4). The Comp type is, in effect, a 64-bit integer and is not available on all target platforms. To get more information on the supported types for each platform, refer to the [Programmers guide](#).

Table 3.4: Supported Real types

Type	Range	Significant digits	Size
Real	platform dependant	???	4 or 8
Single	1.5E-45 .. 3.4E38	7-8	4
Double	5.0E-324 .. 1.7E308	15-16	8
Extended	1.9E-4951 .. 1.1E4932	19-20	10
Comp	-2E64+1 .. 2E63-1	19-20	8


3.2 Character types

Char

Free Pascal supports the type `Char`. A `Char` is exactly 1 byte in size, and contains one character. A character constant can be specified by enclosing the character in single quotes, as follows: `'a'` or `'A'` are both character constants. A character can also be specified by its ASCII value, by preceding the ASCII value with the number symbol (`#`). For example specifying `#65` would be the same as `'A'`. Also, the caret character (`^`) can be used in combination with a letter to specify a character with ASCII value less than 27. Thus `^G` equals `#7` (G is the seventh letter in the alphabet.) When the single quote character must be represented, it should be typed two times successively, thus `" "` represents the single quote character.

Strings

Free Pascal supports the `String` type as it is defined in Turbo Pascal (A sequence of characters with a specified length) and it supports `ansistring`s as in Delphi. To declare a variable as a string, use the following type specification:



The meaning of a string declaration statement is interpreted differently depending on the `{ $H }` switch. The above declaration can declare an `ansistring` or a short string.

Whatever the actual type, `ansistring`s and short strings can be used interchangeably. The compiler always takes care of the necessary type conversions. Note, however, that the result of an expression that contains `ansistring`s and short strings will always be an `ansistring`.

Short strings

A string declaration declares a short string in the following cases:

1. If the switch is off: `{ $H- }`, the string declaration will always be a short string declaration.
2. If the switch is on `{ $H+ }`, and there is a length specifier, the declaration is a short string declaration.

The predefined type `ShortString` is defined as a string of length 255:

```
ShortString = String[255];
```

If the size of the string is not specified, 255 is taken as a default. The length of the string can be obtained with the `Length` (163) standard runtime routine. For example in

```
{ $H- }
```

```
Type
```

```
  NameString = String[10];
  StreetString = String;
```

`NameString` can contain a maximum of 10 characters. While `StreetString` can contain up to 255 characters.

Ansistrings

Ansistrings are strings that have no length limit. They are reference counted and null terminated. Internally, an ansistring is treated as a pointer. This is all handled transparently, i.e. they can be manipulated as a normal short string. Ansistrings can be defined using the predefined `AnsiString` type.

If the `{ $H }` switch is on, then a string definition using the regular `String` keyword and that doesn't contain a length specifier, will be regarded as an ansistring as well. If a length specifier is present, a short string will be used, regardless of the `{ $H }` setting.

If the string is empty (`"`), then the internal pointer representation of the string pointer is `Nil`. If the string is not empty, then the pointer points to a structure in heap memory.

The internal representation as a pointer, and the automatic null-termination make it possible to type-cast an ansistring to a `pchar`. If the string is empty (so the pointer is nil) then the compiler makes sure that the typecasted `pchar` will point to a null byte.

Assigning one ansistring to another doesn't involve moving the actual string. A statement

```
S2 := S1;
```

results in the reference count of `S2` being decreased by one, The reference count of `S1` is increased by one, and finally `S1` (as a pointer) is copied to `S2`. This is a significant speed-up in the code.

If the reference count reaches zero, then the memory occupied by the string is deallocated automatically, so no memory leaks arise.

When an ansistring is declared, the Free Pascal compiler initially allocates just memory for a pointer, not more. This pointer is guaranteed to be nil, meaning that the string is initially empty. This is true for local and global ansistrings or anstrings that are part of a structure (arrays, records or objects).

This does introduce an overhead. For instance, declaring

```
Var
  A : Array[1..100000] of string;
```

Will copy 100,000 times nil into `A`. When `A` goes out of scope, then the reference count of the 100,000 strings will be decreased by 1 for each of these strings. All this happens invisibly for the programmer, but when considering performance issues, this is important.

Memory will be allocated only when the string is assigned a value. If the string goes out of scope, then its reference count is automatically decreased by 1. If the reference count reaches zero, the memory reserved for the string is released.

If a value is assigned to a character of a string that has a reference count greater than 1, such as in the following statements:

```
S:=T; { reference count for S and T is now 2 }  
S[I]:='@';
```

then a copy of the string is created before the assignment. This is known as *copy-on-write* semantics.

The `Length` (163) function must be used to get the length of an `ansistring`.

To set the length of an `ansistring`, the `SetLength` (182) function must be used. Constant `ansistrings` have a reference count of -1 and are treated specially.

`Ansistrings` are converted to short strings by the compiler if needed, this means that the use of `ansistrings` and short strings can be mixed without problems.

`Ansistrings` can be typecasted to `PChar` or `Pointer` types:

```
Var P : Pointer;  
    PC : PChar;  
    S : AnsiString;  
  
begin  
  S := 'This is an ansistring';  
  PC := PChar(S);  
  P := Pointer(S);
```

There is a difference between the two typecasts. When an empty `ansistring` is typecasted to a pointer, the pointer will be `Nil`. If an empty `ansistring` is typecasted to a `PChar`, then the result will be a pointer to a zero byte (an empty string).

The result of such a typecast must be used with care. In general, it is best to consider the result of such a typecast as read-only, i.e. suitable for passing to a procedure that needs a constant `pchar` argument.

It is therefore NOT advisable to typecast one of the following:

1. expressions.
2. strings that have reference count larger than 0. (call `uniquestring` to ensure a string has reference count 1)

WideStrings

`Widestrings` (used to represent unicode character strings) are implemented in much the same way as `ansistrings`: reference counted, null-terminated arrays, only they are implemented as arrays of `WideChars` instead of regular `Chars`. A `WideChar` is a two-byte character (an element of a DBCS: Double Byte Character Set). Mostly the same rules apply for `WideStrings` as for `Ansistrings`. The compiler transparently converts `WideStrings` to `Ansistrings` and vice versa.

Similarly to the typecast of an `Ansistring` to a `PChar` null-terminated array of characters, a `WideString` can be converted to a `PWideChar` null-terminated array of characters. Note that the `PWideChar` array is terminated by 2 null bytes instead of 1, so a typecast to a `pchar` is not automatic.

The compiler itself provides no support for any conversion from Unicode to `ansistrings` or vice versa; 2 procedural variables are present in the system unit which can be set to handle the conversion. For more information, see the system units reference.

Constant strings

To specify a constant string, it must be enclosed in single-quotes, just as a `Char` type, only now more than one character is allowed. Given that `S` is of type `String`, the following are valid assignments:

```
S := 'This is a string.';
S := 'One' + ' ', 'Two' + ' ', 'Three';
S := 'This isn't difficult !';
S := 'This is a weird character : '#145' !';
```

As can be seen, the single quote character is represented by 2 single-quote characters next to each other. Strange characters can be specified by their ASCII value. The example shows also that two strings can be added. The resulting string is just the concatenation of the first with the second string, without spaces in between them. Strings can not be subtracted, however.

Whether the constant string is stored as an ansistring or a short string depends on the settings of the `{ $H }` switch.

PChar - Null terminated strings

Free Pascal supports the Delphi implementation of the PChar type. PChar is defined as a pointer to a Char type, but allows additional operations. The PChar type can be understood best as the Pascal equivalent of a C-style null-terminated string, i.e. a variable of type PChar is a pointer that points to an array of type Char, which is ended by a null-character (#0). Free Pascal supports initializing of PChar typed constants, or a direct assignment. For example, the following pieces of code are equivalent:

```
program one;
var p : PChar;
begin
  p := 'This is a null-terminated string.';
  WriteLn (p);
end.
```

Results in the same as

```
program two;
const P : PChar = 'This is a null-terminated string.'
begin
  WriteLn (P);
end.
```

These examples also show that it is possible to write *the contents* of the string to a file of type Text. The **strings** unit contains procedures and functions that manipulate the PChar type as in the standard C library. Since it is equivalent to a pointer to a type Char variable, it is also possible to do the following:

```
Program three;
Var S : String[30];
    P : PChar;
begin
  S := 'This is a null-terminated string.'#0;
  P := @S[1];
  WriteLn (P);
end.
```

This will have the same result as the previous two examples. Null-terminated strings cannot be added as normal Pascal strings. If two PChar strings must be concatenated; the functions from the unit **strings** must be used.


However, it is possible to do some pointer arithmetic. The operators `+` and `-` can be used to do operations on `PChar` pointers. In table (3.5), `P` and `Q` are of type `PChar`, and `I` is of type `Longint`.

Table 3.5: `PChar` pointer arithmetic

Operation	Result
<code>P + I</code>	Adds <code>I</code> to the address pointed to by <code>P</code> .
<code>I + P</code>	Adds <code>I</code> to the address pointed to by <code>P</code> .
<code>P - I</code>	Subtracts <code>I</code> from the address pointed to by <code>P</code> .
<code>P - Q</code>	Returns, as an integer, the distance between 2 addresses (or the number of characters between <code>P</code> and <code>Q</code>)

3.3 Structured Types


A structured type is a type that can hold multiple values in one variable. Structured types can be nested to unlimited levels.



Unlike Delphi, Free Pascal does not support the keyword `Packed` for all structured types, as can be seen in the syntax diagram. It will be mentioned when a type supports the `packed` keyword. In the following, each of the possible structured types is discussed.

Arrays

Free Pascal supports arrays as in Turbo Pascal, multi-dimensional arrays and packed arrays are also supported, as well as the dynamic arrays of Delphi:



Static arrays

When the range of the array is included in the array definition, it is called a static array. Trying to access an element with an index that is outside the declared range will generate a run-time error (if range checking is on). The following is an example of a valid array declaration:

```
Type
  RealArray = Array [1..100] of Real;
```

Valid indexes for accessing an element of the array are between 1 and 100, where the borders 1 and 100 are included. As in Turbo Pascal, if the array component type is in itself an array, it is possible to combine the two arrays into one multi-dimensional array. The following declaration:

```
Type
  APoints = array[1..100] of Array[1..3] of Real;
```

is equivalent to the following declaration:

```
Type
  APoints = array[1..100,1..3] of Real;
```

The functions [High \(155\)](#) and [Low \(164\)](#) return the high and low bounds of the leftmost index type of the array. In the above case, this would be 100 and 1.

When static array-type variables are assigned to each other, the contents of the whole array is copied. This is also true for multi-dimensional arrays:

```
program testarray1;

Type
  TA = Array[0..9,0..9] of Integer;

var
  A,B : TA;
  I,J : Integer;
begin
  For I:=0 to 9 do
    For J:=0 to 9 do
      A[I,J]:=I*J;
  For I:=0 to 9 do
    begin
      For J:=0 to 9 do
        Write(A[I,J]:2,' ');
      Writeln;
    end;
  B:=A;
  Writeln;
  For I:=0 to 9 do
    For J:=0 to 9 do
      A[9-I,9-J]:=I*J;
  For I:=0 to 9 do
    begin
      For J:=0 to 9 do
        Write(B[I,J]:2,' ');
      Writeln;
```

```
    end;  
end.
```

The output will be 2 identical matrices.

Dynamic arrays

As of version 1.1, Free Pascal also knows dynamic arrays: In that case, the array range is omitted, as in the following example:

```
Type  
  TArray : Array of Byte;
```

When declaring a variable of a dynamic array type, the initial length of the array is zero. The actual length of the array must be set with the standard `SetLength` function, which will allocate the memory to contain the array elements on the heap. The following example will set the length to 1000:

```
Var  
  A : TArray;  
  
begin  
  SetLength(A,1000);
```

After a call to `SetLength`, valid array indexes are 0 to 999: the array index is always zero-based.

Note that the length of the array is set in elements, not in bytes of allocated memory (although these may be the same). The amount of memory allocated is the size of the array multiplied by the size of 1 element in the array. The memory will be disposed of at the exit of the current procedure or function.

It is also possible to resize the array: in that case, as much of the elements in the array as will fit in the new size, will be kept. The array can be resized to zero, which effectively resets the variable.

At all times, trying to access an element of the array that is not in the current length of the array will generate a run-time error.

Assignment of one dynamic array-type variable to another will let both variables point to the same array. Contrary to `ansistring`s, an assignment to an element of one array will be reflected in the other:

```
Var  
  A,B : TArray;  
  
begin  
  SetLength(A,10);  
  A[1]:=33;  
  B:=A;  
  A[1]:=31;
```

After the second assignment, the first element in B will also contain 31.

It can also be seen from the output of the following example:

```
program testarray1;  
  
Type  
  TA = Array of array of Integer;
```

```
var
  A,B : TA;
  I,J : Integer;
begin
  Setlength(A,10,10);
  For I:=0 to 9 do
    For J:=0 to 9 do
      A[I,J]:=I*J;
  For I:=0 to 9 do
    begin
      For J:=0 to 9 do
        Write(A[I,J]:2,' ');
      Writeln;
    end;
  B:=A;
  Writeln;
  For I:=0 to 9 do
    For J:=0 to 9 do
      A[9-I,9-J]:=I*J;
  For I:=0 to 9 do
    begin
      For J:=0 to 9 do
        Write(B[I,J]:2,' ');
      Writeln;
    end;
end.
```

The output will be a matrix of numbers, and then the same matrix, mirrored.

Dynamic arrays are reference counted: if in one of the previous examples A goes out of scope and B does not, then the array is not yet disposed of: the reference count of A (and B) is decreased with 1. As soon as the reference count reaches zero, the memory is disposed of.

It is also possible to copy and/or resize the array with the standard Copy function, which acts as the copy function for strings:


```
program testarray3;

Type
  TA = array of Integer;

var
  A,B : TA;
  I,J : Integer;
begin
  Setlength(A,10);
  For I:=0 to 9 do
    A[I]:=I;
  B:=Copy(A,3,9);
  For I:=0 to 5 do
    Writeln(B[I]);
end.
```

The Copy function will copy 9 elements of the array to a new array. Starting at the element at index 3 (i.e. the fourth element) of the array.

Free Pascal supports fixed records and records with variant parts. The syntax diagram for a record type is



```
Type
  Point = Record
    X,Y,Z : Real;
  end;

  RPoint = Record
    Case Boolean of
      False : (X,Y,Z : Real);
      True : (R,theta,phi : Real);
    end;

  BetterRPoint = Record
    Case UsePolar : Boolean of
      False : (X,Y,Z : Real);
      True : (R,theta,phi : Real);
    end;
```

Remark: It is possible to nest variant parts, as in:

35

```
Case byte of
  2 : (Y : Longint;
      case byte of
        3 : (Z : Longint);
      );
end;
```

The size of a record is the sum of the sizes of its fields, each size of a field is rounded up to a power of two. If the record contains a variant part, the size of the variant part is the size of the biggest variant, plus the size of the tag field type *if an identifier was declared for it*. Here also, the size of each part is first rounded up to two. So in the above example, `SizeOf (185)` would return 24 for `Point`, 24 for `RPoint` and 26 for `BetterRPoint`. For `MyRec`, the value would be 12. If a typed file with records, produced by a Turbo Pascal program, must be read, then chances are that attempting to read that file correctly will fail. The reason for this is that by default, elements of a record are aligned at 2-byte boundaries, for performance reasons. This default behaviour can be changed with the `{$PackRecords n}` switch. Possible values for `n` are 1, 2, 4, 16 or `Default`. This switch tells the compiler to align elements of a record or object or class that have size larger than `n` on `n` byte boundaries. Elements that have size smaller or equal than `n` are aligned on natural boundaries, i.e. to the first power of two that is larger than or equal to the size of the record element. The keyword `Default` selects the default value for the platform that the code is compiled for (currently, this is 2 on all platforms) Take a look at the following program:

```
Program PackRecordsDemo;
type
  {$PackRecords 2}
  Trec1 = Record
    A : byte;
    B : Word;
  end;

  {$PackRecords 1}
  Trec2 = Record
    A : Byte;
    B : Word;
  end;

  {$PackRecords 2}
  Trec3 = Record
    A,B : byte;
  end;

  {$PackRecords 1}
  Trec4 = Record
    A,B : Byte;
  end;

  {$PackRecords 4}
  Trec5 = Record
    A : Byte;
    B : Array[1..3] of byte;
    C : byte;
  end;

  {$PackRecords 8}
  Trec6 = Record
    A : Byte;
```

```
        B : Array[1..3] of byte;
        C : byte;
    end;
{$PackRecords 4}
Trec7 = Record
    A : Byte;
    B : Array[1..7] of byte;
    C : byte;
end;

{$PackRecords 8}
Trec8 = Record
    A : Byte;
    B : Array[1..7] of byte;
    C : byte;
end;
Var rec1 : Trec1;
    rec2 : Trec2;
    rec3 : TRec3;
    rec4 : TRec4;
    rec5 : Trec5;
    rec6 : TRec6;
    rec7 : TRec7;
    rec8 : TRec8;

begin
    Write ('Size Trec1 : ',SizeOf(Trec1));
    Writeln (' Offset B : ',Longint(@rec1.B)-Longint(@rec1));
    Write ('Size Trec2 : ',SizeOf(Trec2));
    Writeln (' Offset B : ',Longint(@rec2.B)-Longint(@rec2));
    Write ('Size Trec3 : ',SizeOf(Trec3));
    Writeln (' Offset B : ',Longint(@rec3.B)-Longint(@rec3));
    Write ('Size Trec4 : ',SizeOf(Trec4));
    Writeln (' Offset B : ',Longint(@rec4.B)-Longint(@rec4));
    Write ('Size Trec5 : ',SizeOf(Trec5));
    Writeln (' Offset B : ',Longint(@rec5.B)-Longint(@rec5),
            ' Offset C : ',Longint(@rec5.C)-Longint(@rec5));
    Write ('Size Trec6 : ',SizeOf(Trec6));
    Writeln (' Offset B : ',Longint(@rec6.B)-Longint(@rec6),
            ' Offset C : ',Longint(@rec6.C)-Longint(@rec6));
    Write ('Size Trec7 : ',SizeOf(Trec7));
    Writeln (' Offset B : ',Longint(@rec7.B)-Longint(@rec7),
            ' Offset C : ',Longint(@rec7.C)-Longint(@rec7));
    Write ('Size Trec8 : ',SizeOf(Trec8));
    Writeln (' Offset B : ',Longint(@rec8.B)-Longint(@rec8),
            ' Offset C : ',Longint(@rec8.C)-Longint(@rec8));
end.
```

The output of this program will be :

```
Size Trec1 : 4 Offset B : 2
Size Trec2 : 3 Offset B : 1
Size Trec3 : 2 Offset B : 1
Size Trec4 : 2 Offset B : 1
```

```

Size Trec5 : 8 Offset B : 4 Offset C : 7
Size Trec6 : 8 Offset B : 4 Offset C : 7
Size Trec7 : 12 Offset B : 4 Offset C : 11
Size Trec8 : 16 Offset B : 8 Offset C : 15

```

And this is as expected. In Trec1, since B has size 2, it is aligned on a 2 byte boundary, thus leaving an empty byte between A and B, and making the total size 4. In Trec2, B is aligned on a 1-byte boundary, right after A, hence, the total size of the record is 3. For Trec3, the sizes of A, B are 1, and hence they are aligned on 1 byte boundaries. The same is true for Trec4. For Trec5, since the size of B – 3 – is smaller than 4, B will be on a 4-byte boundary, as this is the first power of two that is larger than it's size. The same holds for Trec6. For Trec7, B is aligned on a 4 byte boundary, since it's size – 7 – is larger than 4. However, in Trec8, it is aligned on a 8-byte boundary, since 8 is the first power of two that is greater than 7, thus making the total size of the record 16. Free Pascal supports also the 'packed record', this is a record where all the elements are byte-aligned. Thus the two following declarations are equivalent:

```

{$PackRecords 1}
Trec2 = Record
  A : Byte;
  B : Word;
end;
{$PackRecords 2}

```

and

```

Trec2 = Packed Record
  A : Byte;
  B : Word;
end;

```

Note the `{ $PackRecords 2 }` after the first declaration !

Set types

Free Pascal supports the set types as in Turbo Pascal. The prototype of a set declaration is:

Set Types

➡ set type – **set** – **of** – ordinal type ➡

Each of the elements of `SetType` must be of type `TargetType`. `TargetType` can be any ordinal type with a range between 0 and 255. A set can contain maximally 255 elements. The following are valid set declaration:

Type

```

Junk = Set of Char;

Days = (Mon, Tue, Wed, Thu, Fri, Sat, Sun);
WorkDays : Set of days;

```

Given this set declarations, the following assignment is legal:

Table 3.6: Set Manipulation operators

Operation	Operator
Union	+
Difference	-
Intersection	*
Add element	include
Delete element	exclude

File types

File types

→ file type – **file** [of – type] →

```
Type
  Point = Record
    X,Y,Z : real;
  end;
  PointFile = File of Point;
```

A special file type is the `Text` file type, represented by the `TextRec` record. A file of type `Text` uses special input-output routines.

3.4 Pointers

39

Pointer types

↔ pointer type – ^ – type identifier ↔

As can be seen from this diagram, pointers are typed, which means that they point to a particular kind of data. The type of this data must be known at compile time. Dereferencing the pointer (denoted by adding ^ after the variable name) behaves then like a variable. This variable has the type declared in the pointer declaration, and the variable is stored in the address that is pointed to by the pointer variable. Consider the following example:

```
Program pointers;
type
  Buffer = String[255];
  BufPtr = ^Buffer;
Var B   : Buffer;
  BP   : BufPtr;
  PP   : Pointer;
etc..
```

In this example, BP is a pointer to a Buffer type; while B is a variable of type Buffer. B takes 256 bytes memory, and BP only takes 4 bytes of memory (enough to keep an address in memory).

Remark: Free Pascal treats pointers much the same way as C does. This means that a pointer to some type can be treated as being an array of this type. The pointer then points to the zeroeth element of this array. Thus the following pointer declaration

```
Var p : ^Longint;
```

Can be considered equivalent to the following array declaration:

```
Var p : array[0..Infinity] of Longint;
```

The difference is that the former declaration allocates memory for the pointer only (not for the array), and the second declaration allocates memory for the entire array. If the former is used, the memory must be allocated manually, using the [Getmem \(153\)](#) function. The reference P^ is then the same as p[0]. The following program illustrates this maybe more clear:

```
program PointerArray;
var i : Longint;
  p : ^Longint;
  pp : array[0..100] of Longint;
begin
  for i := 0 to 100 do pp[i] := i; { Fill array }
  p := @pp[0];                    { Let p point to pp }
  for i := 0 to 100 do
    if p[i] <> pp[i] then
      WriteLn ('Ohoh, problem !')
  end.
```

Free Pascal supports pointer arithmetic as C does. This means that, if P is a typed pointer, the instructions

```
Inc(P);  
Dec(P);
```

Will increase, respectively decrease the address the pointer points to with the size of the type P is a pointer to. For example

```
Var P : ^Longint;  
...  
Inc (p);
```

will increase P with 4. Normal arithmetic operators on pointers can also be used, that is, the following are valid pointer arithmetic operations:

```
var  p1,p2 : ^Longint;  
     L : Longint;  
begin  
  P1 := @P2;  
  P2 := @L;  
  L := P1-P2;  
  P1 := P1-4;  
  P2 := P2+4;  
end.
```

Here, the value that is added or subtracted *is* multiplied by the size of the type the pointer points to. In the previous example P1 will be decremented by 16 bytes, and P2 will be incremented by 16.

3.5 Forward type declarations

Programs often need to maintain a linked list of records. Each record then contains a pointer to the next record (and possibly to the previous record as well). For type safety, it is best to define this pointer as a typed pointer, so the next record can be allocated on the heap using the New call. In order to do so, the record should be defined something like this:

```
Type  
  TListItem = Record  
    Data : Integer;  
    Next : ^TListItem;  
  end;
```

When trying to compile this, the compiler will complain that the TListItem type is not yet defined when it encounters the Next declaration: This is correct, as the definition is still being parsed.

To be able to have the Next element as a typed pointer, a 'Forward type declaration' must be introduced:

```
Type  
  PListItem = ^TListItem;  
  TListItem = Record  
    Data : Integer;  
    Next : PListItem;  
  end;
```


When the compiler encounters a typed pointer declaration where the referenced type is not yet known, it postpones resolving the reference later on: The pointer definition is a 'Forward type declaration'.

The referenced type should be introduced later in the same `TYPE` block. No other block may come between the definition of the pointer type and the referenced type. Indeed, even the word `TYPE` itself may not re-appear: in effect it would start a new type-block, causing the compiler to resolve all pending declarations in the current block. In most cases, the definition of the referenced type will follow immediately after the definition of the pointer type, as shown in the above listing. The forward defined type can be used in any type definition following its declaration.

Note that a forward type declaration is only possible with pointer types and classes, not with other types.

3.6 Procedural types

Free Pascal has support for procedural types, although it differs a little from the Turbo Pascal implementation of them. The type declaration remains the same, as can be seen in the following syntax diagram:



For a description of formal parameter lists, see chapter 9, page 84. The two following examples are valid type declarations:

```

Type TOneArg = Procedure (Var X : integer);
      TNoArg = Function : Real;
var proc : TOneArg;
      func : TNoArg;

```

One can assign the following values to a procedural type variable:

1. `Nil`, for both normal procedure pointers and method pointers.
2. A variable reference of a procedural type, i.e. another variable of the same type.
3. A global procedure or function address, with matching function or procedure header and calling convention.
4. A method address.

Given these declarations, the following assignments are valid:

```
Procedure printit (Var X : Integer);
begin
  WriteLn (x);
end;
...
Proc := @printit;
Func := @Pi;
```

From this example, the difference with Turbo Pascal is clear: In Turbo Pascal it isn't necessary to use the address operator (@) when assigning a procedural type variable, whereas in Free Pascal it is required (unless the -So switch is used, in which case the address operator can be dropped.)

Remark: The modifiers concerning the calling conventions must be the same as the declaration; i.e. the following code would give an error:

```
Type TOneArgCcall = Procedure (Var X : integer);cdecl;
var proc : TOneArgCcall;
Procedure printit (Var X : Integer);
begin
  WriteLn (x);
end;
begin
Proc := @printit;
end.
```

Because the TOneArgCcall type is a procedure that uses the cdecl calling convention.

3.7 Variant types

Definition

As of version 1.1, FPC has support for variants. For variant support to be enabled, the `variants` unit must be included in every unit that uses variants in some way. Furthermore, the compiler must be in Delphi or ObjFPC mode.

The type of a value stored in a variant is only determined at runtime: it depends what has been assigned to the to the variant. Almost any type can be assigned to variants: ordinal types, string types, int64 types. Structured types such as sets, records, arrays, files, objects and classes are not assign-compatible with a variant, as well as pointers. Interfaces and COM or CORBA objects can be assigned to a variant.

This means that the following assignments are valid:

```
Type
  TMyEnum = (One,Two,Three);

Var
  V : Variant;
  I : Integer;
  B : Byte;
  W : Word;
  Q : Int64;
  E : Extended;
  D : Double;
  En : TMyEnum;
```

```
AS : AnsiString;  
WS : WideString;  
  
begin  
  V:=I;  
  V:=B;  
  V:=W;  
  V:=Q;  
  V:=E;  
  V:=En;  
  V:=D;  
  V:=AS;  
  V:=WS;  
end;
```

And of course vice-versa as well.

Remark: The enumerated type assignment is broken in the early 1.1 development series of the compiler. It is expected that this is fixed soon.

A variant can hold an array of values: All elements in the array have the same type (but can be of type 'variant'). For a variant that contains an array, the variant can be indexed:

```
Program testv;  
  
uses variants;  
  
Var  
  A : Variant;  
  I : integer;  
  
begin  
  A:=VarArrayCreate([1,10],varInteger);  
  For I:=1 to 10 do  
    A[I]:=I;  
end.
```

(for the explanation of `VarArrayCreate`, see [Unit reference](#).)

Note that when the array contains a string, this is not considered an 'array of characters', and so the variant cannot be indexed to retrieve a character at a certain position in the string.

Remark: The array functionality is broken in the early 1.1 development series of the compiler. It is expected that this is fixed soon.

Variants in assignments and expressions

As can be seen from the definition above, most simple types can be assigned to a variant. Likewise, a variant can be assigned to a simple type: If possible, the value of the variant will be converted to the type that is being assigned to. This may fail: Assigning a variant containing a string to an integer will fail unless the string represents a valid integer. In the following example, the first assignment will work, the second will fail:

```
program testv3;  
  
uses Variants;
```

```
Var
  V : Variant;
  I : Integer;

begin
  V:= '100';
  I:=V;
  Writeln('I : ',I);
  V:= 'Something else';
  I:=V;
  Writeln('I : ',I);
end.
```

The first assignment will work, but the second will not, as `Something else` cannot be converted to a valid integer value. An `EConvertError` exception will be the result.

The result of an expression involving a variant will be of type variant again, but this can be assigned to a variable of a different type - if the result can be converted to a variable of this type.

Note that expressions involving variants take more time to be evaluated, and should therefore be used with caution. If a lot of calculations need to be made, it is best to avoid the use of variants.

Variants and interfaces

Remark: Dispatch interface support for variants is currently broken in the compiler.

Variants can contain a reference to an interface - a normal interface (descending from `IInterface`) or a dispatchinterface (descending from `IDispatch`). Variants containing a reference to a dispatch interface can be used to control the object behind it: the compiler will use late binding to perform the call to the dispatch interface: there will be no run-time checking of the function names and parameters or arguments given to the functions. The result type is also not checked. The compiler will simply insert code to make the dispatch call and retrieve the result.

This means basically, that you can do the following on Windows:

```
Var
  W : Variant;
  V : String;

begin
  W:=CreateOleObject('Word.Application');
  V:=W.Application.Version;
  Writeln('Installed version of MS Word is : ',V);
end;
```

The line

```
V:=W.Application.Version;
```

is executed by inserting the necessary code to query the dispatch interface stored in the variant `W`, and execute the call if the needed dispatch information is found.

Chapter 4


Objects


4.1 Declaration

Free Pascal supports object oriented programming. In fact, most of the compiler is written using objects. Here we present some technical questions regarding object oriented programming in Free Pascal. Objects should be treated as a special kind of record. The record contains all the fields that are declared in the objects definition, and pointers to the methods that are associated to the objects' type.

An object is declared just as a record would be declared; except that now, procedures and functions can be declared as if they were part of the record. Objects can "inherit" fields and methods from "parent" objects. This means that these fields and methods can be used as if they were included in the objects declared as a "child" object.

Furthermore, a concept of visibility is introduced: fields, procedures and functions can be declared as `public` or `private`. By default, fields and methods are `public`, and are exported outside the current unit. Fields or methods that are declared `private` are only accessible in the current unit. The prototype declaration of an object is as follows:





As can be seen, as many `private` and `public` blocks as needed can be declared. Method definitions are normal function or procedure declarations. Fields cannot be declared after methods in the same block, i.e. the following will generate an error when compiling:

```
Type MyObj = Object
  Procedure Doit;
  Field : Longint;
end;
```

But the following will be accepted:

```
Type MyObj = Object
  Public
    Procedure Doit;
  Private
    Field : Longint;
end;
```

because the field is in a different section.

Remark: Free Pascal also supports the packed object. This is the same as an object, only the elements (fields) of the object are byte-aligned, just as in the packed record. The declaration of a packed object is similar to the declaration of a packed record :

```
Type
  TObj = packed object;
  Constructor init;
  ...
end;
Pobj = ^TObj;
Var PP : Pobj;
```

Similarly, the `{ $PackRecords }` directive acts on objects as well.

4.2 Fields

Object Fields are like record fields. They are accessed in the same way as a record field would be accessed : by using a qualified identifier. Given the following declaration:

```
Type TAnObject = Object
  AField : Longint;
  Procedure AMethod;
end;
Var AnObject : TAnObject;
```

then the following would be a valid assignment:

```
AnObject.AField := 0;
```


Inside methods, fields can be accessed using the short identifier:

```
Procedure TAnObject.AMethod;
begin
    ...
    AField := 0;
    ...
end;
```

Or, one can use the `self` identifier. The `self` identifier refers to the current instance of the object:

```
Procedure TAnObject.AMethod;
begin
    ...
    Self.AField := 0;
    ...
end;
```


One cannot access fields that are in a private section of an object from outside the objects' methods. If this is attempted anyway, the compiler will complain about an unknown identifier. It is also possible to use the `with` statement with an object instance:

```
With AnObject do
begin
    Afield := 12;
    AMethod;
end;
```

In this example, between the `begin` and `end`, it is as if `AnObject` was prepended to the `Afield` and `AMethod` identifiers. More about this in section 8.2, page 80

4.3 Constructors and destructors

As can be seen in the syntax diagram for an object declaration, Free Pascal supports constructors and destructors. The programmer is responsible for calling the constructor and the destructor explicitly when using objects. The declaration of a constructor or destructor is as follows:



A constructor/destructor pair is *required* if the object uses virtual methods. In the declaration of the object type, a simple identifier should be used for the name of the constructor or destructor. When the constructor or destructor is implemented, a qualified method identifier should be used, i.e. an identifier of the form `objectidentifier.methodidentifier`. Free Pascal supports also the extended syntax of the `New` and `Dispose` procedures. In case a dynamic variable of an object type must be allocated the constructor's name can be specified in the call to `New`. The `New` is implemented as a function which returns a pointer to the instantiated object. Consider the following declarations:

```
Type
  TObj = object;
    Constructor init;
    ...
  end;
  PObj = ^TObj;
Var PP : PObj;
```

Then the following 3 calls are equivalent:

```
pp := new (PObj, Init);
```

and

```
new(pp, init);
```

and also

```
new (pp);
pp^.init;
```

In the last case, the compiler will issue a warning that the extended syntax of `new` and `dispose` must be used to generate instances of an object. It is possible to ignore this warning, but it's better programming practice to use the extended syntax to create instances of an object. Similarly, the `Dispose` procedure accepts the name of a destructor. The destructor will then be called, before removing the object from the heap.

In view of the compiler warning remark, the following chapter presents the Delphi approach to object-oriented programming, and may be considered a more natural way of object-oriented programming.

4.4 Methods

Object methods are just like ordinary procedures or functions, only they have an implicit extra parameter: `self`. `Self` points to the object with which the method was invoked. When implementing methods, the fully qualified identifier must be given in the function header. When declaring methods, a normal identifier must be given.

4.5 Method invocation

Methods are called just as normal procedures are called, only they have an object instance identifier prepended to them (see also chapter 8, page 73). To determine which method is called, it is necessary to know the type of the method. We treat the different types in what follows.

Static methods

Static methods are methods that have been declared without a `abstract` or `virtual` keyword. When calling a static method, the declared (i.e. compile time) method of the object is used. For example, consider the following declarations:

```
Type
  TParent = Object
    ...
    procedure Doit;
    ...
  end;
  PParent = ^TParent;
  TChild = Object(TParent)
    ...
    procedure Doit;
    ...
  end;
  PChild = ^TChild;
```

As it is visible, both the parent and child objects have a method called `Doit`. Consider now the following declarations and calls:

```
Var ParentA, ParentB : PParent;
    Child           : PChild;
    ParentA := New(PParent, Init);
    ParentB := New(PChild, Init);
    Child := New(PChild, Init);
    ParentA^.Doit;
    ParentB^.Doit;
    Child^.Doit;
```

Of the three invocations of `Doit`, only the last one will call `TChild.Doit`, the other two calls will call `TParent.Doit`. This is because for static methods, the compiler determines at compile time which method should be called. Since `ParentB` is of type `TParent`, the compiler decides that it must be called with `TParent.Doit`, even though it will be created as a `TChild`. There may be times when the method that is actually called should depend on the actual type of the object at run-time. If so, the method cannot be a static method, but must be a virtual method.

Virtual methods

To remedy the situation in the previous section, virtual methods are created. This is simply done by appending the method declaration with the `virtual` modifier. Going back to the previous example, consider the following alternative declaration:

```
Type
  TParent = Object
    ...
    procedure Doit; virtual;
    ...
  end;
  PParent = ^TParent;
  TChild = Object(TParent)
    ...
```

```
procedure Doit;virtual;  
...  
end;  
PChild = ^TChild;
```

As it is visible, both the parent and child objects have a method called `Doit`. Consider now the following declarations and calls :

```
Var ParentA,ParentB : PParent;  
    Child           : PChild;  
ParentA := New(PParent,Init);  
ParentB := New(PChild,Init);  
Child := New(PChild,Init);  
ParentA^.Doit;  
ParentB^.Doit;  
Child^.Doit;
```

Now, different methods will be called, depending on the actual run-time type of the object. For `ParentA`, nothing changes, since it is created as a `TParent` instance. For `Child`, the situation also doesn't change: it is again created as an instance of `TChild`. For `ParentB` however, the situation does change: Even though it was declared as a `TParent`, it is created as an instance of `TChild`. Now, when the program runs, before calling `Doit`, the program checks what the actual type of `ParentB` is, and only then decides which method must be called. Seeing that `ParentB` is of type `TChild`, `TChild.Doit` will be called. The code for this run-time checking of the actual type of an object is inserted by the compiler at compile time. The `TChild.Doit` is said to *override* the `TParent.Doit`. It is possible to access the `TParent.Doit` from within the `varTChild.Doit`, with the `inherited` keyword:

```
Procedure TChild.Doit;  
begin  
    inherited Doit;  
    ...  
end;
```

In the above example, when `TChild.Doit` is called, the first thing it does is call `TParent.Doit`. The `inherited` keyword cannot be used in static methods, only on virtual methods.

Abstract methods

An abstract method is a special kind of virtual method. A method can not be abstract if it is not virtual (this is not obvious from the syntax diagram). An instance of an object that has an abstract method cannot be created directly. The reason is obvious: there is no method where the compiler could jump to ! A method that is declared `abstract` does not have an implementation for this method. It is up to inherited objects to override and implement this method. Continuing our example, take a look at this:

```
Type  
TParent = Object  
...  
procedure Doit;virtual;abstract;  
...  
end;  
PParent=^TParent;  
TChild = Object(TParent)
```

```
...  
procedure Doit;virtual;  
...  
end;  
PChild = ^TChild;
```

As it is visible, both the parent and child objects have a method called `Doit`. Consider now the following declarations and calls :

```
Var ParentA,ParentB : PParent;  
    Child           : PChild;  
ParentA := New(PParent,Init);  
ParentB := New(PChild,Init);  
Child := New(PChild,Init);  
ParentA^.Doit;  
ParentB^.Doit;  
Child^.Doit;
```

First of all, Line 3 will generate a compiler error, stating that one cannot generate instances of objects with abstract methods: The compiler has detected that `PParent` points to an object which has an abstract method. Commenting line 3 would allow compilation of the program.

Remark: If an abstract method is overridden, The parent method cannot be called with `inherited`, since there is no parent method; The compiler will detect this, and complain about it, like this:

```
testo.pp(32,3) Error: Abstract methods can't be called directly
```

If, through some mechanism, an abstract method is called at run-time, then a run-time error will occur. (run-time error 211, to be precise)

4.6 Visibility

For objects, 3 visibility specifiers exist : `private`, `protected` and `public`. If a visibility specifier is not specified, `public` is assumed. Both methods and fields can be hidden from a programmer by putting them in a `private` section. The exact visibility rule is as follows:

Private All fields and methods that are in a `private` block, can only be accessed in the module (i.e. unit or program) that contains the object definition. They can be accessed from inside the object's methods or from outside them e.g. from other objects' methods, or global functions.

Protected Is the same as `Private`, except that the members of a `Protected` section are also accessible to descendent types, even if they are implemented in other modules.

Public sections are always accessible, from everywhere. Fields and methods in a `public` section behave as though they were part of an ordinary record type.

Chapter 5


Classes


In the Delphi approach to Object Oriented Programming, everything revolves around the concept of 'Classes'. A class can be seen as a pointer to an object, or a pointer to a record.

Remark: In earlier versions of Free Pascal it was necessary, in order to use classes, to put the `objpas` unit in the uses clause of a unit or program. *This is no longer needed* as of version 0.99.12. As of version 0.99.12 the `system` unit contains the basic definitions of `TObject` and `TClass`, as well as some auxiliary methods for using classes. The `objpas` unit still exists, and contains some redefinitions of basic types, so they coincide with Delphi types. The unit will be loaded automatically when the `-S2` or `-Sd` options are specified.

5.1 Class definitions

The prototype declaration of a class is as follows :





As many `private`, `protected`, `published` and `public` blocks as needed can be repeated. Methods are normal function or procedure declarations. As can be seen, the declaration of a class is almost identical to the declaration of an object. The real difference between objects and classes is in the way they are created (see further in this chapter). The visibility of the different sections is as follows:


Private All fields and methods that are in a `private` block, can only be accessed in the module (i.e. unit) that contains the class definition. They can be accessed from inside the classes' methods or from outside them (e.g. from other classes' methods)

Protected Is the same as `Private`, except that the members of a `Protected` section are also accessible to descendent types, even if they are implemented in other modules.

Public sections are always accessible.

Published Is the same as a `Public` section, but the compiler generates also type information that is needed for automatic streaming of these classes. Fields defined in a `published` section must be of class type. Array peroperties cannot be in a `published` section.

It is also possible to define class reference types:



Class reference types are used to create instances of a certain class, which is not yet known at compile time, but which is specified at run time. Essentially, a variable of a class reference type contains a pointer to the VMT of the specified class. This can be used to construct an instance of the class corresponding to the VMT. The following example shows how it works:

```

Type
  TComponentClass = Class of TComponent;

Function CreateComponent(AClass : TComponentClass; AOwner : TComponent) : TComponent;

begin
  // ...
  Result:=AClass.Create(AOwner);
  // ...
end;
  
```

More about instantiating a class can be found in the next section.

5.2 Class instantiation

Classes must be created using their constructor. Remember that a class is a pointer to an object, so when a variable of some class is declared, the compiler just allocates a pointer, not the entire object. The constructor of a class returns a pointer to an initialized instance of the object. So, to initialize an instance of some class, one would do the following :

```
ClassVar := ClassType.ConstructorName;
```

The extended syntax of `new` and `dispose` can be used to instantiate and destroy class instances. That construct is reserved for use with objects only. Calling the constructor will provoke a call to `getmem`, to allocate enough space to hold the class instance data. After that, the constructor's code is executed. The constructor has a pointer to its data, in `self`.

Remark:

- The `{ $PackRecords }` directive also affects classes. i.e. the alignment in memory of the different fields depends on the value of the `{ $PackRecords }` directive.
- Just as for objects and records, a packed class can be declared. This has the same effect as on an object, or record, namely that the elements are aligned on 1-byte boundaries. i.e. as close as possible.
- `SizeOf(class)` will return 4, since a class is but a pointer to an object. To get the size of the class instance data, use the `TObject.InstanceSize` method.

5.3 Methods

invocation

Method invocation for classes is no different than for objects. The following is a valid method invocation:

```
Var AnObject : TAnObject;
begin
  AnObject := TAnObject.Create;
  AnObject.AMethod;
```

Virtual methods

Classes have virtual methods, just as objects do. There is however a difference between the two. For objects, it is sufficient to redeclare the same method in a descendent object with the keyword `virtual` to override it. For classes, the situation is different: virtual methods *must* be overridden with the `override` keyword. Failing to do so, will start a *new* batch of virtual methods, hiding the previous one. The `Inherited` keyword will not jump to the inherited method, if `virtual` was used.

The following code is *wrong*:

```
Type
  ObjParent = Class
    Procedure MyProc; virtual;
  end;
  ObjChild = Class(ObjParent)
    Procedure MyProc; virtual;
  end;
```


The compiler will produce a warning:

Warning: An inherited method is hidden by OBJCHILD.MYPROC

The compiler will compile it, but using `Inherited` can produce strange effects.

The correct declaration is as follows:

```
Type ObjParent = Class
    Procedure MyProc; virtual;
end;
ObjChild = Class(ObjParent)
    Procedure MyProc; override;
end;
```

This will compile and run without warnings or errors.

Message methods

New in classes are message methods. Pointers to message methods are stored in a special table, together with the integer or string constant that they were declared with. They are primarily intended to ease programming of callback functions in several GUI toolkits, such as Win32 or GTK. In difference with Delphi, Free Pascal also accepts strings as message identifiers.

Message methods that are declared with an integer constant can take only one var argument (typed or not):

```
Procedure TMyObject.MyHandler(Var Msg); Message 1;
```

The method implementation of a message function is no different from an ordinary method. It is also possible to call a message method directly, but this should not be done. Instead, the `TObject.Dispatch` method should be used.

The `TObject.Dispatch` method can be used to call a message handler. It is declared in the `system` unit and will accept a var parameter which must have at the first position a cardinal with the message ID that should be called. For example:


```
Type
    TMsg = Record
        MSGID : Cardinal
        Data : Pointer;
Var
    Msg : TMsg;

MyObject.Dispatch (Msg);
```

In this example, the `Dispatch` method will look at the object and all its ancestors (starting at the object, and searching up the class tree), to see if a message method with message `MSGID` has been declared. If such a method is found, it is called, and passed the `Msg` parameter.

If no such method is found, `DefaultHandler` is called. `DefaultHandler` is a virtual method of `TObject` that doesn't do anything, but which can be overridden to provide any processing that might be needed. `DefaultHandler` is declared as follows:

```
procedure defaulthandler(var message);virtual;
```

A read specifier is either the name of a field that contains the property, or the name of a method function that has the same return type as the property type. In the case of a simple type, this function must not accept an argument. A read specifier is optional, making the property write-only. A write specifier is optional: If there is no write specifier, the property is read-only. A write specifier is either the name of a field, or the name of a method procedure that accepts as a sole argument a variable of the same type as the property. The section (private, published) in which the specified function or procedure resides is irrelevant. Usually, however, this will be a protected or private method. Example: Given the following declaration:

Type

```

MyClass = Class
  Private
    Field1 : Longint;
    Field2 : Longint;
    Field3 : Longint;
    Procedure Sety (value : Longint);
    Function Gety : Longint;
    Function Getz : Longint;
  Public
    Property X : Longint Read Field1 write Field2;
    Property Y : Longint Read GetY Write Sety;
    Property Z : Longint Read GetZ;
  end;
Var MyClass : TMyClass;
  
```

The following are valid statements:

```

WriteLn ('X : ', MyClass.X);
WriteLn ('Y : ', MyClass.Y);
WriteLn ('Z : ', MyClass.Z);
MyClass.X := 0;
MyClass.Y := 0;
  
```

But the following would generate an error:

```
MyClass.Z := 0;
```

because Z is a read-only property. What happens in the above statements is that when a value needs to be read, the compiler inserts a call to the various `getNNN` methods of the object, and the result of this call is used. When an assignment is made, the compiler passes the value that must be assigned as a parameter to the various `setNNN` methods. Because of this mechanism, properties cannot be passed as var arguments to a function or procedure, since there is no known address of the property (at least, not always). If the property definition contains an index, then the read and write specifiers must be a function and a procedure. Moreover, these functions require an additional parameter: An integer parameter. This allows to read or write several properties with the same function. For this, the properties must have the same type. The following is an example of a property with an index:

```
{ $mode objfpc }
Type TPoint = Class(TObject)
  Private
    FX, FY : Longint;
    Function GetCoord (Index : Integer): Longint;
    Procedure SetCoord (Index : Integer; Value : longint);
  Public
    Property X : Longint index 1 read GetCoord Write SetCoord;
    Property Y : Longint index 2 read GetCoord Write SetCoord;
    Property Coords[Index : Integer]:Longint Read GetCoord;
  end;
Procedure TPoint.SetCoord (Index : Integer; Value : Longint);
begin
  Case Index of
    1 : FX := Value;
    2 : FY := Value;
  end;
end;
Function TPoint.GetCoord (Index : Integer) : Longint;
begin
  Case Index of
    1 : Result := FX;
    2 : Result := FY;
  end;
end;
Var P : TPoint;
begin
  P := TPoint.create;
  P.X := 2;
  P.Y := 3;
  With P do
    WriteLn ('X=', X, ' Y=', Y);
  end.
```

When the compiler encounters an assignment to X, then `SetCoord` is called with as first parameter the index (1 in the above case) and with as a second parameter the value to be set. Conversely, when reading the value of X, the compiler calls `GetCoord` and passes it index 1. Indexes can only be integer values. Array properties also exist. These are properties that accept an index, just as an array does. Only now the index doesn't have to be an ordinal type, but can be any type.

A `read` specifier for an array property is the name method function that has the same return type as the property type. The function must accept as a sole argument a variable of the same type as the index type. For an array property, one cannot specify fields as `read` specifiers.

A `write` specifier for an array property is the name of a method procedure that accepts two arguments: The first argument has the same type as the index, and the second argument is a parameter of the same type as the property type. As an example, see the following declaration:

```
Type TIntList = Class
  Private
    Function GetInt (I : Longint) : longint;
    Function GetAsString (A : String) : String;
    Procedure SetInt (I : Longint; Value : Longint);
    Procedure SetAsString (A : String; Value : String);
  Public
    Property Items [i : Longint] : Longint Read GetInt
                                          Write SetInt;
    Property StrItems [S : String] : String Read GetAsString
                                          Write SetAsString;
end;
Var AIntList : TIntList;
```

Then the following statements would be valid:

```
AIntList.Items[26] := 1;
AIntList.StrItems['twenty-five'] := 'zero';
WriteLn ('Item 26 : ',AIntList.Items[26]);
WriteLn ('Item 25 : ',AIntList.StrItems['twenty-five']);
```

While the following statements would generate errors:

```
AIntList.Items['twenty-five'] := 1;
AIntList.StrItems[26] := 'zero';
```

Because the index types are wrong. Array properties can be declared as default properties. This means that it is not necessary to specify the property name when assigning or reading it. If, in the previous example, the definition of the items property would have been

```
Property Items[i : Longint]: Longint Read GetInt
                                          Write SetInt; Default;
```

Then the assignment

```
AIntList.Items[26] := 1;
```

Would be equivalent to the following abbreviation.

```
AIntList[26] := 1;
```

Only one default property per class is allowed, and descendent classes cannot redeclare the default property.

Chapter 6

Interfaces


6.1 Definition

As of version 1.1, FPC supports interfaces. Interfaces are an alternative to multiple inheritance (where a class can have multiple parent classes) as implemented for instance in C++. An interface is basically a named set of methods and properties: A class that *implements* the interface provides *all* the methods as they are enumerated in the Interface definition. It is not possible for a class to implement only part of the interface: it is all or nothing.

Interfaces can also be ordered in a hierarchy, exactly as classes: An interface definition that inherits from another interface definition contains all the methods from the parent interface, as well as the methods explicitly named in the interface definition. A class implementing an interface must then implement all members of the interface as well as the methods of the parent interface(s).

An interface can be uniquely identified by a GUID (GUID is an acronym for Globally Unique Identifier, a 128-bit integer guaranteed always to be unique¹). Especially on Windows systems, the GUID of an interface can and must be used when using COM.

The definition of an Interface has the following form:



Along with this definition the following must be noted:

- Interfaces can only be used in DELPHI mode or in OBJFPC mode.
- There are no visibility specifiers. All members are public (indeed, it would make little sense to make them private or protected).

¹In theory, of course.

- The properties declared in an interface can only have methods as read and write specifiers.
- There are no constructors or destructors. Instances of interfaces cannot be created directly: instead, an instance of a class implementing the interface must be created.
- Only calling convention modifiers may be present in the definition of a method. Modifiers as `virtual`, `abstract` or `dynamic`, and hence also `override` cannot be present in the definition of a interface definition.

6.2 Identification: A GUID

An interface can be identified by a GUID. This is a 128-bit number, which is represented in a text representation (a string literal):

```
[ ' { H H H H H H H H - H H H H - H H H H - H H H H H H H H H H H H } ' ]
```

Each H character represents a hexadecimal number (0-9,A-F). The format contains 8-4-4-4-12 numbers. A GUID can also be represented by the following record, defined in the `objpas` unit (included automatically when in `DELPHI` or `OBJFPC` mode:

```
PGuid = ^TGuid;
TGuid = packed record
  case integer of
    1 : (
      Data1 : DWord;
      Data2 : word;
      Data3 : word;
      Data4 : array[0..7] of byte;
    );
    2 : (
      D1 : DWord;
      D2 : word;
      D3 : word;
      D4 : array[0..7] of byte;
    );
  end;
end;
```

A constant of type `TGUID` can be specified using a string literal:

```
{ $mode objfpc }
program testuid;

Const
  MyGUID : TGUID = ' { 10101010-1010-0101-1001-110110110110 } ' ;

begin
end.
```

6.3 Interfaces and COM

When using interfaces on Windows which should be available to the COM subsystem, the calling convention should be `stdcall` - this is not the default Free Pascal calling convention, so it should be specified explicitly.

COM does not know properties. It only knows methods. So when specifying property definitions as part of an interface definition, be aware that the properties will only be known in the Free Pascal compiled program: other Windows programs will not be aware of the property definitions. For this reason, property definitions must always have interface methods as the read/write specifiers.

Interface implementations

When a class implements an interface, it should implement all methods of the interface. If a method of an interface is not implemented, then the compiler will give an error. For example:

Type

```
IMyInterface = Interface
  Function MyFunc : Integer;
  Function MySecondFunc : Integer;
end;

TMyClass = Class(TInterfacedObject,IMyInterface)
  Function MyFunc : Integer;
  Function MyOtherFunc : Integer;
end;
```

```
Function TMyClass.MyFunc : Integer;
```

```
begin
  Result:=23;
end;
```

```
Function TMyClass.MyOtherFunc : Integer;
```

```
begin
  Result:=24;
end;
```

will result in a compiler error:

```
Error: No matching implementation for interface method
"IMyInterface.MySecondFunc:LongInt" found
```

At the moment of writing, the compiler does not yet support providing aliases for an interface as in Delphi. i.e. the following will not yet compile:

type

```
IMyInterface = Interface
  Function MyFunc : Integer;
end;

TMyClass = Class(TInterfacedObject,IMyInterface)
  Function MyOtherFunction : Integer;
  // The following fails in FPC.
  Function IMyInterface.MyFunc = MyOtherFunction;
end;
```

This declaration should tell the compiler that the MyFunc method of the IMyInterface interface is implemented in the MyOtherFunction method of the TMyClass class.

Chapter 7

Expressions

Expressions occur in assignments or in tests. Expressions produce a value, of a certain type. Expressions are built with two components: Operators and their operands. Usually an operator is binary, i.e. it requires 2 operands. Binary operators occur always between the operands (as in X/Y). Sometimes an operator is unary, i.e. it requires only one argument. A unary operator occurs always before the operand, as in $-X$.

When using multiple operands in an expression, the precedence rules of table (7.1) are used. When

Table 7.1: Precedence of operators

Operator	Precedence	Category
Not, @	Highest (first)	Unary operators
* / div mod and shl shr as	Second	Multiplying operators
+ - or xor	Third	Adding operators
< <> < > <= >= in is	Lowest (Last)	relational operators

determining the precedence, the compiler uses the following rules:

1. In operations with unequal precedences the operands belong to the operator with the highest precedence. For example, in $5*3+7$, the multiplication is higher in precedence than the addition, so it is executed first. The result would be 22.
2. If parentheses are used in an expression, their contents is evaluated first. Thus, $5*(3+7)$ would result in 50.

Remark: The order in which expressions of the same precedence are evaluated is not guaranteed to be left-to-right. In general, no assumptions on which expression is evaluated first should be made in such a case. The compiler will decide which expression to evaluate first based on optimization rules. Thus, in the following expression:

```
a := g(3) + f(2);
```


$f(2)$ may be executed before $g(3)$. This behaviour is distinctly different from Delphior Turbo Pascal.

If one expression *must* be executed before the other, it is necessary to split up the statement using temporary results:

```
e1 := g(3);
a  := e1 + f(2);
```

7.1 Expression syntax

An expression applies relational operators to simple expressions. Simple expressions are a series of terms (what a term is, is explained below), joined by adding operators.




The following are valid expressions:

```
GraphResult<>grError
(DoItToday=Yes) and (DoItTomorrow=No);
Day in Weekend
```

And here are some simple expressions:

```
A + B
-Pi
ToBe or NotToBe
```


Terms consist of factors, connected by multiplication operators.



Here are some valid terms:


```
2 * Pi
A Div B
(DoItToday=Yes) and (DoItTomorrow=No);
```

Factors are all other constructions:



7.2 Function calls

Function calls are part of expressions (although, using extended syntax, they can be statements too). They are constructed as follows:



The `variable reference` must be a procedural type variable reference. A method designator can only be used inside the method of an object. A qualified method designator can be used outside object methods too. The function that will get called is the function with a declared parameter list that matches the actual parameter list. This means that

1. The number of actual parameters must equal the number of declared parameters (unless default parameter values are used).

2. The types of the parameters must be compatible. For variable reference parameters, the parameter types must be exactly the same.

If no matching function is found, then the compiler will generate an error. Depending on the fact of the function is overloaded (i.e. multiple functions with the same name, but different parameter lists) the error will be different. There are cases when the compiler will not execute the function call in an expression. This is the case when assigning a value to a procedural type variable, as in the following example:

```
Type
  FuncType = Function: Integer;
Var A : Integer;
Function AddOne : Integer;
begin
  A := A+1;
  AddOne := A;
end;
Var F : FuncType;
    N : Integer;
begin
  A := 0;
  F := AddOne; { Assign AddOne to F, Don't call AddOne }
  N := AddOne; { N := 1 !! }
end.
```

In the above listing, the assignment to F will not cause the function AddOne to be called. The assignment to N, however, will call AddOne. A problem with this syntax is the following construction:

```
If F = AddOne Then
  DoSomethingHorrible;
```

Should the compiler compare the addresses of F and AddOne, or should it call both functions, and compare the result? Free Pascal solves this by deciding that a procedural variable is equivalent to a pointer. Thus the compiler will give a type mismatch error, since AddOne is considered a call to a function with integer result, and F is a pointer. Hence a type mismatch occurs. How then, should one compare whether F points to the function AddOne? To do this, one should use the address operator @:

```
If F = @AddOne Then
  WriteLn ('Functions are equal');
```


The left hand side of the boolean expression is an address. The right hand side also, and so the compiler compares 2 addresses. How to compare the values that both functions return? By adding an empty parameter list:

```
If F()=Addone then
  WriteLn ('Functions return same values ');
```

Remark that this behaviour is not compatible with Delphi syntax.

7.3 Set constructors

When a set-type constant must be entered in an expression, a set constructor must be given. In essence this is the same thing as when a type is defined, only there is no identifier to identify the set with. A set constructor is a comma separated list of expressions, enclosed in square brackets.



All set groups and set elements must be of the same ordinal type. The empty set is denoted by `[]`, and it can be assigned to any type of set. A set group with a range `[A..Z]` makes all values in the range a set element. If the first range specifier has a bigger ordinal value than the second the set is empty, e.g., `[Z..A]` denotes an empty set. The following are valid set constructors:


```

[ today, tomorrow ]
[ Monday..Friday, Sunday ]
[ 2, 3*2, 6*2, 9*2 ]
[ 'A'..'Z', 'a'..'z', '0'..'9' ]

```

7.4 Value typecasts

Sometimes it is necessary to change the type of an expression, or a part of the expression, to be able to be assignment compatible. This is done through a value typecast. The syntax diagram for a value typecast is as follows:



Value typecasts cannot be used on the left side of assignments, as variable typecasts. Here are some valid typecasts:

```

Byte('A')
Char(48)
boolean(1)
longint(@Buffer)

```

The type size of the expression and the size of the type cast must be the same. That is, the following doesn't work:

```

Integer('A')
Char(4875)
boolean(100)
Word(@Buffer)

```

This is different from Delphi or Turbo Pascal behaviour.

Table 7.2: Binary arithmetic operators

Operator	Operation
+	Addition
-	Subtraction
*	Multiplication
/	Division
Div	Integer division
Mod	Remainder

Table 7.3: Unary arithmetic operators

Operator	Operation
+	Sign identity
-	Sign inversion

```
I mod J = I - (I div J) * J
```

but it executes faster than the right hand side expression.

Logical operators

Logical operators act on the individual bits of ordinal expressions. Logical operators require operands that are of an integer type, and produce an integer type result. The possible logical operators are listed in table (7.4). The following are valid logical expressions:

Table 7.4: Logical operators

Operator	Operation
not	Bitwise negation (unary)
and	Bitwise and
or	Bitwise or
xor	Bitwise xor
shl	Bitwise shift to the left
shr	Bitwise shift to the right

```
A shr 1 { same as A div 2, but faster }
Not 1   { equals -2 }
Not 0   { equals -1 }
Not -1  { equals 0 }
B shl 2 { same as B * 2 for integers }
1 or 2  { equals 3 }
3 xor 1 { equals 2 }
```

Boolean operators

Boolean operators can be considered logical operations on a type with 1 bit size. Therefore the `shl` and `shr` operations have little sense. Boolean operators can only have boolean type operands, and the resulting type is always boolean. The possible operators are listed in table (7.5)

Table 7.5: Boolean operators

Operator	Operation
<code>not</code>	logical negation (unary)
<code>and</code>	logical and
<code>or</code>	logical or
<code>xor</code>	logical xor

Remark: Boolean expressions are always evaluated with short-circuit evaluation. This means that from the moment the result of the complete expression is known, evaluation is stopped and the result is returned. For instance, in the following expression:

```
B := True or MaybeTrue;
```

The compiler will never look at the value of `MaybeTrue`, since it is obvious that the expression will always be true. As a result of this strategy, if `MaybeTrue` is a function, it will not get called ! (This can have surprising effects when used in conjunction with properties)

String operators

There is only one string operator : `+`. It's action is to concatenate the contents of the two strings (or characters) it stands between. One cannot use `+` to concatenate null-terminated (`PChar`) strings. The following are valid string operations:

```
'This is ' + 'VERY ' + 'easy !'  
Dirname+'\'
```

The following is not:

```
Var Dirname = PChar;  
...  
Dirname := Dirname+'\';
```

Because `Dirname` is a null-terminated string.

Set operators

The following operations on sets can be performed with operators: Union, difference and intersection. The operators needed for this are listed in table (7.6). The set type of the operands must be the same, or an error will be generated by the compiler.

Relational operators

The relational operators are listed in table (7.7) Left and right operands must be of the same type.

Table 7.6: Set operators

Operator	Action
+	Union
-	Difference
*	Intersection

Table 7.7: Relational operators


Operator	Action
=	Equal
<>	Not equal
<	Strictly less than
>	Strictly greater than
<=	Less than or equal
>=	Greater than or equal
in	Element of

Only integer and real types can be mixed in relational expressions. Comparing strings is done on the basis of their ASCII code representation. When comparing pointers, the addresses to which they point are compared. This also is true for `PChar` type pointers. To compare the strings the `PChar` point to, the `StrComp` function from the `strings` unit must be used. The `in` returns `True` if the left operand (which must have the same ordinal type as the set type, and which must be in the range `0..255`) is an element of the set which is the right operand, otherwise it returns `False`.

Chapter 8

Statements


The heart of each algorithm are the actions it takes. These actions are contained in the statements of a program or unit. Each statement can be labeled and jumped to (within certain limits) with `Goto` statements. This can be seen in the following syntax diagram:



A label can be an identifier or an integer digit.

8.1 Simple statements


A simple statement cannot be decomposed in separate statements. There are basically 4 kinds of simple statements:



Of these statements, the *raise statement* will be explained in the chapter on Exceptions (chapter [12](#), page [110](#))

Assignments

Assignments give a value to a variable, replacing any previous value the variable might have had:



In addition to the standard Pascal assignment operator (`:=`), which simply replaces the value of the variable with the value resulting from the expression on the right of the `:=` operator, Free Pascal supports some c-style constructions. All available constructs are listed in table (8.1). For these

Table 8.1: Allowed C constructs in Free Pascal

Assignment	Result
<code>a += b</code>	Adds <code>b</code> to <code>a</code> , and stores the result in <code>a</code> .
<code>a -= b</code>	Subtracts <code>b</code> from <code>a</code> , and stores the result in <code>a</code> .
<code>a *= b</code>	Multiplies <code>a</code> with <code>b</code> , and stores the result in <code>a</code> .
<code>a /= b</code>	Divides <code>a</code> through <code>b</code> , and stores the result in <code>a</code> .

constructs to work, the `-Sc` command-line switch must be specified.

Remark: These constructions are just for typing convenience, they don't generate different code. Here are some examples of valid assignment statements:


```

X := X+Y;
X+=Y;      { Same as X := X+Y, needs -Sc command line switch}
X/=2;      { Same as X := X/2, needs -Sc command line switch}
Done := False;
Weather := Good;
MyPi := 4* Tan(1);

```

Procedure statements

Procedure statements are calls to subroutines. There are different possibilities for procedure calls: A normal procedure call, an object method call (fully qualified or not), or even a call to a procedural type variable. All types are present in the following diagram.



The Free Pascal compiler will look for a procedure with the same name as given in the procedure statement, and with a declared parameter list that matches the actual parameter list. The following are valid procedure statements:

```
Usage;
WriteLn('Pascal is an easy language !');
Doit();
```

Goto statements

Free Pascal supports the `goto` jump statement. Its prototype syntax is



When using `goto` statements, the following must be kept in mind:


1. The jump label must be defined in the same block as the `Goto` statement.
2. Jumping from outside a loop to the inside of a loop or vice versa can have strange effects.
3. To be able to use the `Goto` statement, the `-Sg` compiler switch must be used.

`Goto` statements are considered bad practice and should be avoided as much as possible. It is always possible to replace a `goto` statement by a construction that doesn't need a `goto`, although this construction may not be as clear as a `goto` statement. For instance, the following is an allowed `goto` statement:


```
label
  jump to;
...
Jump to :
  Statement;
...
Goto jump to;
...
```

8.2 Structured statements


Structured statements can be broken into smaller simple statements, which should be executed repeatedly, conditionally or sequentially:



Conditional statements come in 2 flavours :

Conditional statements


Repetitive statements come in 3 flavours:

Repetitive statements


The following sections deal with each of these statements.

Compound statements

Compound statements are a group of statements, separated by semicolons, that are surrounded by the keywords `Begin` and `End`. The Last statement doesn't need to be followed by a semicolon, although it is allowed. A compound statement is a way of grouping statements together, executing the statements sequentially. They are treated as one statement in cases where Pascal syntax expects 1 statement, such as in `if ... then` statements.

Compound statements**The Case statement**

Free Pascal supports the `case` statement. Its syntax diagram is

Case statement

The constants appearing in the various case parts must be known at compile-time, and can be of the following types : enumeration types, Ordinal types (except boolean), and chars. The expression must be also of this type, or a compiler error will occur. All case constants must have the same type. The compiler will evaluate the expression. If one of the case constants values matches the value of the expression, the statement that follows this constant is executed. After that, the program continues after the final end. If none of the case constants match the expression value, the statement after the else keyword is executed. This can be an empty statement. If no else part is present, and no case constant matches the expression value, program flow continues after the final end. The case statements can be compound statements (i.e. a begin . . End block).

Remark: Contrary to Turbo Pascal, duplicate case labels are not allowed in Free Pascal, so the following code will generate an error when compiling:

```
Var i : integer;
...
Case i of
  3 : DoSomething;
  1..5 : DoSomethingElse;
end;
```

The compiler will generate a Duplicate case label error when compiling this, because the 3 also appears (implicitly) in the range 1 . . 5. This is similar to Delphi syntax.

The following are valid case statements:

```
Case C of
  'a' : WriteLn ('A pressed');
  'b' : WriteLn ('B pressed');
  'c' : WriteLn ('C pressed');
else
  WriteLn ('unknown letter pressed : ',C);
end;
```

Or

```
Case C of
  'a','e','i','o','u' : WriteLn ('vowel pressed');
  'y' : WriteLn ('This one depends on the language');
else
  WriteLn ('Consonant pressed');
end;
```

```
Case Number of
  1..10 : WriteLn ('Small number');
  11..100 : WriteLn ('Normal, medium number');
else
  WriteLn ('HUGE number');
end;
```

The If...then...else statement

The If .. then .. else.. prototype syntax is

If then statements



The expression between the `if` and `then` keywords must have a boolean return type. If the expression evaluates to `True` then the statement following `then` is executed.

If the expression evaluates to `False`, then the statement following `else` is executed, if it is present.

Be aware of the fact that the boolean expression will be short-cut evaluated. (Meaning that the evaluation will be stopped at the point where the outcome is known with certainty) Also, before the `else` keyword, no semicolon (;) is allowed, but all statements can be compound statements. In nested `If... then... else` constructs, some ambiguity may arise as to which `else` statement pairs with which `if` statement. The rule is that the `else` keyword matches the first `if` keyword not already matched by an `else` keyword. For example:

```

If exp1 Then
  If exp2 then
    Stat1
else
  stat2;

```

Despite its appearance, the statement is syntactically equivalent to

```

If exp1 Then
  begin
    If exp2 then
      Stat1
    else
      stat2
  end;

```

and not to

```

{ NOT EQUIVALENT }
If exp1 Then
  begin
    If exp2 then
      Stat1
    end
else
  stat2

```

If it is this latter construct is needed, the `begin` and `end` keywords must be present. When in doubt, it is better to add them.

The following is a valid statement:

```

If Today in [Monday..Friday] then
  WriteLn ('Must work harder')
else
  WriteLn ('Take a day off.');
```

The `For...to/downto...do` statement

Free Pascal supports the `For` loop construction. A `for` loop is used in case one wants to calculate something a fixed number of times. The prototype syntax is as follows:

For statement

The diagram illustrates the syntax of a For statement. It consists of five horizontal lines, each starting with a right-pointing arrowhead. The first line contains the text "for statement – **for** – control variable – **:=** – initial value" followed by a bracketed section containing "to" and "downto". The second line contains "final value – **do** – statement". The third line contains "control variable – variable identifier". The fourth line contains "initial value – expression". The fifth line contains "final value – expression".

► for statement – **for** – control variable – **:=** – initial value to
downto ►

► final value – **do** – statement ►

► control variable – variable identifier ►

► initial value – expression ►

► final value – expression ►

The following are valid loops:

Repeat statement

repeat statement – **repeat** statement **until** – expression

;

This will execute the statements between `repeat` and `until` up to the moment when `Expression` evaluates to `True`. Since the expression is evaluated *after* the execution of the statements, they are executed at least once. Be aware of the fact that the boolean expression `Expression` will be short-cut evaluated. (Meaning that the evaluation will be stopped at the point where the outcome is known with certainty) The following are valid `repeat` statements

```
repeat
  WriteLn ('I =', i);
  I := I+2;
until I>100;
repeat
  X := X/2
until x<10e-3
```

The `Break` (132) and `Continue` (139) reserved words can be used to jump to the end or just after the end of the `repeat .. until` statement.

The `while..do` statement

A `while` statement is used to execute a statement as long as a certain condition holds. This may imply that the statement is never executed. The prototype syntax of the `While..do` statement is

While statements

→ while statement – **while** – expression – **do** – statement →

This will execute `Statement` as long as `Expression` evaluates to `True`. Since `Expression` is evaluated *before* the execution of `Statement`, it is possible that `Statement` isn't executed at all. `Statement` can be a compound statement. Be aware of the fact that the boolean expression `Expression` will be short-cut evaluated. (Meaning that the evaluation will be stopped at the point where the outcome is known with certainty) The following are valid `while` statements:


```
I := I+2;
while i<=100 do
  begin
    WriteLn ('I =', i);
    I := I+2;
  end;
X := X/2;
while x>=10e-3 do
  X := X/2;
```

They correspond to the example loops for the `repeat` statements.

If the statement is a compound statement, then the `Break` (132) and `Continue` (139) reserved words can be used to jump to the end or just after the end of the `While` statement.

The `with` statement

The `with` statement serves to access the elements of a record or object or class, without having to specify the name of the each time. The syntax for a `with` statement is



The variable reference must be a variable of a record, object or class type. In the `with` statement, any variable reference, or method reference is checked to see if it is a field or method of the record or object or class. If so, then that field is accessed, or that method is called. Given the declaration:

```

Type Passenger = Record
    Name : String[30];
    Flight : String[10];
end;
Var TheCustomer : Passenger;

```

The following statements are completely equivalent:

```

TheCustomer.Name := 'Michael';
TheCustomer.Flight := 'PS901';

```

and

```

With TheCustomer do
begin
    Name := 'Michael';
    Flight := 'PS901';
end;

```

The statement

```

With A,B,C,D do Statement;

```

is equivalent to

```

With A do
    With B do
        With C do
            With D do Statement;

```

This also is a clear example of the fact that the variables are tried *last to first*, i.e., when the compiler encounters a variable reference, it will first check if it is a field or method of the last variable. If not, then it will check the last-but-one, and so on. The following example shows this;

```

Program testw;
Type AR = record
    X,Y : Longint;
end;
PAR = Record;

```

```

Var S,T : Ar;
begin
    S.X := 1;S.Y := 1;

```

The output of this program is

2 2

Showing thus that the X, Y in the `WriteLn` statement match the T record variable.

Remark: When using a `With` statement with a pointer, or a class, it is not permitted to change the pointer or the class in the `With` block. With the definitions of the previous example, the following illustrates what it is about:

```
Var p : PAR;

begin
  With P^ do
    begin
      // Do some operations
      P:=OtherP;
      X:=0.0; // Wrong X will be used !!
    end;
  end;
end;
```


The reason the pointer cannot be changed is that the address is stored by the compiler in a temporary register. Changing the pointer won't change the temporary address. The same is true for classes.

Exception Statements

Free Pascal supports exceptions. Exceptions provide a convenient way to program error and error-recovery mechanisms, and are closely related to classes. Exception support is explained in [chapter 12](#), [page 110](#)

8.3 Assembler statements

An assembler statement allows to insert assembler code right in the pascal code.



More information about assembler blocks can be found in the [Programmers guide](#). The register list is used to indicate the registers that are modified by an assembler statement in the assembler block. The compiler stores certain results in the registers. If the registers are modified in an assembler statement,

the compiler should, sometimes, be told about it. The registers are denoted with their Intel names for the I386 processor, i.e., 'EAX', 'ESI' etc... As an example, consider the following assembler code:

```
asm
  Movl $1,%ebx
  Movl $0,%eax
  addl %eax,%ebx
end; [ 'EAX', 'EBX' ];
```

This will tell the compiler that it should save and restore the contents of the EAX and EBX registers when it encounters this asm statement.

Free Pascal supports various styles of assembler syntax. By default, AT&T syntax is assumed for the 80386 and compatibles platform. The default assembler style can be changed with the {`$asmmode xxx`} switch in the code, or the `-R` command-line option. More about this can be found in the [Programmers guide](#).

Chapter 9


Using functions and procedures

Free Pascal supports the use of functions and procedures, but with some extras: Function overloading is supported, as well as `Const` parameters and open arrays.

Remark: In many of the subsequent paragraphs the words `procedure` and `function` will be used interchangeably. The statements made are valid for both, except when indicated otherwise.

9.1 Procedure declaration

A procedure declaration defines an identifier and associates it with a block of code. The procedure can then be called with a procedure statement.




See section 9.3, page 85 for the list of parameters. A procedure declaration that is followed by a block implements the action of the procedure in that block. The following is a valid procedure :

```
Procedure DoSomething (Para : String);
begin
  Writeln ('Got parameter : ', Para);
  Writeln ('Parameter in upper case : ', Upper(Para));
end;
```

Note that it is possible that a procedure calls itself.

9.2 Function declaration


A function declaration defines an identifier and associates it with a block of code. The block of code will return a result. The function can then be called inside an expression, or with a procedure statement, if extended syntax is on.



The result type of a function can be any previously declared type. contrary to Turbo pascal, where only simple types could be returned.

9.3 Parameter lists

When arguments must be passed to a function or procedure, these parameters must be declared in the formal parameter list of that function or procedure. The parameter list is a declaration of identifiers that can be referred to only in that procedure or function's block.




Constant parameters and variable parameters can also be untyped parameters if they have no type identifier.

As of version 1.1, Free Pascal supports default values for both constant parameters and value parameters, but only for simple types. The compiler must be in OBJFPC or DELPHI mode to accept default values.

Value parameters

Value parameters are declared as follows:



When parameters are declared as value parameters, the procedure gets *a copy* of the parameters that the calling block passes. Any modifications to these parameters are purely local to the procedure's block, and do not propagate back to the calling block. A block that wishes to call a procedure with value parameters must pass assignment compatible parameters to the procedure. This means that the types should not match exactly, but can be converted (conversion code is inserted by the compiler itself)

Care must be taken when using value parameters: Value parameters makes heavy use of the stack, especially when using large parameters. The total size of all parameters in the formal parameter list should be below 32K for portability's sake (the Intel version limits this to 64K).

Open arrays can be passed as value parameters. See section 9.3, page 88 for more information on using open arrays.

For a parameter of a simple type (i.e. not a structured type), a default value can be specified. This can be an untyped constant. If the function call omits the parameter, the default value will be passed on to the function. For dynamic arrays or other types that can be considered as equivalent to a pointer, the only possible default value is Nil.

The following example will print 20 on the screen:

```
program testp;

Const
  MyConst = 20;


Procedure MyRealFunc(I : Integer = MyConst);


begin
  Writeln('Function received : ',I);
end;

begin
  MyRealFunc;
end.
```

Variable parameters

Variable parameters are declared as follows:





When parameters are declared as variable parameters, the procedure or function accesses immediately the variable that the calling block passed in its parameter list. The procedure gets a pointer to the variable that was passed, and uses this pointer to access the variable's value. From this, it follows that any changes made to the parameter, will propagate back to the calling block. This mechanism can be used to pass values back in procedures. Because of this, the calling block must pass a parameter of *exactly* the same type as the declared parameter's type. If it does not, the compiler will generate an error.

Variable and constant parameters can be untyped. In that case the variable has no type, and hence is incompatible with all other types. However, the address operator can be used on it, or it can be passed to a function that has also an untyped parameter. If an untyped parameter is used in an assignment, or a value must be assigned to it, a typecast must be used.


File type variables must always be passed as variable parameters.

Open arrays can be passed as variable parameters. See section 9.3, page 88 for more information on using open arrays.

Note that default values are not supported for variable parameters. This would make little sense since it defeats the purpose of being able to pass a value back to the caller.

Out parameters

Out parameters (output parameters) are declared as follows:




The purpose of an `out` parameter is to pass values back to the calling routine: The variable is passed by reference. The initial value of the parameter on function entry is discarded, and should not be used.


If a variable must be used to pass a value to a function and retrieve data from the function, then a variable parameter must be used. If only a value must be retrieved, a `out` parameter can be used.

Needless to say, default values are not supported for `out` parameters.

Constant parameters

In addition to variable parameters and value parameters Free Pascal also supports Constant parameters. A constant parameter as can be specified as follows:





A constant argument is passed by reference if it's size is larger than a pointer. It is passed by value if the size is equal or is less then the size of a native pointer. This means that the function or procedure receives a pointer to the passed argument, but it cannot be assigned to, this will result in a compiler error. Furthermore a const parameter cannot be passed on to another function that requires a variable parameter. The main use for this is reducing the stack size, hence improving performance, and still retaining the semantics of passing by value...

Constant parameters can also be untyped. See section 9.3, page 86 for more information about untyped parameters.

As for value parameters, constant parameters can get default values.

Open arrays can be passed as constant parameters. See section 9.3, page 88 for more information on using open arrays.

Open array parameters

Free Pascal supports the passing of open arrays, i.e. a procedure can be declared with an array of unspecified length as a parameter, as in Delphi. Open array parameters can be accessed in the procedure or function as an array that is declared with starting index 0, and last element index `High(parameter)`. For example, the parameter

```
Row : Array of Integer;
```

would be equivalent to

```
Row : Array[0..N-1] of Integer;
```

Where N would be the actual size of the array that is passed to the function. N-1 can be calculated as `High(Row)`. Open parameters can be passed by value, by reference or as a constant parameter. In the latter cases the procedure receives a pointer to the actual array. In the former case, it receives a copy of the array. In a function or procedure, open arrays can only be passed to functions which are also declared with open arrays as parameters, *not* to functions or procedures which accept arrays of fixed length. The following is an example of a function using an open array:

```
Function Average (Row : Array of integer) : Real;
Var I : longint;
    Temp : Real;
begin
    Temp := Row[0];
    For I := 1 to High(Row) do
        Temp := Temp + Row[i];
    Average := Temp / (High(Row)+1);
end;
```

Array of const

In Object Pascal or Delphi mode, Free Pascal supports the `Array of Const` construction to pass parameters to a subroutine.

This is a special case of the Open array construction, where it is allowed to pass any expression in an array to a function or procedure.

In the procedure, passed the arguments can be examined using a special record:

Type

```
PVarRec = ^TVarRec;
TVarRec = record
  case VType : Longint of
    vtInteger      : (VInteger: Longint);
    vtBoolean      : (VBoolean: Boolean);
    vtChar         : (VChar: Char);
    vtExtended     : (VExtended: PExtended);
    vtString       : (VString: PShortString);
    vtPointer      : (VPointer: Pointer);
    vtPChar        : (VPChar: PChar);
    vtObject       : (VObject: TObject);
    vtClass        : (VClass: TClass);
    vtAnsiString   : (VAnsiString: Pointer);
    vtWideString   : (VWideString: Pointer);
    vtInt64        : (VInt64: PInt64);
  end;
```

Inside the procedure body, the array of const is equivalent to an open array of TVarRec:

```
Procedure Testit (Args: Array of const);
```

```
Var I : longint;
```

```
begin
```

```
  If High(Args)<0 then
```

```
    begin
```

```
      Writeln ('No arguments');
```

```
      exit;
```

```
    end;
```

```
  Writeln ('Got ',High(Args)+1,' arguments :');
```

```
  For i:=0 to High(Args) do
```

```
    begin
```

```
      write ('Argument ',i,' has type ');
```

```
      case Args[i].vtype of
```

```
        vtinteger      :
```

```
          Writeln ('Integer, Value : ',args[i].vinteger);
```

```
        vtboolean      :
```

```
          Writeln ('Boolean, Value : ',args[i].vboolean);
```

```
        vtchar         :
```

```
          Writeln ('Char, value : ',args[i].vchar);
```

```
        vtextended     :
```

```
          Writeln ('Extended, value : ',args[i].VExtended^);
```

```
        vtString       :
```

```
          Writeln ('ShortString, value : ',args[i].VString^);
```

```
        vtPointer      :
```

```
          Writeln ('Pointer, value : ',Longint(Args[i].VPointer));
```

```
        vtPChar        :
```

```
          Writeln ('PChar, value : ',Args[i].VPChar);
```

```
        vtObject       :
```

```

        Writeln ('Object, name : ',Args[i].VObject.Classname);
    vtClass      :
        Writeln ('Class reference, name : ',Args[i].VClass.Classname);
    vtAnsiString :
        Writeln ('AnsiString, value : ',AnsiString(Args[I].VAnsiStr
else
        Writeln ('(Unknown) : ',args[i].vtype);
end;
end;
end;

```

In code, it is possible to pass an arbitrary array of elements to this procedure:

```

S:='Ansistring 1';
T:='AnsiString 2';
Testit ([]);
Testit ([1,2]);
Testit (['A','B']);
Testit ([TRUE,FALSE,TRUE]);
Testit (['String','Another string']);
Testit ([S,T]) ;
Testit ([P1,P2]);
Testit ([@testit,Nil]);
Testit ([ObjA,ObjB]);
Testit ([1.234,1.234]);
TestIt ([AClass]);

```

If the procedure is declared with the `cdecl` modifier, then the compiler will pass the array as a C compiler would pass it. This, in effect, emulates the C construct of a variable number of arguments, as the following example will show:

```

program testaocc;
{$mode objfpc}

Const
    P : Pchar = 'example';
    Fmt : PChar =
        'This %s uses printf to print numbers (%d) and strings.'#10;

// Declaration of standard C function printf:
procedure printf (fm : pchar; args : array of const);cdecl; external 'c';

begin
    printf(Fmt,[P,123]);
end.

```

Remark that this is not true for Delphi, so code relying on this feature will not be portable.

9.4 Function overloading

Function overloading simply means that the same function is defined more than once, but each time with a different formal parameter list. The parameter lists must differ at least in one of its elements type. When the compiler encounters a function call, it will look at the function parameters to decide

which one of the defined functions it should call. This can be useful when the same function must be defined for different types. For example, in the RTL, the `Dec` procedure could be defined as:

```
...
Dec(Var I : Longint;decrement : Longint);
Dec(Var I : Longint);
Dec(Var I : Byte;decrement : Longint);
Dec(Var I : Byte);
...
```

When the compiler encounters a call to the `dec` function, it will first search which function it should use. It therefore checks the parameters in a function call, and looks if there is a function definition which matches the specified parameter list. If the compiler finds such a function, a call is inserted to that function. If no such function is found, a compiler error is generated. Functions that have a `cdecl` modifier cannot be overloaded. (Technically, because this modifier prevents the mangling of the function name by the compiler).

Prior to version 1.9 of the compiler, the overloaded functions needed to be in the same unit. Now the compiler will continue searching in other units if it doesn't find a matching version of an overloaded function in one unit.

The compiler accepts the presence of the `overload` modifier as in Delphi, but it is not required, unless in Delphi mode.

9.5 Forward defined functions

A function can be declared without having it followed by its implementation, by having it followed by the `forward` procedure. The effective implementation of that function must follow later in the module. The function can be used after a `forward` declaration as if it had been implemented already. The following is an example of a forward declaration.

```
Program testforward;
Procedure First (n : longint); forward;
Procedure Second;
begin
  WriteLn ('In second. Calling first...');
  First (1);
end;
Procedure First (n : longint);
begin
  WriteLn ('First received : ',n);
end;
begin
  Second;
end.
```

A function can be defined as `forward` only once. Likewise, in units, it is not allowed to have a `forward` declared function of a function that has been declared in the interface part. The interface declaration counts as a `forward` declaration. The following unit will give an error when compiled:

```
Unit testforward;
interface
Procedure First (n : longint);
Procedure Second;
```


```

implementation
Procedure First (n : longint); forward;
Procedure Second;
begin
    WriteLn ('In second. Calling first...');
    First (1);
end;
Procedure First (n : longint);
begin
    WriteLn ('First received : ',n);
end;
end.

```

9.6 External functions

The `external` modifier can be used to declare a function that resides in an external object file. It allows to use the function in some code, and at linking time, the object file containing the implementation of the function or procedure must be linked in.



It replaces, in effect, the function or procedure code block. As an example:

```

program CmodDemo;
{$Linklib c}
Const P : PChar = 'This is fun !';
Function strlen (P : PChar) : Longint; cdecl; external;
begin
    WriteLn ('Length of (' ,p,') : ',strlen(p))
end.

```

Remark: The parameters in our declaration of the `external` function should match exactly the ones in the declaration in the object file.

If the `external` modifier is followed by a string constant:

```
external 'lname';
```

Then this tells the compiler that the function resides in library 'lname'. The compiler will then automatically link this library to the program.

The name that the function has in the library can also be specified:

```
external 'lname' name 'Fname';
```

This tells the compiler that the function resides in library 'lname', but with name 'Fname'. The compiler will then automatically link this library to the program, and use the correct name for the function. Under WINDOWS and OS/2, the following form can also be used:

```
external 'lname' Index Ind;
```

This tells the compiler that the function resides in library 'lname', but with index Ind. The compiler will then automatically link this library to the program, and use the correct index for the function.

Finally, the external directive can be used to specify the external name of the function :

```
{ $L myfunc.o }
external name 'Fname' ;
```

This tells the compiler that the function has the name 'Fname'. The correct library or object file (in this case myfunc.o) must still be linked. so that the function 'Fname' is included in the linking stage.

9.7 Assembler functions

Functions and procedures can be completely implemented in assembly language. To indicate this, use the `assembler` keyword:

Assembler functions


→ asm block – **assembler** – ; – declaration part – asm statement →

Contrary to Delphi, the assembler keyword must be present to indicate an assembler function. For more information about assembler functions, see the chapter on using assembler in the [Programmers guide](#).

9.8 Modifiers

A function or procedure declaration can contain modifiers. Here we list the various possibilities:

Modifiers



Free Pascal doesn't support all Turbo Pascal modifiers, but does support a number of additional modifiers. They are used mainly for assembler and reference to C object files.

alias

The `alias` modifier allows the programmer to specify a different name for a procedure or function. This is mostly useful for referring to this procedure from assembly language constructs or from another object file. As an example, consider the following program:

```
Program Aliases;

Procedure Printit;alias : 'DOIT';
begin
  WriteLn ('In Printit (alias : "DOIT")');
end;
begin
  asm
    call DOIT
  end;
end.
```

Remark: the specified alias is inserted straight into the assembly code, thus it is case sensitive.

The `alias` modifier does not make the symbol public to other modules, unless the routine is also declared in the interface part of a unit, or the `public` modifier is used to force it as public. Consider the following:

```
unit testalias;

interface

procedure testroutine;

implementation

procedure testroutine;alias:'ARoutine';
begin
  WriteLn('Hello world');
end;

end.
```

This will make the routine `testroutine` available publicly to external object files under the label name `ARoutine`.

cdecl

The `cdecl` modifier can be used to declare a function that uses a C type calling convention. This must be used when accessing functions residing in an object file generated by standard C compilers. It allows to use the function in the code, and at linking time, the object file containing the C implementation of the function or procedure must be linked in. As an example:

```
program CmodDemo;
{$LINKLIB c}
Const P : PChar = 'This is fun !';
Function strlen (P : PChar) : Longint; cdecl; external name 'strlen';
begin
```

```
WriteLn ('Length of (' ,p,') : ',strlen(p))  
end.
```

When compiling this, and linking to the C-library, the `strlen` function can be called throughout the program. The `external` directive tells the compiler that the function resides in an external object filelibrary with the 'strlen' name (see 9.6).

Remark: The parameters in our declaration of the C function should match exactly the ones in the declaration in C.

export

The `export` modifier is used to export names when creating a shared library or an executable program. This means that the symbol will be publicly available, and can be imported from other programs. For more information on this modifier, consult the section on Programming dynamic libraries in the [Programmers guide](#).

inline

Procedures that are declared `inline` are copied to the places where they are called. This has the effect that there is no actual procedure call, the code of the procedure is just copied to where the procedure is needed, this results in faster execution speed if the function or procedure is used a lot.

By default, `inline` procedures are not allowed. Inline code must be enabled using the command-line switch `-Si` or `{$inline on}` directive.

1. Inline code is NOT exported from a unit. This means that when calling an inline procedure from another unit, a normal procedure call will be performed. Only inside units, Inline procedures are really inlined.
2. Recursive inline functions are not allowed. i.e. an inline function that calls itself is not allowed.

interrupt

The `interrupt` keyword is used to declare a routine which will be used as an interrupt handler. On entry to this routine, all the registers will be saved and on exit, all registers will be restored and an interrupt or trap return will be executed (instead of the normal return from subroutine instruction).

On platforms where a return from interrupt does not exist, the normal exit code of routines will be done instead. For more information on the generated code, consult the [Programmers guide](#).

pascal

The `pascal` modifier can be used to declare a function that uses the classic pascal type calling convention (passing parameters from left to right). For more information on the pascal calling convention, consult the [Programmers guide](#).

popstack

`Popstack` does the same as `cdecl`, namely it tells the Free Pascal compiler that a function uses the C calling convention. In difference with the `cdecl` modifier, it still mangles the name of the function as it would for a normal pascal function. With `popstack`, functions can be called by their pascal names in a library.

public

The `Public` keyword is used to declare a function globally in a unit. This is useful if the function should not be accessible from the unit file (i.e. another unit/program using the unit doesn't see the function), but must be accessible from the object file. as an example:

```
Unit someunit;
interface
Function First : Real;
Implementation
Function First : Real;
begin
    First := 0;
end;
Function Second : Real; [Public];
begin
    Second := 1;
end;
end.
```

If another program or unit uses this unit, it will not be able to use the function `Second`, since it isn't declared in the interface part. However, it will be possible to access the function `Second` at the assembly-language level, by using its mangled name (see the [Programmers guide](#)).

register

The `register` keyword is used for compatibility with Delphi. In version 1.0.x of the compiler, this directive has no effect on the generated code.

saveregisters

If this modifier is specified after a procedure or function, then the Free Pascal compiler will save all registers on procedure entry, and restore them when the procedure exits (except for registers where return values are stored).

This modifier is not used under normal circumstances, except maybe when calling assembler code.

safecall

This modifier resembles closely the `stdcall` modifier. It sends parameters from right to left on the stack.

More information about this modifier can be found in the [Programmers guide](#), in the section on the calling mechanism and the chapter on linking.

stdcall

This modifier pushes the parameters from right to left on the stack, it also aligns all the parameters to a default alignment.

More information about this modifier can be found in the [Programmers guide](#), in the section on the calling mechanism and the chapter on linking.

9.9 Unsupported Turbo Pascal modifiers

The modifiers that exist in Turbo pascal, but aren't supported by Free Pascal, are listed in table (9.1).

Table 9.1: Unsupported modifiers

Modifier	Why not supported ?
Near	Free Pascal is a 32-bit compiler.
Far	Free Pascal is a 32-bit compiler.

Chapter 10

Operator overloading

10.1 Introduction


Free Pascal supports operator overloading. This means that it is possible to define the action of some operators on self-defined types, and thus allow the use of these types in mathematical expressions.

Defining the action of an operator is much like the definition of a function or procedure, only there are some restrictions on the possible definitions, as will be shown in the subsequent.

Operator overloading is, in essence, a powerful notational tool; but it is also not more than that, since the same results can be obtained with regular function calls. When using operator overloading, It is important to keep in mind that some implicit rules may produce some unexpected results. This will be indicated.

10.2 Operator declarations

To define the action of an operator is much like defining a function:



The parameter list for a comparison operator or an arithmetic operator must always contain 2 parameters. The result type of the comparison operator must be `Boolean`.

Remark: When compiling in `Delphi` mode or `Objfpc` mode, the result identifier may be dropped. The result can then be accessed through the standard `Result` symbol.

If the result identifier is dropped and the compiler is not in one of these modes, a syntax error will occur.

The statement block contains the necessary statements to determine the result of the operation. It can contain arbitrary large pieces of code; it is executed whenever the operation is encountered in some expression. The result of the statement block must always be defined; error conditions are not checked by the compiler, and the code must take care of all possible cases, throwing a run-time error if some error condition is encountered.

In the following, the three types of operator definitions will be examined. As an example, throughout this chapter the following type will be used to define overloaded operators on :

```
type
  complex = record
    re : real;
    im : real;
  end;
```

this type will be used in all examples.

The sources of the Run-Time Library contain a unit `ucomplex`, which contains a complete calculus for complex numbers, based on operator overloading.

10.3 Assignment operators

The assignment operator defines the action of a assignment of one type of variable to another. The result type must match the type of the variable at the left of the assignment statement, the single parameter to the assignment operator must have the same type as the expression at the right of the assignment operator.

This system can be used to declare a new type, and define an assignment for that type. For instance, to be able to assign a newly defined type 'Complex'

```
Var
  C,Z : Complex; // New type complex

begin
  Z:=C; // assignments between complex types.
end;
```

The following assignment operator would have to be defined:

```
Operator := (C : Complex) z : complex;
```

To be able to assign a real type to a complex type as follows:

```
var
  R : real;
  C : complex;
```

```
begin
  C:=R;
end;
```

the following assignment operator must be defined:

```
Operator := (r : real) z : complex;
```

As can be seen from this statement, it defines the action of the operator := with at the right a real expression, and at the left a complex expression.

an example implementation of this could be as follows:

```
operator := (r : real) z : complex;

begin
  z.re:=r;
  z.im:=0.0;
end;
```

As can be seen in the example, the result identifier (z in this case) is used to store the result of the assignment. When compiling in Delphi mode or objfpc mode, the use of the special identifier Result is also allowed, and can be substituted for the z, so the above would be equivalent to

```
operator := (r : real) z : complex;

begin
  Result.re:=r;
  Result.im:=0.0;
end;
```

The assignment operator is also used to convert types from one type to another. The compiler will consider all overloaded assignment operators till it finds one that matches the types of the left hand and right hand expressions. If no such operator is found, a 'type mismatch' error is given.

Remark: The assignment operator is not commutative; the compiler will never reverse the role of the two arguments. in other words, given the above definition of the assignment operator, the following is *not* possible:

```
var
  R : real;
  C : complex;

begin
  R:=C;
end;
```

if the reverse assignment should be possible (this is not so for reals and complex numbers) then the assignment operator must be defined for that as well.

Remark: The assignment operator is also used in implicit type conversions. This can have unwanted effects. Consider the following definitions:

```
operator := (r : real) z : complex;
function exp(c : complex) : complex;
```

then the following assignment will give a type mismatch:

```
Var
  r1,r2 : real;

begin
  r1:=exp(r2);
end;
```

because the compiler will encounter the definition of the `exp` function with the complex argument. It implicitly converts `r2` to a complex, so it can use the above `exp` function. The result of this function is a complex, which cannot be assigned to `r1`, so the compiler will give a 'type mismatch' error. The compiler will not look further for another `exp` which has the correct arguments.

It is possible to avoid this particular problem by specifying

```
  r1:=system.exp(r2);
```

An experimental solution for this problem exists in the compiler, but is not enabled by default. Maybe someday it will be.

10.4 Arithmetic operators

Arithmetic operators define the action of a binary operator. Possible operations are:

multiplication to multiply two types, the `*` multiplication operator must be overloaded.

division to divide two types, the `/` division operator must be overloaded.

addition to add two types, the `+` addition operator must be overloaded.

subtraction to subtract two types, the `-` subtraction operator must be overloaded.

exponentiation to exponentiate two types, the `**` exponentiation operator must be overloaded.

The definition of an arithmetic operator takes two parameters. The first parameter must be of the type that occurs at the left of the operator, the second parameter must be of the type that is at the right of the arithmetic operator. The result type must match the type that results after the arithmetic operation.

To compile an expression as

```
var
  R : real;
  C,Z : complex;

begin
  C:=R*Z;
end;
```

one needs a definition of the multiplication operator as:

```
Operator * (r : real; z1 : complex) z : complex;

begin
  z.re := z1.re * r;
  z.im := z1.im * r;
end;
```

As can be seen, the first operator is a real, and the second is a complex. The result type is complex.

Multiplication and addition of reals and complexes are commutative operations. The compiler, however, has no notion of this fact so even if a multiplication between a real and a complex is defined, the compiler will not use that definition when it encounters a complex and a real (in that order). It is necessary to define both operations.

So, given the above definition of the multiplication, the compiler will not accept the following statement:

```
var
  R : real;
  C,Z : complex;

begin
  C:=Z*R;
end;
```

since the types of Z and R don't match the types in the operator definition.

The reason for this behaviour is that it is possible that a multiplication is not always commutative. e.g. the multiplication of a (n, m) with a (m, n) matrix will result in a (n, n) matrix, while the multiplication of a (m, n) with a (n, m) matrix is a (m, m) matrix, which needn't be the same in all cases.

10.5 Comparision operator

The comparison operator can be overloaded to compare two different types or to compare two equal types that are not basic types. The result type of a comparison operator is always a boolean.

The comparison operators that can be overloaded are:

equal to (=) to determine if two variables are equal.

less than (<) to determine if one variable is less than another.

greater than (>) to determine if one variable is greater than another.

greater than or equal to (>=) to determine if one variable is greater than or equal to another.

less than or equal to (<=) to determine if one variable is greater than or equal to another.

There is no separate operator for *unequal to* (<>). To evaluate a statement that contains the *unequal to* operator, the compiler uses the *equal to* operator (=), and negates the result.

As an example, the following operator allows to compare two complex numbers:

```
operator = (z1, z2 : complex) b : boolean;
```

the above definition allows comparisons of the following form:

```
Var
  C1,C2 : Complex;

begin
  If C1=C2 then
    Writeln('C1 and C2 are equal');
end;
```

The comparison operator definition needs 2 parameters, with the types that the operator is meant to compare. Here also, the compiler doesn't apply commutativity; if the two types are different, then it is necessary to define 2 comparison operators.

In the case of complex numbers, it is, for instance necessary to define 2 comparisons: one with the complex type first, and one with the real type first.

Given the definitions

```
operator = (z1 : complex; r : real) b : boolean;  
operator = (r : real; z1 : complex) b : boolean;
```

the following two comparisons are possible:

```
Var  
  R,S : Real;  
  C : Complex;  
  
begin  
  If (C=R) or (S=C) then  
    Writeln ('Ok');  
end;
```

Note that the order of the real and complex type in the two comparisons is reversed.


Chapter 11

Programs, units, blocks

A Pascal program consists of modules called `units`. A unit can be used to group pieces of code together, or to give someone code without giving the sources. Both programs and units consist of code blocks, which are mixtures of statements, procedures, and variable or type declarations.

11.1 Programs

A pascal program consists of the program header, followed possibly by a 'uses' clause, and a block.



The program header is provided for backwards compatibility, and is ignored by the compiler. The uses clause serves to identify all units that are needed by the program. The system unit doesn't have to be in this list, since it is always loaded by the compiler. The order in which the units appear is significant, it determines in which order they are initialized. Units are initialized in the same order as they appear in the uses clause. Identifiers are searched in the opposite order, i.e. when the compiler searches for an identifier, then it looks first in the last unit in the uses clause, then the last but one, and so on. This is important in case two units declare different types with the same identifier. When the compiler looks for unit files, it adds the extension `.ppu` (`.ppw` for Win32 platforms) to the name of the unit. On LINUX and in operating systems where filenames are case sensitive, when looking for a unit, the unit name is first looked for in the original case, and when not found, converted to all lowercase and searched for.

If a unit name is longer than 8 characters, the compiler will first look for a unit name with this length, and then it will truncate the name to 8 characters and look for it again. For compatibility reasons, this is also true on platforms that support long file names.


```
Unit UnitA;
interface
Uses UnitB;
implementation
end.
```

```
Unit UnitB
interface
Uses UnitA;
implementation
end.
```


But this is allowed :


```
Unit UnitA;
interface
Uses UnitB;
implementation
end.
Unit UnitB
implementation
Uses UnitA;
end.
```

Because **UnitB** uses **UnitA** only in it's implentation section. In general, it is a bad idea to have circular unit dependencies, even if it is only in implementation sections.

11.3 Blocks

Units and programs are made of blocks. A block is made of declarations of labels, constants, types variables and functions or procedures. Blocks can be nested in certain ways, i.e., a procedure or function declaration can have blocks in themselves. A block looks like the following:





Labels that can be used to identify statements in a block are declared in the label declaration part of that block. Each label can only identify one statement. Constants that are to be used only in one block should be declared in that block's constant declaration part. Variables that are to be used only in one block should be declared in that block's constant declaration part. Types that are to be used only in one block should be declared in that block's constant declaration part. Lastly, functions and procedures that will be used in that block can be declared in the procedure/function declaration part. After the different declaration parts comes the statement part. This contains any actions that the block should execute. All identifiers declared before the statement part can be used in that statement part.

11.4 Scope

Identifiers are valid from the point of their declaration until the end of the block in which the declaration occurred. The range where the identifier is known is the *scope* of the identifier. The exact scope of an identifier depends on the way it was defined.

Block scope

The *scope* of a variable declared in the declaration part of a block, is valid from the point of declaration until the end of the block. If a block contains a second block, in which the identifier is redeclared, then inside this block, the second declaration will be valid. Upon leaving the inner block, the first declaration is valid again. Consider the following example:

```

Program Demo;
Var X : Real;
{ X is real variable }
Procedure NewDeclaration
Var X : Integer; { Redeclare X as integer}
begin
  // X := 1.234; {would give an error when trying to compile}
  X := 10; { Correct assignment}
end;
{ From here on, X is Real again}
begin

```

```
X := 2.468;  
end.
```

In this example, inside the procedure, X denotes an integer variable. It has its own storage space, independent of the variable X outside the procedure.

Record scope

The field identifiers inside a record definition are valid in the following places:

1. to the end of the record definition.
2. field designators of a variable of the given record type.
3. identifiers inside a `With` statement that operates on a variable of the given record type.

Class scope

A component identifier is valid in the following places:

1. From the point of declaration to the end of the class definition.
2. In all descendent types of this class, unless it is in the private part of the class declaration.
3. In all method declaration blocks of this class and descendent classes.
4. In a `with` statement that operates on a variable of the given class's definition.

Note that method designators are also considered identifiers.

Unit scope

All identifiers in the interface part of a unit are valid from the point of declaration, until the end of the unit. Furthermore, the identifiers are known in programs or units that have the unit in their `uses` clause. Identifiers from indirectly dependent units are *not* available. Identifiers declared in the implementation part of a unit are valid from the point of declaration to the end of the unit. The system unit is automatically used in all units and programs. Its identifiers are therefore always known, in each pascal program, library or unit. The rules of unit scope imply that an identifier of a unit can be redefined. To have access to an identifier of another unit that was redeclared in the current unit, precede it with that other unit's name, as in the following example:


```
unit unitA;  
interface  
Type  
  MyType = Real;  
implementation  
end.  
Program prog;  
Uses UnitA;  
  
{ Redeclaration of MyType}  
Type MyType = Integer;  
Var A : MyType;      { Will be Integer }  
    B : UnitA.MyType { Will be real }  
begin  
end.
```

This is especially useful when redeclaring the system unit's identifiers.

11.5 Libraries


Free Pascal supports making of dynamic libraries (DLLs under Win32 and OS/2) through the use of the `Library` keyword.

A Library is just like a unit or a program:



By default, functions and procedures that are declared and implemented in library are not available to a programmer that wishes to use this library.

In order to make functions or procedures available from the library, they must be exported in an export clause:



Under Win32, an index clause can be added to an exports entry. an index entry must be a positive number larger or equal than 1.

Optionally, an exports entry can have a name specifier. If present, the name specifier gives the exact name (case sensitive) of the function in the library.

If neither of these constructs is present, the functions or procedures are exported with the exact names as specified in the exports clause.

Chapter 12

Exceptions

Exceptions provide a convenient way to program error and error-recovery mechanisms, and are closely related to classes. Exception support is based on 3 constructs:


Raise statements. To raise an exception. This is usually done to signal an error condition.

Try ... Except blocks. These block serve to catch exceptions raised within the scope of the block, and to provide exception-recovery code.

Try ... Finally blocks. These block serve to force code to be executed irrespective of an exception occurrence or not. They generally serve to clean up memory or close files in case an exception occurs. The compiler generates many implicit `Try ... Finally` blocks around procedure, to force memory consistence.

12.1 The raise statement

The `raise` statement is as follows:



This statement will raise an exception. If it is specified, the exception instance must be an initialized instance of a class, which is the raise type. The address exception is optional. If it is not specified, the compiler will provide the address by itself. If the exception instance is omitted, then the current exception is re-raised. This construct can only be used in an exception handling block (see further).

Remark: Control *never* returns after an exception block. The control is transferred to the first `try...finally` or `try...except` statement that is encountered when unwinding the stack. If no such statement is found, the Free Pascal Run-Time Library will generate a run-time error 217 (see also section 12.5, page 113).

As an example: The following division checks whether the denominator is zero, and if so, raises an exception of type `EDivException`

The class `Exception` is defined in the `Sysutils` unit of the rtl. (section 12.5, page 113)

A try...except exception handling block is of the following form :

If no exception is raised during the execution of the statement `list`, then all statements in the list will be executed sequentially, and the `except` block will be skipped, transferring program flow to the statement after the final `end`.


```

Try
  Z := DoDiv (X,Y);
Except
  On EDivException do Z := 0;
end;

```

If Y happens to be zero, then the DoDiv function code will raise an exception. When this happens, program flow is transferred to the except statement, where the Exception handler will set the value of Z to zero. If no exception is raised, then program flow continues past the last end statement. To allow error recovery, the Try ... Finally block is supported. A Try...Finally block ensures that the statements following the Finally keyword are guaranteed to be executed, even if an exception occurs.

12.3 The try...finally statement

A Try...Finally statement has the following form:

Try...finally statement

→ trystatement – **try** – statement list – **finally** – finally statements – **end** →

→ finally statements – statementlist →

If no exception occurs inside the statement List, then the program runs as if the Try, Finally and End keywords were not present.

If, however, an exception occurs, the program flow is immediately transferred from the point where the exception was raised to the first statement of the Finally statements.

All statements after the finally keyword will be executed, and then the exception will be automatically re-raised. Any statements between the place where the exception was raised and the first statement of the Finally Statements are skipped.

As an example consider the following routine:

```

Procedure Doit (Name : string);
Var F : Text;
begin
  Try
    Assign (F,Name);
    Rewrite (name);
    ... File handling ...
  Finally
    Close(F);
end;

```

If during the execution of the file handling an exception occurs, then program flow will continue at the close(F) statement, skipping any file operations that might follow between the place where the exception was raised, and the Close statement. If no exception occurred, all file operations will be executed, and the file will be closed at the end.

12.4 Exception handling nesting

It is possible to nest `Try...Except` blocks with `Try...Finally` blocks. Program flow will be done according to a `lifo` (last in, first out) principle: The code of the last encountered `Try...Except` or `Try...Finally` block will be executed first. If the exception is not caught, or it was a finally statement, program flow will be transferred to the last-but-one block, *ad infinitum*.

If an exception occurs, and there is no exception handler present, then a `runerror 217` will be generated. When using the `sysutils` unit, a default handler is installed which will show the exception object message, and the address where the exception occurred, after which the program will exit with a `Halt` instruction.

12.5 Exception classes

The `sysutils` unit contains a great deal of exception handling. It defines the following exception types:

```
Exception = class(TObject)
private
    fmessage : string;
    fhelpcontext : longint;
public
    constructor create(const msg : string);
    constructor createres(indent : longint);
    property helpcontext : longint read fhelpcontext write fhelpcontext;
    property message : string read fmessage write fmessage;
end;
ExceptClass = Class of Exception;
{ mathematical exceptions }
EIntError = class(Exception);
EDivByZero = class(EIntError);
ERangeError = class(EIntError);
EIntOverflow = class(EIntError);
EMathError = class(Exception);
```

The `sysutils` unit also installs an exception handler. If an exception is unhandled by any exception handling block, this handler is called by the Run-Time library. Basically, it prints the exception address, and it prints the message of the `Exception` object, and exits with a exit code of 217. If the exception object is not a descendent object of the `Exception` object, then the class name is printed instead of the exception message.

It is recommended to use the `Exception` object or a descendant class for all `raise` statements, since then the message field of the exception object can be used.

Chapter 13

Using assembler

Free Pascal supports the use of assembler in code, but not inline assembler macros. To have more information on the processor specific assembler syntax and its limitations, see the [Programmers guide](#).

13.1 Assembler statements

The following is an example of assembler inclusion in pascal code.

```
...
Statements;
...
Asm
    the asm code here
    ...
end;
...
Statements;
```

The assembler instructions between the `Asm` and `end` keywords will be inserted in the assembler generated by the compiler. Conditionals can be used in assembler, the compiler will recognise it, and treat it as any other conditionals.

13.2 Assembler procedures and functions

Assembler procedures and functions are declared using the `Assembler` directive. This permits the code generator to make a number of code generation optimizations.

The code generator does not generate any stack frame (entry and exit code for the routine) if it contains no local variables and no parameters. In the case of functions, ordinal values must be returned in the accumulator. In the case of floating point values, these depend on the target processor and emulation options.

Part II

Reference : The System unit

Chapter 14

The system unit

The system unit contains the standard supported functions of Free Pascal. It is the same for all platforms. Basically it is the same as the system unit provided with Borland or Turbo Pascal.

Functions are listed in alphabetical order. Arguments of functions or procedures that are optional are put between square brackets.

The pre-defined constants and variables are listed in the first section. The second section contains an overview of all functions, grouped by functionality, and the last section contains the supported functions and procedures.

14.1 Types, Constants and Variables

Types

The following integer types are defined in the System unit:

```
Shortint = -128..127;  
SmallInt = -32768..32767;  
Longint = $80000000..$7fffffff;  
byte = 0..255;  
word = 0..65535;  
dword = longword;  
cardinal = longword;  
Integer = smallint;
```

The following types are used for the functions that need compiler magic such as [Val \(190\)](#) or [Str \(187\)](#):

```
StrLenInt = LongInt;  
ValSInt = Longint;  
ValUInt = Cardinal;  
ValReal = Extended;
```

The Real48 type is defined to emulate the old Turbo Pascal Real type:

```
Real48 = Array[0..5] of byte;
```

The assignment operator has been overloaded so this type can be assigned to the Free Pascal native Double and Extended types. [Real2Double \(175\)](#).

The following character types are defined for Delphi compatibility:

```
TAnsiChar    = Char;  
AnsiChar     = TAnsiChar;
```

And the following pointer types as well:

```
PChar = ^char;  
pPChar = ^PChar;  
PAnsiChar = PChar;  
PQWord = ^QWord;  
PInt64 = ^Int64;  
pshortstring = ^shortstring;  
plongstring = ^longstring;  
pansistring = ^ansistring;  
pwidestring = ^widestring;  
pextended = ^extended;  
ppointer = ^pointer;
```

For the [SetJump \(182\)](#) and [LongJump \(164\)](#) calls, the following jump bufer type is defined (for the I386 processor):

```
jmp_buf = record  
    ebx,esi,edi : Longint;  
    bp,sp,pc : Pointer;  
end;  
PJmp_buf = ^jmp_buf;
```

The following records and pointers can be used to scan the entries in the string message handler tables:

```
tmsgstrtable = record  
    name : pshortstring;  
    method : pointer;  
end;  
pmsgstrtable = ^tmsgstrtable;  
  
tstringmessagetable = record  
    count : dword;  
    msgstrtable : array[0..0] of tmsgstrtable;  
end;  
pstringmessagetable = ^tstringmessagetable;
```

The base class for all classes is defined as:

```
Type  
TObject = Class  
Public  
    constructor create;  
    destructor destroy;virtual;  
    class function newinstance : tobject;virtual;  
    procedure freeinstance;virtual;  
    function safecallexception(exceptobject : tobject;  
        exceptaddr : pointer) : longint;virtual;
```

```

procedure defaulthandler(var message);virtual;
procedure free;
class function initinstance(instance : pointer) : tobject;
procedure cleanupinstance;
function classtype : tclass;
class function classinfo : pointer;
class function classname : shortstring;
class function classnameis(const name : string) : boolean;
class function classparent : tclass;
class function instancesize : longint;
class function inheritsfrom(aclass : tclass) : boolean;
class function inheritsfrom(aclass : tclass) : boolean;
class function stringmessagetable : pstringmessagetable;
procedure dispatch(var message);
procedure dispatchstr(var message);
class function methodaddress(const name : shortstring) : pointer;
class function methodname(address : pointer) : shortstring;
function fieldaddress(const name : shortstring) : pointer;
procedure AfterConstruction;virtual;
procedure BeforeDestruction;virtual;
procedure DefaultHandlerStr(var message);virtual;
end;
TClass = Class Of TObject;
PClass = ^TClass;

```

Unhandled exceptions can be treated using a constant of the TExceptProc type:

```
TExceptProc = Procedure (Obj : TObject; Addr,Frame: Pointer);
```

Obj is the exception object that was used to raise the exception, Addr and Frame contain the exact address and stack frame where the exception was raised.

The TVarRec type is used to access the elements passed in a Array of Const argument to a function or procedure:

Type

```

PVarRec = ^TVarRec;
TVarRec = record
  case VType : Longint of
    vtInteger      : (VInteger: Longint);
    vtBoolean      : (VBoolean: Boolean);
    vtChar         : (VChar: Char);
    vtExtended     : (VExtended: PExtended);
    vtString       : (VString: PShortString);
    vtPointer      : (VPointer: Pointer);
    vtPChar        : (VPChar: PChar);
    vtObject       : (VObject: TObject);
    vtClass        : (VClass: TClass);
    vtAnsiString   : (VAnsiString: Pointer);
    vtWideString   : (VWideString: Pointer);
    vtInt64        : (VInt64: PInt64);
  end;
end;

```

The heap manager uses the TMemoryManager type:

```
PMemoryManager = ^TMemoryManager;
TMemoryManager = record
    Getmem      : Function(Size:Longint):Pointer;
    Freemem     : Function(var p:pointer):Longint;
    FreememSize : Function(var p:pointer;Size:Longint):Longint;
    AllocMem    : Function(Size:longint):Pointer;
    ReAllocMem  : Function(var p:pointer;Size:longint):Pointer;
    MemSize     : function(p:pointer):Longint;
    MemAvail    : Function:Longint;
    MaxAvail    : Function:Longint;
    HeapSize    : Function:Longint;
end;
```

More information on using this record can be found in [Programmers guide](#).

Constants

The following constants define the maximum values that can be used with various types:

```
MaxSIntValue = High(ValSInt);
MaxUIntValue = High(ValUInt);
maxint      = maxsmallint;
maxLongint  = $7fffffff;
maxSmallint = 32767;
```

The following constants for file-handling are defined in the system unit:

```
Const
    fmclosed = $D7B0;
    fminput  = $D7B1;
    fmoutput = $D7B2;
    fminout  = $D7B3;
    fmappend = $D7B4;
    filemode : byte = 2;
```

The `filemode` variable is used when a non-text file is opened using `Reset`. It indicates how the file will be opened. `filemode` can have one of the following values:

- 0** The file is opened for reading.
- 1** The file is opened for writing.
- 2** The file is opened for reading and writing.

The default value is 2. Other values are possible but are operating system specific.

Further, the following non processor specific general-purpose constants are also defined:

```
const
    erroraddr : pointer = nil;
    errorcode : word = 0;
    { max level in dumping on error }
    max_frame_dump : word = 20;
```


Remark: Processor specific global constants are named Testxxxx where xxxx represents the processor number (such as Test8086, Test68000), and are used to determine on what generation of processor the program is running on.

The following constants are defined to access VMT entries:

```
vmtInstanceSize      = 0;
vmtParent            = 8;
vmtClassName         = 12;
vmtDynamicTable      = 16;
vmtMethodTable       = 20;
vmtFieldTable        = 24;
vmtTypeInfo          = 28;
vmtInitTable         = 32;
vmtAutoTable         = 36;
vmtIntfTable         = 40;
vmtMsgStrPtr         = 44;
vmtMethodStart       = 48;
vmtDestroy           = vmtMethodStart;
vmtNewInstance       = vmtMethodStart+4;
vmtFreeInstance      = vmtMethodStart+8;
vmtSafeCallException = vmtMethodStart+12;
vmtDefaultHandler    = vmtMethodStart+16;
vmtAfterConstruction = vmtMethodStart+20;
vmtBeforeDestruction = vmtMethodStart+24;
vmtDefaultHandlerStr = vmtMethodStart+28;
```

The constant names should be used, and never their values, because the VMT table can change, breaking code that uses direct values.

The following constants will be used for the planned variant support:

```
varEmpty    = $0000;
varNull     = $0001;
varSmallint = $0002;
varInteger  = $0003;
varSingle   = $0004;
varDouble   = $0005;
varCurrency = $0006;
varDate     = $0007;
varOleStr   = $0008;
varDispatch = $0009;
varError    = $000A;
varBoolean  = $000B;
varVariant  = $000C;
varUnknown  = $000D;
varByte     = $0011;
varString   = $0100;
varAny      = $0101;
varTypeMask = $0FFF;
varArray    = $2000;
varByRef    = $4000;
```

The following constants are used in the TVarRec record:

```
vtInteger    = 0;
```

```
vtBoolean      = 1;
vtChar         = 2;
vtExtended     = 3;
vtString       = 4;
vtPointer      = 5;
vtPChar        = 6;
vtObject       = 7;
vtClass        = 8;
vtWideChar     = 9;
vtPWideChar    = 10;
vtAnsiString   = 11;
vtCurrency     = 12;
vtVariant      = 13;
vtInterface    = 14;
vtWideString   = 15;
vtInt64        = 16;
vtQWord        = 17;
```

The `ExceptProc` is called when an unhandled exception occurs:

```
Const
    ExceptProc : TExceptProc = Nil;
```

It is set in the `objpas` unit, but it can be set by the programmer to change the default exception handling.

The following constants are defined to describe the operating system's file system:

```
LineEnding = #10;
LFNSupport = true;
DirectorySeparator = '//';
DriveSeparator = ':';
PathSeparator = ':';
FileNameCaseSensitive : Boolean = True;
```

The shown values are for UNIX platforms, but will be different on other platforms. The meaning of the constants is the following:

LineEnding End of line marker. This constant is used when writing end of lines to text files.

LFNSupport This is `True` if the system supports long file names, i.e. filenames that are not restricted to 8.3 characters.

DirectorySeparator The character that is used as a directory separator, i.e. it appears between various parts of a path to a file.

DriveSeparator On systems that support drive letters, this character separates the drive indication from the rest of a filename.

PathSeparator This character can be found between elements in a series of paths (such as the contents of the `PATH` environment variable).

FileNameCaseSensitive Indicates whether filenames are case sensitive.

When programming cross-platform, use these constants instead of hard-coded characters. This will enhance portability of an application.

Variables

The following variables are defined and initialized in the system unit:

```
var
  output,input,stderr : text;
  exitproc : pointer;
  exitcode : word;
  stackbottom : Cardinal;
```

The variables `ExitProc`, `exitcode` are used in the Free Pascal exit scheme. It works similarly to the one in Turbo Pascal:

When a program halts (be it through the call of the `Halt` function or `Exit` or through a run-time error), the exit mechanism checks the value of `ExitProc`. If this one is non-`Nil`, it is set to `Nil`, and the procedure is called. If the exit procedure exits, the value of `ExitProc` is checked again. If it is non-`Nil` then the above steps are repeated. So when an exit procedure must be installed, the old value of `ExitProc` should be saved (it may be non-`Nil`, since other units could have set it). In the exit procedure the value of `ExitProc` should be restored to the previous value, such that if it was non-`Nil` the exit-procedure can be called.

Listing: `refex/ex98.pp`

Program `Example98`;

```
{ Program to demonstrate the exitproc function. }
```

Var

```
  OldExitProc : Pointer;
```

Procedure `MyExit`;

begin

```
  Writeln('My Exitproc was called. Exitcode = ',ExitCode);
```

```
  { restore old exit procedure }
```

```
  ExitProc:=OldExitProc;
```

end;

begin

```
  OldExitProc:=ExitProc;
```

```
  ExitProc:=@MyExit;
```

```
  If ParamCount>0 Then
```

```
    Halt(66);
```

end.

The `ErrorAddr` and `ExitCode` can be used to check for error-conditions. If `ErrorAddr` is non-`Nil`, a run-time error has occurred. If so, `ExitCode` contains the error code. If `ErrorAddr` is `Nil`, then `ExitCode` contains the argument to `Halt` or 0 if the program terminated normally.

`ExitCode` is always passed to the operating system as the exit-code of the current process.

Remark: The maximum error code under LINUX and UNIX like operating systems is 127.

Under GO32, the following constants are also defined :

```
const
  seg0040 = $0040;
  segA000 = $A000;
  segB000 = $B000;
  segB800 = $B800;
```

These constants allow easy access to the bios/screen segment via mem/absolute.

The randomize function uses a seed stored in the RandSeed variable:

```
RandSeed      : Cardinal;
```

This variable is initialized in the initialization code of the system unit.

Other variables indicate the state of the application.

```
IsLibrary      : boolean;  
IsMultiThread  : boolean;
```

The IsLibrary variable is set to true if this module is a shared library instead of an application. The IsMultiThread variable is set to True if the application has spawned other threads, otherwise, and by default, it is set to False.

14.2 Function list by category

What follows is a listing of the available functions, grouped by category. For each function there is a reference to the page where the function can be found:

File handling

Functions concerning input and output from and to file.

Name	Description	Page
Append	Open a file in append mode	128
Assign	Assign a name to a file	129
Blockread	Read data from a file into memory	131
Blockwrite	Write data from memory to a file	131
Close	Close a file	133
Eof	Check for end of file	143
Eoln	Check for end of line	144
Erase	Delete file from disk	144
Filepos	Position in file	148
Filesize	Size of file	148
Flush	Write file buffers to disk	151
IOresult	Return result of last file IO operation	161
Read	Read from file into variable	174
Readln	Read from file into variable and goto next line	175
Rename	Rename file on disk	176
Reset	Open file for reading	177
Rewrite	Open file for writing	177
Seek	Set file position	179
SeekEof	Set file position to end of file	180

SeekEoln	Set file position to end of line	181
SetTextBuf	Set size of file buffer	183
Truncate	Truncate the file at position	189
Write	Write variable to file	191
WriteLn	Write variable to file and append newline	191

Memory management

Functions concerning memory issues.

Name	Description	Page
Addr	Return address of variable	127
Assigned	Check if a pointer is valid	130
CompareByte	Compare 2 memory buffers byte per byte	134
CompareChar	Compare 2 memory buffers byte per byte	135
CompareDWord	Compare 2 memory buffers byte per byte	136
CompareWord	Compare 2 memory buffers byte per byte	137
CSeg	Return code segment	140
Dispose	Free dynamically allocated memory	142
DSeg	Return data segment	143
FillByte	Fill memory region with 8-bit pattern	149
Fillchar	Fill memory region with certain character	150
FillDWord	Fill memory region with 32-bit pattern	150
Fillword	Fill memory region with 16-bit pattern	151
Freemem	Release allocated memory	152
Getmem	Allocate new memory	153
GetMemoryManager	Return current memory manager	154
High	Return highest index of open array or enumerated	155
IsMemoryManagerSet	Is the memory manager set	161
Low	Return lowest index of open array or enumerated	164
Mark	Mark current memory position	165
Maxavail	Return size of largest free memory block	166
Memavail	Return total available memory	166
Move	Move data from one location in memory to another	167
MoveChar0	Move data till first zero character	168
New	Dynamically allocate memory for variable	168
Ofs	Return offset of variable	169
Ptr	Combine segment and offset to pointer	173
ReAllocMem	Resize a memory block on the heap	198
Release	Release memory above mark point	176
Seg	Return segment	181

SetMemoryManager	Set a memory manager	182
Sptr	Return current stack pointer	185
SSeg	Return stack segment register value	186

Mathematical routines

Functions connected to calculating and converting numbers.

Name	Description	Page
Abs	Calculate absolute value	127
Arctan	Calculate inverse tangent	128
Cos	Calculate cosine of angle	140
Dec	Decrease value of variable	141
Exp	Exponentiate	147
Frac	Return fractional part of floating point value	152
Hi	Return high byte/word of value	155
Inc	Increase value of variable	156
Int	Calculate integer part of floating point value	161
Ln	Calculate logarithm	163
Lo	Return low byte/word of value	164
Odd	Is a value odd or even ?	168
Pi	Return the value of pi	171
Power	Raise float to integer power	172
Random	Generate random number	173
Randomize	Initialize random number generator	174
Round	Round floating point value to nearest integer number	179
Sin	Calculate sine of angle	184
Sqr	Calculate the square of a value	185
Sqrt	Calculate the square root of a value	186
Swap	Swap high and low bytes/words of a variable	188
Trunc	Truncate a floating point value	188

String handling

All things connected to string handling.

Name	Description	Page
BinStr	Construct binary representation of integer	130
Chr	Convert ASCII code to character	133
Concat	Concatenate two strings	138
Copy	Copy part of a string	139
Delete	Delete part of a string	141

HexStr	Construct hexadecimal representation of integer	154
Insert	Insert one string in another	160
Length	Return length of string	163
Lowercase	Convert string to all-lowercase	165
OctStr	Construct octal representation of integer	169
Pos	Calculate position of one string in another	172
SetLength	Set length of a string	182
SetString	Set contents and length of a string	183
Str	Convert number to string representation	187
StringOfChar	Create string consisting of a number of characters	187
Ucase	Convert string to all-uppercase	189
Val	Convert string to number	190

Operating System functions

Functions that are connected to the operating system.

Name	Description	Page
Chdir	Change working directory	132
Getdir	Return current working directory	153
Halt	Halt program execution	154
Paramcount	Number of parameters with which program was called	170
Paramstr	Retrieve parameters with which program was called	171
Mkdir	Make a directory	167
Rmdir	Remove a directory	178
Runerror	Abort program execution with error condition	179

Miscellaneous functions

Functions that do not belong in one of the other categories.

Name	Description	Page
Assert	Conditionally abort program with error	129
Break	Abort current loop	132
Continue	Next cycle in current loop	139
Exclude	Exclude an element from a set	145
Exit	Exit current function or procedure	146
Include	Include an element into a set	157
LongJump	Jump to execution point	164
Ord	Return ordinal value of enumerated type	170
Pred	Return previous value of ordinal type	172
SetJump	Mark execution point for jump	182

SizeOf	Return size of variable or type	185
Succ	Return next value of ordinal type	188

14.3 Functions and Procedures

Abs

Declaration: `Function Abs (X : Every numerical type) : Every numerical type;`

Description: Abs returns the absolute value of a variable. The result of the function has the same type as its argument, which can be any numerical type.

Errors: None.

See also: Round ([179](#))

Listing: refex/ex1.pp

```
Program Example1;  
  
  { Program to demonstrate the Abs function. }  
  
Var  
  r : real;  
  i : integer;  
  
begin  
  r:=abs(-1.0);    { r:=1.0 }  
  i:=abs(-21);     { i:=21 }  
end.
```

Addr

Declaration: `Function Addr (X : Any type) : Pointer;`

Description: Addr returns a pointer to its argument, which can be any type, or a function or procedure name. The returned pointer isn't typed. The same result can be obtained by the @ operator, which can return a typed pointer ([Programmers guide](#)).

Errors: None

See also: SizeOf ([185](#))

Listing: refex/ex2.pp

```
Program Example2;  
  
  { Program to demonstrate the Addr function. }  
  
Const Zero : integer = 0;  
  
Var p : pointer;  
    i : Integer;  
  
begin  
  p:=Addr(p);    { P points to itself }
```



```
p:=Addr(I);      { P points to I }  
p:=Addr(Zero);   { P points to 'Zero' }  
end.
```

Append

Declaration: `Procedure Append (Var F : Text);`

Description: `Append` opens an existing file in append mode. Any data written to `F` will be appended to the file. Only text files can be opened in append mode. After a call to `Append`, the file `F` becomes write-only. File sharing is not taken into account when calling `Append`.

Errors: If the file doesn't exist when appending, a run-time error will be generated. This behaviour has changed on Windows and Linux platforms, where in versions prior to 1.0.6, the file would be created in append mode.

See also: `Rewrite` ([177](#)), `Close` ([133](#)), `Reset` ([177](#))

Listing: `refex/ex3.pp`

Program `Example3;`

```
{ Program to demonstrate the Append function. }
```

```
Var f : text;
```

```
begin
```

```
  Assign (f, 'test.txt');
```

```
  Rewrite (f);           { file is opened for write , and emptied }
```

```
  Writeln (F, 'This is the first line of text.txt');
```

```
  close (f);
```

```
  Append(f);             { file is opened for write , but NOT emptied.  
                          any text written to it is appended. }
```

```
  Writeln (f, 'This is the second line of text.txt');
```

```
  close (f);
```

```
end.
```

Arctan

Declaration: `Function Arctan (X : Real) : Real;`

Description: `Arctan` returns the Arctangent of `X`, which can be any `Real` type. The resulting angle is in radial units.

Errors: None

See also: `Sin` ([184](#)), `Cos` ([140](#))

Listing: `refex/ex4.pp`

Program `Example4;`

```
{ Program to demonstrate the ArcTan function. }
```

```
Var R : Real;
```

```
begin
  R:=ArcTan(0);      { R:=0 }
  R:=ArcTan(1)/pi;   { R:=0.25 }
end.
```

Assert

Declaration: `Procedure Assert(expr : Boolean [; const msg: string]);`

Description: With assertions on, `Assert` tests if `expr` is false, and if so, aborts the application with a Runtime error 227 and an optional error message in `msg`. If `expr` is true, program execution continues normally.

If assertions are not enabled at compile time, this routine does nothing, and no code is generated for the `Assert` call.

Enabling and disabling assertions at compile time is done via the `$C` or `$ASSERTIONS` compiler switches. These are global switches.

The default behavior of the assert call can be changed by setting a new handler in the `AssertErrorProc` variable. `Sysutils` overrides the default handler to raise a `EAssertionFailed` exception.

Errors: None.

See also: [Halt \(154\)](#), [Runerror \(179\)](#)

Assign

Declaration: `Procedure Assign (Var F; Name : String);`

Description: `Assign` assigns a name to `F`, which can be any file type. This call doesn't open the file, it just assigns a name to a file variable, and marks the file as closed.

Errors: None.

See also: [Reset \(177\)](#), [Rewrite \(177\)](#), [Append \(128\)](#)

Listing: `refex/ex5.pp`

Program `Example5;`

{ Program to demonstrate the Assign function. }

Var `F : text;`

begin

`Assign (F, '');`

Rewrite `(f);`

*{ The following can be put in any file by redirecting it
from the command line. }*

Writeln `(f, 'This goes to standard output !');`

`Close (f);`

`Assign (F, 'Test.txt');`

rewrite `(f);`

writeln `(f, 'This doesn't go to standard output !');`

`close (f);`

end.

Assigned

Declaration: `Function Assigned (P : Pointer) : Boolean;`

Description: `Assigned` returns `True` if `P` is non-nil and returns `False` if `P` is nil. The main use of `Assigned` is that Procedural variables, method variables and class-type variables also can be passed to `Assigned`.

Errors: None

See also: `New` ([168](#))

Listing: `refex/ex96.pp`

Program `Example96;`

{ Program to demonstrate the Assigned function. }

Var `P : Pointer;`

begin

If Not Assigned(`P`) **then**

WriteLn ('Pointer is initially NIL');

`P:=@P;`

If Not Assigned(`P`) **then**

WriteLn('Internal inconsistency')

else

WriteLn('All is well in FPC')

end.

BinStr

Declaration: `Function BinStr (Value : longint; cnt : byte) : String;`

Description: `BinStr` returns a string with the binary representation of `Value`. The string has at most `cnt` characters. (i.e. only the `cnt` rightmost bits are taken into account) To have a complete representation of any longint-type value, 32 bits are needed, i.e. `cnt=32`

Errors: None.

See also: `Str` ([187](#)), `Val` ([190](#)), `HexStr` ([154](#)), `OctStr` ([169](#))

Listing: `refex/ex82.pp`

Program `example82;`

{ Program to demonstrate the BinStr function }

Const `Value = 45678;`

Var `I : longint;`

begin

For `I:=8 to 20 do`

WriteLn (`BinStr`(`Value`, `I`):20);

end.

Blockread

Declaration: `Procedure Blockread (Var F : File; Var Buffer; Var Count : Longint
[; var Result : Longint]);`

Description: `Blockread` reads `count` or less records from file `F`. A record is a block of bytes with size specified by the `Rewrite` (177) or `Reset` (177) statement.

The result is placed in `Buffer`, which must contain enough room for `Count` records. The function cannot read partial records. If `Result` is specified, it contains the number of records actually read. If `Result` isn't specified, and less than `Count` records were read, a run-time error is generated. This behavior can be controlled by the `{ $i }` switch.

Errors: Depending on the state of the `{ $I }` switch, a runtime error can be generated if there is an error. In the `{ $I- }` state, use `IOResult` to check for errors.

See also: `Blockwrite` (131), `Close` (133), `Reset` (177), `Assign` (129)

Listing: `refex/ex6.pp`

Program `Example6`;

{ Program to demonstrate the BlockRead and BlockWrite functions. }

```
Var Fin , fout : File;  
    NumRead, NumWritten : Word;  
    Buf : Array[1..2048] of byte;  
    Total : Longint;  
  
begin  
    Assign ( Fin , Paramstr(1));  
    Assign ( Fout , Paramstr(2));  
    Reset ( Fin , 1);  
    Rewrite ( Fout , 1);  
    Total:=0;  
    Repeat  
        BlockRead ( Fin , buf , Sizeof( buf ) , NumRead);  
        BlockWrite ( Fout , Buf , NumRead , NumWritten);  
        inc( Total , NumWritten);  
    Until ( NumRead=0) or ( NumWritten<>NumRead);  
    Write ( 'Copied ' , Total , ' bytes from file ' , paramstr(1));  
    Writeln ( ' to file ' , paramstr(2));  
    close( fin );  
    close( fout );  
end.
```

Blockwrite

Declaration: `Procedure Blockwrite (Var F : File; Var Buffer; Var Count : Longint);`

Description: `BlockWrite` writes `count` records from buffer to the file `F`. A record is a block of bytes with size specified by the `Rewrite` (177) or `Reset` (177) statement.

If the records couldn't be written to disk, a run-time error is generated. This behavior can be controlled by the `{ $i }` switch.

Errors: Depending on the state of the `{ $I }` switch, a runtime error can be generated if there is an error. In the `{ $I- }` state, use `IOResult` to check for errors.

See also: Blockread ([131](#)), Close ([133](#)), Rewrite ([177](#)), Assign ([129](#))

For the example, see Blockread ([131](#)).

Break

Declaration: Procedure Break;

Description: Break jumps to the statement following the end of the current repetitive statement. The code between the Break call and the end of the repetitive statement is skipped. The condition of the repetitive statement is NOT evaluated.

This can be used with For, varrepeat and While statements.

Note that while this is a procedure, Break is a reserved word and hence cannot be redefined.

Errors: None.

See also: Continue ([139](#)), Exit ([146](#))

Listing: refex/ex87.pp

Program Example87;

{ Program to demonstrate the Break function. }

Var I : longint;

begin

 I:=0;

While I<10 **Do**

begin

 Inc(I);

If I>5 **Then**

 Break;

 WriteLn (i);

end;

 I:=0;

Repeat

 Inc(I);

If I>5 **Then**

 Break;

 WriteLn (i);

Until I>=10;

For I:=1 to 10 **do**

begin

If I>5 **Then**

 Break;

 WriteLn (i);

end;

end.

Chdir

Declaration: Procedure Chdir (const S : string);

Description: Chdir changes the working directory of the process to S.

Errors:

Errors: Depending on the state of the `{SI}` switch, a runtime error can be generated if there is an error. In the `{SI-}` state, use `IOResult` to check for errors.

See also: `Mkdir` ([167](#)), `Rmdir` ([178](#))

Listing: `refex/ex7.pp`

```
Program Example7;

{ Program to demonstrate the ChDir function. }

begin
  {SI-}
  ChDir (ParamStr(1));
  if IOResult <> 0 then
    WriteLn ('Cannot change to directory : ', paramstr (1));
end.
```

Chr

Declaration: `Function Chr (X : byte) : Char;`

Description: `Chr` returns the character which has ASCII value X.

Errors: None.

See also: `Ord` ([170](#)), `Str` ([187](#))

Listing: `refex/ex8.pp`

```
Program Example8;

{ Program to demonstrate the Chr function. }

begin
  Write (chr(10),chr(13)); { The same effect as WriteLn; }
end.
```

Close

Declaration: `Procedure Close (Var F : Anyfiletype);`

Description: `Close` flushes the buffer of the file F and closes F. After a call to `Close`, data can no longer be read from or written to F. To reopen a file closed with `Close`, it isn't necessary to assign the file again. A call to `Reset` ([177](#)) or `Rewrite` ([177](#)) is sufficient.

Errors: Depending on the state of the `{SI}` switch, a runtime error can be generated if there is an error. In the `{SI-}` state, use `IOResult` to check for errors.

See also: `Assign` ([129](#)), `Reset` ([177](#)), `Rewrite` ([177](#)), `Flush` ([151](#))

Listing: `refex/ex9.pp`

```
Program Example9;

{ Program to demonstrate the Close function. }

Var F : text;

begin
  Assign (f, 'Test.txt');
  ReWrite (F);
  WriteIn (F, 'Some text written to Test.txt');
  close (f); { Flushes contents of buffer to disk,
               closes the file. Omitting this may
               cause data NOT to be written to disk. }
end.
```

CompareByte

Declaration: `function CompareByte(var buf1,buf2:len:longint):longint;`

Description: CompareByte compares two memory regions buf1,buf2 on a byte-per-byte basis for a total of len bytes.

The function returns one of the following values:

less than 0 if buf1 and buf2 contain different bytes in the first len bytes, and the first such byte is smaller in buf1 than the byte at the same position in buf2.

0 if the first len bytes in buf1 and buf2 are equal.

greater than 0 if buf1 and buf2 contain different bytes in the first len bytes, and the first such byte is larger in buf1 than the byte at the same position in buf2.

Errors: None.

See also: CompareChar ([135](#)), CompareWord ([137](#)), CompareDWord ([136](#))

Listing: refex/ex99.pp

```
Program Example99;

{ Program to demonstrate the CompareByte function. }

Const
  ArraySize      = 100;
  HalfArraySize = ArraySize Div 2;

Var
  Buf1,Buf2 : Array[1..ArraySize] of byte;
  I : longint;

Procedure CheckPos(Len : Longint);

Begin
  Write('First ',Len,' positions are ');
  if CompareByte(Buf1,Buf2,Len)<>0 then
    Write('NOT ');
    WriteIn('equal');
end;
```

```
begin
  For I:=1 to ArraySize do
    begin
      Buf1[I]:=I;
      If I<=HalfArraySize Then
        Buf2[I]:=I
      else
        Buf2[I]:=HalfArraySize-I;
      end;
      CheckPos(HalfArraySize div 2);
      CheckPos(HalfArraySize);
      CheckPos(HalfArraySize+1);
      CheckPos(HalfArraySize + HalfArraySize Div 2);
    end.
```

CompareChar

Declaration: `function CompareChar(var buf1,buf2:len:longint):longint; function CompareChar0(var buf1,buf2:len:longint):longint;`

Description: `CompareChar` compares two memory regions `buf1,buf2` on a character-per-character basis for a total of `len` characters.

The `CompareChar0` variant compares `len` bytes, or until a zero character is found.

The function returns one of the following values:

-1 if `buf1` and `buf2` contain different characters in the first `len` positions, and the first such character is smaller in `buf1` than the character at the same position in `buf2`.

0 if the first `len` characters in `buf1` and `buf2` are equal.

1 if `buf1` and `buf2` contain different characters in the first `len` positions, and the first such character is larger in `buf1` than the character at the same position in `buf2`.

Errors: None.

See also: [CompareByte \(134\)](#), [CompareWord \(137\)](#), [CompareDWord \(136\)](#)

Listing: refex/ex100.pp

Program Example100;

{ Program to demonstrate the CompareChar function. }

Const

```
  ArraySize      = 100;
  HalfArraySize = ArraySize Div 2;
```

Var

```
  Buf1,Buf2 : Array[1..ArraySize] of char;
  I : longint;
```

Procedure CheckPos(Len : Longint);

Begin

```
  Write('First ',Len,' characters are ');
```



```
    if CompareChar(Buf1 , Buf2 , Len)<>0 then
        Write( 'NOT ');
        Writeln( 'equal' );
    end;

    Procedure CheckNullPos(Len : Longint);

    Begin
        Write( 'First ',Len,' non-null characters are ');
        if CompareChar0( Buf1 , Buf2 , Len)<>0 then
            Write( 'NOT ');
            Writeln( 'equal' );
        end;

    begin
        For I:=1 to ArraySize do
            begin
                Buf1[ I ]:=chr( I );
                If I<=HalfArraySize Then
                    Buf2[ I ]:=chr( I )
                else
                    Buf2[ I ]:=chr( HalfArraySize-I );
                end;
            CheckPos( HalfArraySize div 2 );
            CheckPos( HalfArraySize );
            CheckPos( HalfArraySize+1 );
            CheckPos( HalfArraySize + HalfArraySize Div 2 );
            For I:=1 to 4 do
                begin
                    buf1[Random( ArraySize)+1]:=Chr(0);
                    buf2[Random( ArraySize)+1]:=Chr(0);
                end;
            Randomize;
            CheckNullPos( HalfArraySize div 2 );
            CheckNullPos( HalfArraySize );
            CheckNullPos( HalfArraySize+1 );
            CheckNullPos( HalfArraySize + HalfArraySize Div 2 );
        end.
```

CompareDWord

Declaration: `function CompareDWord(var buf1,buf2;len:longint):longint;`

Description: `CompareDWord` compares two memory regions `buf1`,`buf2` on a DWord-per-DWord basis for a total of `len` DWords. (A DWord is 4 bytes).

The function returns one of the following values:

-1 if `buf1` and `buf2` contain different DWords in the first `len` DWords, and the first such DWord is smaller in `buf1` than the DWord at the same position in `buf2`.

0 if the first `len` DWords in `buf1` and `buf2` are equal.

1 if `buf1` and `buf2` contain different DWords in the first `len` DWords, and the first such DWord is larger in `buf1` than the DWord at the same position in `buf2`.

Errors: None.

See also: [CompareChar \(135\)](#), [CompareByte \(134\)](#), [CompareWord \(137\)](#),

Listing: refex/ex101.pp**Program** Example101;*{ Program to demonstrate the CompareDWord function. }***Const**

```
ArraySize      = 100;  
HalfArraySize = ArraySize Div 2;
```

Var

```
Buf1, Buf2 : Array[1..ArraySize] of Dword;  
I : longint;
```

```
Procedure CheckPos(Len : Longint);
```

Begin

```
  Write('First ', Len, ' DWords are ');  
  if CompareDWord(Buf1, Buf2, Len) < > 0 then  
    Write('NOT ');  
    Writeln('equal');  
end;
```

begin

```
  For I:=1 to ArraySize do  
    begin  
      Buf1[I]:=I;  
      If I<=HalfArraySize Then  
        Buf2[I]:=I  
      else  
        Buf2[I]:=HalfArraySize-I;  
      end;  
    CheckPos(HalfArraySize Div 2);  
    CheckPos(HalfArraySize);  
    CheckPos(HalfArraySize+1);  
    CheckPos(HalfArraySize + HalfArraySize Div 2);  
end.
```

CompareWord

Declaration: function CompareWord(var buf1, buf2; len: longint): longint;

Description: CompareWord compares two memory regions buf1, buf2 on a Word-per-Word basis for a total of len Words. (A Word is 2 bytes).

The function returns one of the following values:

- 1 if buf1 and buf2 contain different Words in the first len Words, and the first such Word is smaller in buf1 than the Word at the same position in buf2.
- 0 if the first len Words in buf1 and buf2 are equal.
- 1 if buf1 and buf2 contain different Words in the first len Words, and the first such Word is larger in buf1 than the Word at the same position in buf2.

Errors: None.

See also: CompareChar ([135](#)), CompareByte ([134](#)), CompareWord ([137](#)),

Listing: refex/ex102.pp

Program Example102;

{ Program to demonstrate the CompareWord function. }

Const

 ArraySize = 100;
 HalfArraySize = ArraySize **Div** 2;

Var

 Buf1, Buf2 : **Array**[1..ArraySize] **of** Word;
 I : longint;

Procedure CheckPos(Len : Longint);

Begin

Write('First ', Len, ' words are ');
 if CompareWord(Buf1, Buf2, Len) <> 0 **then**
 Write('NOT ');
 Writeln('equal');
 end;

begin

For I:=1 **to** ArraySize **do**
 begin
 Buf1[I]:=I;
 If I<=HalfArraySize **Then**
 Buf2[I]:=I
 else
 Buf2[I]:=HalfArraySize-I;
 end;
 CheckPos(HalfArraySize **div** 2);
 CheckPos(HalfArraySize);
 CheckPos(HalfArraySize+1);
 CheckPos(HalfArraySize + HalfArraySize **Div** 2);
end.

Concat

Declaration: Function Concat (S1,S2 [,S3, ... ,Sn]) : String;

Description: Concat concatenates the strings S1,S2 etc. to one long string. The resulting string is truncated at a length of 255 bytes. The same operation can be performed with the + operation.

Errors: None.

See also: Copy ([139](#)), Delete ([141](#)), Insert ([160](#)), Pos ([172](#)), Length ([163](#))

Listing: refex/ex10.pp

Program Example10;

{ Program to demonstrate the Concat function. }

Var

 S : **String**;

```
begin
  S:=Concat('This can be done',' Easier ','with the + operator !');
end.
```

Continue

Declaration: Procedure Continue;

Description: Continue jumps to the end of the current repetitive statement. The code between the Continue call and the end of the repetitive statement is skipped. The condition of the repetitive statement is then checked again.

This can be used with For, varrepeat and While statements.

Note that while this is a procedure, Continue is a reserved word and hence cannot be redefined.

Errors: None.

See also: Break ([132](#)), Exit ([146](#))

Listing: refex/ex86.pp

Program Example86;

{ Program to demonstrate the Continue function. }

Var I : longint;

```
begin
  I:=0;
  While I<10 Do
    begin
      Inc(I);
      If I<5 Then
        Continue;
      Writeln (i);
    end;
  I:=0;
  Repeat
    Inc(I);
    If I<5 Then
      Continue;
    Writeln (i);
  Until I>=10;
  For I:=1 to 10 do
    begin
      If I<5 Then
        Continue;
      Writeln (i);
    end;
end.
```

Copy

Declaration: Function Copy (Const S : String; Index : Integer; Count : Integer) : String;

Description: Copy returns a string which is a copy of the Count characters in S, starting at position Index. If Count is larger than the length of the string S, the result is truncated. If Index is larger than the length of the string S, then an empty string is returned.

Errors: None.

See also: Delete ([141](#)), Insert ([160](#)), Pos ([172](#))

Listing: refex/ex11.pp

```
Program Example11;

{ Program to demonstrate the Copy function. }

Var S,T : String;

begin
  T:= '1234567';
  S:=Copy (T,1,2);   { S:= '12' }
  S:=Copy (T,4,2);   { S:= '45' }
  S:=Copy (T,4,8);   { S:= '4567' }
end.
```

Cos

Declaration: Function Cos (X : Real) : Real;

Description: Cos returns the cosine of X, where X is an angle, in radians.

If the absolute value of the argument is larger than 2^{63} , then the result is undefined.

Errors: None.

See also: Arctan ([128](#)), Sin ([184](#))

Listing: refex/ex12.pp

```
Program Example12;

{ Program to demonstrate the Cos function. }

Var R : Real;

begin
  R:=Cos(Pi);      { R:= -1 }
  R:=Cos(Pi/2);    { R:= 0 }
  R:=Cos(0);       { R:= 1 }
end.
```

CSeg

Declaration: Function CSeg : Word;

Description: CSeg returns the Code segment register. In Free Pascal, it returns always a zero, since Free Pascal is a 32 bit compiler.

Errors: None.

See also: DSeg ([143](#)), Seg ([181](#)), Ofs ([169](#)), Ptr ([173](#))

Listing: refex/ex13.pp

```
Program Example13;

{ Program to demonstrate the CSeg function. }

var W : word;

begin
  W:=CSeg; {W:=0, provided for compatibility,
            FPC is 32 bit.}
end.
```

Dec

Declaration: Procedure Dec (Var X : Any ordinal type[; Decrement : Any ordinal type]);

Description: Dec decreases the value of X with Decrement. If Decrement isn't specified, then 1 is taken as a default.

Errors: A range check can occur, or an underflow error, if an attempt is made to decrease X below its minimum value.

See also: Inc ([156](#))

Listing: refex/ex14.pp

```
Program Example14;

{ Program to demonstrate the Dec function. }

Var
  I : Integer;
  L : Longint;
  W : Word;
  B : Byte;
  Si : ShortInt;

begin
  I:=1;
  L:=2;
  W:=3;
  B:=4;
  Si:=5;
  Dec (I); { I:=0 }
  Dec (L,2); { L:=0 }
  Dec (W,2); { W:=1 }
  Dec (B,-2); { B:=6 }
  Dec (Si,0); { Si:=5 }
end.
```

Delete

Declaration: Procedure Delete (var S : string; Index : Integer; Count : Integer);

Description: Delete removes Count characters from string S, starting at position Index. All characters after the deleted characters are shifted Count positions to the left, and the length of the string is adjusted.

Errors: None.

See also: Copy ([139](#)), Pos ([172](#)), Insert ([160](#))

Listing: refex/ex15.pp

Program Example15;

{ Program to demonstrate the Delete function. }

Var

S : **String**;

begin

S := 'This is not easy !';

Delete (S,9,4); { S := 'This is easy !' }

end.

Dispose

Declaration: Procedure Dispose (P : pointer);

Procedure Dispose (P : Typed Pointer; Des : Procedure);

Description: The first form Dispose releases the memory allocated with a call to New ([168](#)). The pointer P must be typed. The released memory is returned to the heap.

The second form of Dispose accepts as a first parameter a pointer to an object type, and as a second parameter the name of a destructor of this object. The destructor will be called, and the memory allocated for the object will be freed.

Errors: An runtime error will occur if the pointer doesn't point to a location in the heap.

See also: New ([168](#)), Getmem ([153](#)), Freemem ([152](#))

Listing: refex/ex16.pp

Program Example16;

{ Program to demonstrate the Dispose and New functions. }

Type SS = **String**[20];

AnObj = **Object**

l : integer;

Constructor Init;

Destructor Done;

end;

Var

P : ^SS;

T : ^AnObj;

Constructor Anobj.Init;

begin

```
  Writeln ( 'Initializing an instance of AnObj ! ');
end;

Destructor AnObj.Done;

begin
  Writeln ( 'Destroying an instance of AnObj ! ');
end;

begin
  New (P);
  P^:= 'Hello , World ! ';
  Dispose (P);
  { P is undefined from here on ! }
  New(T, Init);
  T^.i:=0;
  Dispose (T, Done);
end.
```

DSeg

Declaration: `Function DSeg : Word;`

Description: `DSeg` returns the data segment register. In Free Pascal, it returns always a zero, since Free Pascal is a 32 bit compiler.

Errors: None.

See also: `CSeg` ([140](#)), `Seg` ([181](#)), `Ofs` ([169](#)), `Ptr` ([173](#))

Listing: `refex/ex17.pp`

Program `Example17;`

{ Program to demonstrate the DSeg function. }

Var

W : Word;

begin

W:=`DSeg`; *{W:=0, This function is provided for compatibility,
FPC is a 32 bit comiler.}*

end.

Eof

Declaration: `Function Eof [(F : Any file type)] : Boolean;`

Description: `Eof` returns `True` if the file-pointer has reached the end of the file, or if the file is empty. In all other cases `Eof` returns `False`. If no file `F` is specified, standard input is assumed.

Errors: Depending on the state of the `{ $I }` switch, a runtime error can be generated if there is an error. In the `{ $I- }` state, use `IOResult` to check for errors.

See also: `Eoln` ([144](#)), `Assign` ([129](#)), `Reset` ([177](#)), `Rewrite` ([177](#))

Listing: refex/ex18.pp

Program Example18;

```
{ Program to demonstrate the Eof function. }

Var T1,T2 : text;
    C : Char;

begin
  { Set file to read from. Empty means from standard input. }
  assign (t1,paramstr(1));
  reset (t1);
  { Set file to write to. Empty means to standard output. }
  assign (t2,paramstr(2));
  rewrite (t2);
  While not eof(t1) do
    begin
      read (t1,C);
      write (t2,C);
    end;
  Close (t1);
  Close (t2);
end.
```

Eoln

Declaration: Function Eoln [(F : Text)] : Boolean;

Description: Eof returns True if the file pointer has reached the end of a line, which is demarcated by a line-feed character (ASCII value 10), or if the end of the file is reached. In all other cases Eof returns False. If no file F is specified, standard input is assumed. It can only be used on files of type Text.

Errors: None.

See also: Eof ([143](#)), Assign ([129](#)), Reset ([177](#)), Rewrite ([177](#))**Listing:** refex/ex19.pp

Program Example19;

```
{ Program to demonstrate the Eoln function. }

begin
  { This program waits for keyboard input. }
  { It will print True when an empty line is put in,
    and false when you type a non-empty line.
    It will only stop when you press enter. }
  While not Eoln do
    Writeln (eoln);
end.
```

Erase

Declaration: Procedure Erase (Var F : Any file type);

Description: `Erase` removes an unopened file from disk. The file should be assigned with `Assign`, but not opened with `Reset` or `Rewrite`

Errors: Depending on the state of the `{SI}` switch, a runtime error can be generated if there is an error. In the `{SI-}` state, use `IOResult` to check for errors.

See also: `Assign` ([129](#))

Listing: `refex/ex20.pp`

Program `Example20`;

{ Program to demonstrate the Erase function. }

Var `F : Text`;

begin

{ Create a file with a line of text in it }

`Assign (F, 'test.txt');`

`Rewrite (F);`

`Writeln (F, 'Try and find this when I'm finished !');`

`close (f);`

{ Now remove the file }

`Erase (f);`

end.

Exclude

Declaration: `Procedure Exclude (Var S : Any set type; E : Set element);`

Description: `Exclude` removes `E` from the set `S` if it is included in the set. `E` should be of the same type as the base type of the set `S`.

Thus, the two following statements do the same thing:

`S:=S-[E];`

`Exclude(S,E);`

Errors: If the type of the element `E` is not equal to the base type of the set `S`, the compiler will generate an error.

See also: `Include` ([157](#))

Listing: `refex/ex111.pp`

program `Example111`;

{ Program to demonstrate the Include/Exclude functions }

Type

`TEnumA = (aOne,aTwo,aThree);`

`TEnumAs = Set of TEnumA;`

Var

`SA : TEnumAs;`

Procedure `PrintSet(S : TEnumAs);`

```
var
  B : Boolean;

  procedure DoEl(A : TEnumA; Desc : String);

  begin
    If A in S then
      begin
        If B then
          Write( ' , ' );
        B:=True;
        Write(Desc);
      end;
    end;

  begin
    Write( ' [ ' );
    B:=False;
    DoEl(aOne, 'aOne');
    DoEl(aTwo, 'aTwo');
    DoEl(aThree, 'aThree');
    WriteLn( ' ] ' )
  end;

begin
  SA:=[];
  Include(SA,aOne);
  PrintSet(SA);
  Include(SA,aThree);
  PrintSet(SA);
  Exclude(SA,aOne);
  PrintSet(SA);
  Exclude(SA,aTwo);
  PrintSet(SA);
  Exclude(SA,aThree);
  PrintSet(SA);
end.
```

Exit

Declaration: `Procedure Exit ([Var X : return type]);`

Description: `Exit` exits the current subroutine, and returns control to the calling routine. If invoked in the main program routine, exit stops the program. The optional argument `X` allows to specify a return value, in the case `Exit` is invoked in a function. The function result will then be equal to `X`.

Errors: None.

See also: `Halt` ([154](#))

Listing: `refex/ex21.pp`

Program `Example21;`

{ Program to demonstrate the Exit function. }

```
Procedure DoAnExit (Yes : Boolean);  
  
{ This procedure demonstrates the normal Exit }  
  
begin  
  Writeln ( 'Hello from DoAnExit ! ');  
  If Yes then  
    begin  
      Writeln ( 'Bailing out early.' );  
      exit;  
    end;  
  Writeln ( 'Continuing to the end.' );  
end;  
  
Function Positive (Which : Integer) : Boolean;  
  
{ This function demonstrates the extra FPC feature of Exit :  
  You can specify a return value for the function }  
  
begin  
  if Which > 0 then  
    exit (True)  
  else  
    exit (False);  
end;  
  
begin  
  { This call will go to the end }  
  DoAnExit (False);  
  { This call will bail out early }  
  DoAnExit (True);  
  if Positive (-1) then  
    Writeln ( 'The compiler is nuts, -1 is not positive.' )  
  else  
    Writeln ( 'The compiler is not so bad, -1 seems to be negative.' );  
end.
```

Exp

Declaration: `Function Exp (Var X : Real) : Real;`

Description: Exp returns the exponent of X, i.e. the number e to the power X.

Errors: None.

See also: Ln ([163](#)), Power ([172](#))

Listing: `refex/ex22.pp`

```
Program Example22;  
  
{ Program to demonstrate the Exp function. }  
  
begin  
  Writeln (Exp(1):8:2); { Should print 2.72 }  
end.
```

Filepos

Declaration: `Function Filepos (Var F : Any file type) : Longint;`

Description: `Filepos` returns the current record position of the file-pointer in file `F`. It cannot be invoked with a file of type `Text`. A compiler error will be generated if this is attempted.

Errors: Depending on the state of the `{SI}` switch, a runtime error can be generated if there is an error. In the `{SI-}` state, use `IOResult` to check for errors.

See also: `Filesize` ([148](#))

Listing: `refex/ex23.pp`

Program `Example23;`

{ Program to demonstrate the FilePos function. }

Var `F : File of Longint;`
 `L,FP : longint;`

begin

*{ Fill a file with data :
 Each position contains the position ! }*

`Assign (F, 'test.tmp');`

`Rewrite (F);`

For `L:=0 to 100 do`

begin

`FP:=FilePos(F);`

`Write (F,FP);`

end;

`Close (F);`

`Reset (F);`

{ If all goes well, nothing is displayed here. }

While not (Eof(F)) do

begin

`FP:=FilePos (F);`

`Read (F,L);`

if `L<>FP then`

`Writeln ('Something wrong: Got ',L, ' on pos ',FP);`

end;

`Close (F);`

`Erase (f);`

end.

Filesize

Declaration: `Function Filesize (Var F : Any file type) : Longint;`

Description: `Filesize` returns the total number of records in file `F`. It cannot be invoked with a file of type `Text`. (under LINUX and UNIX, this also means that it cannot be invoked on pipes). If `F` is empty, 0 is returned.

Errors: Depending on the state of the `{SI}` switch, a runtime error can be generated if there is an error. In the `{SI-}` state, use `IOResult` to check for errors.

See also: `Filepos` ([148](#))

Listing: refex/ex24.pp

Program Example24;*{ Program to demonstrate the FileSize function. }***Var** F : **File Of** byte;
 L : **File Of** Longint;**begin** Assign (F, **paramstr** (1)); **Reset** (F); **WriteLn** ('File size in bytes : ', **FileSize** (F));

Close (F);

 Assign (L, **paramstr** (1)); **Reset** (L); **WriteLn** ('File size in Longints : ', **FileSize** (L));

Close (f);

end.

FillByteDeclaration: **Procedure** FillByte(var X;Count:longint;Value:byte);

Description: FillByte fills the memory starting at X with Count bytes with value equal to Value.

This is useful for quickly zeroing out a memory location. When the size of the memory location to be filled out is a multiple of 2 bytes, it is better to use **Fillword** ([151](#)), and if it is a multiple of 4 bytes it is better to use **FillDWord** ([150](#)), these routines are optimized for their respective sizes.

Errors: No checking on the size of X is done.

See also: **Fillchar** ([150](#)), **FillDWord** ([150](#)), **Fillword** ([151](#)), **Move** ([167](#))**Listing:** refex/ex102.pp

Program Example102;*{ Program to demonstrate the CompareWord function. }***Const**

ArraySize = 100;

 HalfArraySize = ArraySize **Div** 2;**Var** Buf1, Buf2 : **Array** [1.. ArraySize] **of** Word;

I : longint;

Procedure CheckPos(Len : Longint);**Begin** **Write** ('First ', Len, ' words are '); **if** CompareWord (Buf1, Buf2, Len) <> 0 **then** **Write** ('NOT '); **WriteLn** ('equal ');**end;**

```
begin
  For I:=1 to ArraySize do
    begin
      Buf1[I]:=I;
      If I<=HalfArraySize Then
        Buf2[I]:=I
      else
        Buf2[I]:=HalfArraySize-I;
      end;
      CheckPos(HalfArraySize div 2);
      CheckPos(HalfArraySize);
      CheckPos(HalfArraySize+1);
      CheckPos(HalfArraySize + HalfArraySize Div 2);
    end.
end.
```

Fillchar

Declaration: Procedure Fillchar (Var X;Count : Longint;Value : char or byte);;

Description: Fillchar fills the memory starting at X with Count bytes or characters with value equal to Value.

Errors: No checking on the size of X is done.

See also: Fillword ([151](#)), Move ([167](#)), FillByte ([149](#)), FillDWord ([150](#))

Listing: refex/ex25.pp

Program Example25;

{ Program to demonstrate the FillChar function. }

```
Var S : String[10];
    I : Byte;
begin
  For i:=10 downto 0 do
    begin
      { Fill S with i spaces }
      FillChar (S,SizeOf(S),' ');
      { Set Length }
      SetLength(S,I);
      WriteLn (s,'*');
    end;
  end.
end.
```

FillDWord

Declaration: Procedure FillDWord (Var X;Count : Longint;Value : DWord);;

Description: Fillword fills the memory starting at X with Count DWords with value equal to Value. A DWord is 4 bytes in size.

Errors: No checking on the size of X is done.

See also: FillByte ([149](#)), Fillchar ([150](#)), Fillword ([151](#)), Move ([167](#))

Listing: refex/ex103.pp

Program Example103;

{ Program to demonstrate the FillByte function. }

```
Var S : String[10];
    I : Byte;

begin
  For i:=10 downto 0 do
    begin
      { Fill S with i bytes }
      FillChar (S, SizeOf(S), 32);
      { Set Length }
      SetLength(S, I);
      Writeln (s, '* ');
    end;
  end.
```

Fillword

Declaration: Procedure Fillword (Var X; Count : Longint; Value : Word);;

Description: Fillword fills the memory starting at X with Count words with value equal to Value. A word is 2 bytes in size.

Errors: No checking on the size of X is done.

See also: Fillchar ([150](#)), Move ([167](#))

Listing: refex/ex76.pp

Program Example76;

{ Program to demonstrate the FillWord function. }

```
Var W : Array[1..100] of Word;

begin
  { Quick initialization of array W }
  FillWord(W, 100, 0);
end.
```

Flush

Declaration: Procedure Flush (Var F : Text);

Description: Flush empties the internal buffer of an opened file F and writes the contents to disk. The file is *not* closed as a result of this call.

Errors: Depending on the state of the {SI} switch, a runtime error can be generated if there is an error. In the {SI-} state, use IOResult to check for errors.

See also: Close ([133](#))

Listing: refex/ex26.pp

Program Example26;

```
{ Program to demonstrate the Flush function. }

Var F : Text;

begin
  { Assign F to standard output }
  Assign (F, '');
  Rewrite (F);
  Writeln (F, 'This line is written first , but appears later !');
  { At this point the text is in the internal pascal buffer ,
    and not yet written to standard output }
  Writeln ('This line appears first , but is written later !');
  { A writeln to 'output' always causes a flush – so this text is
    written to screen }
  Flush (f);
  { At this point , the text written to F is written to screen. }
  Write (F, 'Finishing ');
  Close (f); { Closing a file always causes a flush first }
  Writeln ('off. ');
end.
```

Frac

Declaration: Function Frac (X : Real) : Real;

Description: Frac returns the non-integer part of X.

Errors: None.

See also: Round ([179](#)), Int ([161](#))**Listing:** refex/ex27.pp

Program Example27;

```
{ Program to demonstrate the Frac function. }

Var R : Real;

begin
  Writeln (Frac (123.456):0:3); { Prints 0.456 }
  Writeln (Frac (-123.456):0:3); { Prints -0.456 }
end.
```

Freemem

Declaration: Procedure Freemem (Var P : pointer; Count : Longint);

Description: Freemem releases the memory occupied by the pointer P, of size Count (in bytes), and returns it to the heap. P should point to the memory allocated to a dynamic variable.

Errors: An error will occur when P doesn't point to the heap.

See also: [Getmem \(153\)](#), [New \(168\)](#), [Dispose \(142\)](#)

Listing: [refex/ex28.pp](#)

```
Program Example28;

{ Program to demonstrate the FreeMem and GetMem functions. }

Var P : Pointer;
    MM : Longint;

begin
  { Get memory for P }
  MM:=MemAvail;
  Writeln ( 'Memory available before GetMem : ',MemAvail);
  GetMem (P,80);
  MM:=MM-Memavail;
  Write ( 'Memory available after GetMem : ',MemAvail);
  Writeln ( ' or ',MM,' bytes less than before the call. ');
  { fill it with spaces }
  FillChar (P^,80,' ');
  { Free the memory again }
  FreeMem (P,80);
  Writeln ( 'Memory available after FreeMem : ',MemAvail);
end.
```

Getdir

Declaration: `Procedure Getdir (drivenr : byte;var dir : string);`

Description: Getdir returns in dir the current directory on the drive drivenr, where drivenr is 1 for the first floppy drive, 3 for the first hard disk etc. A value of 0 returns the directory on the current disk. On LINUX and UNIX systems, drivenr is ignored, as there is only one directory tree.

Errors: An error is returned under DOS, if the drive requested isn't ready.

See also: [Chdir \(132\)](#)

Listing: [refex/ex29.pp](#)

```
Program Example29;

{ Program to demonstrate the GetDir function. }

Var S : String;

begin
  GetDir (0,S);
  Writeln ( 'Current directory is : ',S);
end.
```

Getmem

Declaration: `Procedure Getmem (var p : pointer;size : Longint);`

Description: `Getmem` reserves `Size` bytes memory on the heap, and returns a pointer to this memory in `p`. If no more memory is available, `nil` is returned.

Errors: None.

See also: `Freemem` ([152](#)), `Dispose` ([142](#)), `New` ([168](#))

For an example, see `Freemem` ([152](#)).

GetMemoryManager

Declaration: `procedure GetMemoryManager (var MemMgr : TMemoryManager);`

Description: `GetMemoryManager` stores the current Memory Manager record in `MemMgr`.

Errors: None.

See also: `SetMemoryManager` ([182](#)), `IsMemoryManagerSet` ([161](#)).

For an example, see [Programmers guide](#).

Halt

Declaration: `Procedure Halt [(Errnum : byte)];`

Description: `Halt` stops program execution and returns control to the calling program. The optional argument `Errnum` specifies an exit value. If omitted, zero is returned.

Errors: None.

See also: `Exit` ([146](#))

Listing: `refex/ex30.pp`

Program `Example30;`

{ Program to demonstrate the Halt function. }

```
begin
  Writeln ('Before Halt. ');
  Halt (1); { Stop with exit code 1 }
  Writeln ('After Halt doesn't get executed. ');
end.
```

HexStr

Declaration: `Function HexStr (Value : longint; cnt : byte) : String; Function HexStr (Value : int64; cnt : byte) : String;`

Description: `HexStr` returns a string with the hexadecimal representation of `Value`. The string has exactly `cnt` characters. (i.e. only the `cnt` rightmost nibbles are taken into account) To have a complete representation of a Longint-type value, 8 nibbles are needed, i.e. `cnt=8`.

Errors: None.

See also: `Str` ([187](#)), `Val` ([190](#)), `BinStr` ([130](#))

Listing: refex/ex81.pp

```
Program example81;  
  
{ Program to demonstrate the HexStr function }  
  
Const Value = 45678;  
  
Var I : longint;  
  
begin  
  For I:=1 to 10 do  
    WriteLn ( HexStr(Value, I));  
end.
```

Hi

Declaration: Function Hi (X : Ordinal type) : Word or byte;

Description: Hi returns the high byte or word from X, depending on the size of X. If the size of X is 4, then the high word is returned. If the size is 2 then the high byte is returned. Hi cannot be invoked on types of size 1, such as byte or char.

Errors: None

See also: Lo ([164](#))

Listing: refex/ex31.pp

```
Program Example31;  
  
{ Program to demonstrate the Hi function. }  
  
var  
  L : Longint;  
  W : Word;  
  
begin  
  L:=1 Shl 16;      { = $10000 }  
  W:=1 Shl 8;       { = $100 }  
  WriteLn ( Hi(L)); { Prints 1 }  
  WriteLn ( Hi(W)); { Prints 1 }  
end.
```

High

Declaration: Function High (Type identifier or variable reference) : Ordinal;

Description: The return value of High depends on it's argument:

- 1.If the argument is an ordinal type, High returns the highest value in the range of the given ordinal type.
- 2.If the argument is an array type or an array type variable then High returns the highest possible value of it's index.
- 3.If the argument is an open array identifier in a function or procedure, then High returns the highest index of the array, as if the array has a zero-based index.

The return type is always the same type as the type of the argument (This can lead to some nasty surprises!).

Errors: None.

See also: [Low \(164\)](#), [Ord \(170\)](#), [Pred \(172\)](#), [Succ \(188\)](#)

Listing: `refex/ex80.pp`

Program `example80;`

{ Example to demonstrate the High and Low functions. }

Type `TEnum = (North , East , South , West);`
 `TRange = 14..55;`
 `TArray = Array [2..10] of Longint;`

Function `Average (Row : Array of Longint) : Real;`

Var `I : longint;`
 `Temp : Real;`

begin

`Temp := Row[0];`
 For `I := 1 to High(Row) do`
 `Temp := Temp + Row[i];`
 `Average := Temp / (High(Row)+1);`

end;

Var `A : TEnum;`
 `B : TRange;`
 `C : TArray;`
 `I : longint;`

begin

`Writeln ('TEnum goes from : ',Ord(Low(TEnum)), ' to ', Ord(high(TEnum)), '. ');`
 `Writeln ('A goes from : ',Ord(Low(A)), ' to ', Ord(high(A)), '. ');`
 `Writeln ('TRange goes from : ',Ord(Low(TRange)), ' to ', Ord(high(TRange)), '. ');`
 `Writeln ('B goes from : ',Ord(Low(B)), ' to ', Ord(high(B)), '. ');`
 `Writeln ('TArray index goes from : ',Ord(Low(TArray)), ' to ', Ord(high(TArray)), '. ');`
 `Writeln ('C index goes from : ',Low(C), ' to ', high(C), '. ');`
 For `I:=Low(C) to High(C) do`
 `C[i]:=I;`
 `Writeln ('Average : ',Average(c));`
 `Write ('Type of return value is always same as type of argument:');`
 `Writeln(high(high(word)));`

end.

Inc

Declaration: `Procedure Inc (Var X : Any ordinal type[; Increment : Any ordinal type]);`

Description: `Inc` increases the value of `X` with `Increment`. If `Increment` isn't specified, then `1` is taken as a default.

Errors: If range checking is on, then a range check can occur, or an overflow error, when an attempt is made to increase `X` over its maximum value.

See also: [Dec \(141\)](#)

Listing: `refex/ex32.pp`

Program `Example32;`

{ Program to demonstrate the Inc function. }

Const

`C : Cardinal = 1;
L : Longint = 1;
I : Integer = 1;
W : Word = 1;
B : Byte = 1;
SI : ShortInt = 1;
CH : Char = 'A';`

begin

`Inc (C); { C:=2 }
Inc (L, 5); { L:=6 }
Inc (I, -3); { I:=-2 }
Inc (W, 3); { W:=4 }
Inc (B, 100); { B:=101 }
Inc (SI, -3); { SI:=-2 }
Inc (CH, 1); { ch:='B' }`

end.

Include

Declaration: `Procedure Include (Var S : Any set type; E : Set element);`

Description: `Include` includes `E` in the set `S` if it is not yet part of the set. `E` should be of the same type as the base type of the set `S`.

Thus, the two following statements do the same thing:

`S:=S+[E];
Include(S,E);`

Errors: If the type of the element `E` is not equal to the base type of the set `S`, the compiler will generate an error.

See also: [Exclude \(145\)](#)

For an example, see [Exclude \(145\)](#)

IndexByte

Declaration: `function IndexByte(var buf;len:longint;b:byte):longint;`

Description: `IndexByte` searches the memory at `buf` for maximally `len` positions for the byte `b` and returns it's position if it found one. If `b` is not found then -1 is returned.

The position is zero-based.

Errors: `Buf` and `Len` are not checked to see if they are valid values.

See also: [IndexChar \(158\)](#), [IndexDWord \(159\)](#), [IndexWord \(160\)](#), [CompareByte \(134\)](#)

Listing: [refex/ex105.pp](#)

Program Example105;

{ Program to demonstrate the IndexByte function. }

Const

 ArraySize = 256;
 MaxValue = 256;

Var

 Buffer : **Array**[1..ArraySize] **of** Byte;
 I,J : longint;
 K : Byte;

begin

Randomize;

For I:=1 **To** ArraySize **do**

 Buffer[I]:=Random(MaxValue);

For I:=1 **to** 10 **do**

begin

 K:=Random(MaxValue);

 J:=IndexByte(Buffer, ArraySize, K);

if J=-1 **then**

 Writeln('Value ',K,' was not found in buffer.')

else

 Writeln('Found ',K,' at position ',J,' in buffer');

end;

end.

IndexChar

Declaration: `function IndexChar(var buf;len:longint;b:char):longint;`

Declaration: `function IndexChar0(var buf;len:longint;b:char):longint;`

Description: IndexChar searches the memory at buf for maximally len positions for the character b and returns it's position if it found one. If b is not found then -1 is returned.

The position is zero-based. The IndexChar0 variant stops looking if a null character is found, and returns -1 in that case.

Errors: Buf and Len are not checked to see if they are valid values.

See also: [IndexByte \(157\)](#), [IndexDWord \(159\)](#), [IndexWord \(160\)](#), [CompareChar \(135\)](#)

Listing: [refex/ex108.pp](#)

Program Example108;

{ Program to demonstrate the IndexChar function. }

Const

 ArraySize = 1000;
 MaxValue = 26;

```
Var
  Buffer : Array[1..ArraySize] of Char;
  I,J : longint;
  K : Char;

begin
  Randomize;
  For I:=1 To ArraySize do
    Buffer[I]:=chr(Ord('A')+Random(MaxValue));
  For I:=1 to 10 do
    begin
      K:=chr(Ord('A')+Random(MaxValue));
      J:=IndexChar(Buffer, ArraySize, K);
      if J=-1 then
        Writeln('Value ',K,' was not found in buffer.')
      else
        Writeln('Found ',K,' at position ',J,' in buffer');
      end;
    end.
end.
```

IndexDWord

Declaration: `function IndexDWord(var buf; len:longint; DW:DWord):longint;`

Description: `IndexChar` searches the memory at `buf` for maximally `len` positions for the `DWord` `DW` and returns it's position if it found one. If `DW` is not found then -1 is returned.

The position is zero-based.

Errors: `Buf` and `Len` are not checked to see if they are valid values.

See also: `IndexByte` ([157](#)), `IndexChar` ([158](#)), `IndexWord` ([160](#)), `CompareDWord` ([136](#))

Listing: `refex/ex106.pp`

Program `Example106;`

{ Program to demonstrate the IndexDWord function. }

Const

```
  ArraySize = 1000;
  MaxValue = 1000;
```

Var

```
  Buffer : Array[1..ArraySize] of DWord;
  I,J : longint;
  K : DWord;
```

begin

```
  Randomize;
  For I:=1 To ArraySize do
    Buffer[I]:=Random(MaxValue);
  For I:=1 to 10 do
    begin
      K:=Random(MaxValue);
      J:=IndexDWord(Buffer, ArraySize, K);
      if J=-1 then
        Writeln('Value ',K,' was not found in buffer.')
```



```
    else
      Writeln('Found ',K,' at position ',J,' in buffer');
    end;
end.
```

IndexWord

Declaration: `function IndexWord(var buf:len:longint;W:word):longint;`

Description: `IndexChar` searches the memory at `buf` for maximally `len` positions for the Word `W` and returns it's position if it found one. If `W` is not found then -1 is returned.

Errors: `Buf` and `Len` are not checked to see if they are valid values.

See also: `IndexByte` ([157](#)), `IndexDWord` ([159](#)), `IndexChar` ([158](#)), `CompareWord` ([137](#))

Listing: `refex/ex107.pp`

Program `Example107;`

```
{ Program to demonstrate the IndexWord function. }
```

Const

```
  ArraySize = 1000;
  MaxValue = 1000;
```

Var

```
  Buffer : Array[1..ArraySize] of Word;
  I,J : longint;
  K : Word;
```

begin

```
  Randomize;
  For I:=1 To ArraySize do
    Buffer[I]:=Random(MaxValue);
  For I:=1 to 10 do
    begin
      K:=Random(MaxValue);
      J:=IndexWord(Buffer,ArraySize,K);
      if J=-1 then
        Writeln('Value ',K,' was not found in buffer.')
      else
        Writeln('Found ',K,' at position ',J,' in buffer');
      end;
    end;
```

```
end.
```

Insert

Declaration: `Procedure Insert (Const Source : String;var S : String;Index : Integer);`

Description: `Insert` inserts string `Source` in string `S`, at position `Index`, shifting all characters after `Index` to the right. The resulting string is truncated at 255 characters, if needed. (i.e. for shortstrings)

Errors: None.

See also: `Delete` ([141](#)), `Copy` ([139](#)), `Pos` ([172](#))

Listing: refex/ex33.pp

```
Program Example33;  
  
  { Program to demonstrate the Insert function. }  
  
Var S : String;  
  
begin  
  S:= 'Free Pascal is difficult to use !';  
  Insert ( 'NOT ',S,pos( 'difficult ',S));  
  writeln (s);  
end.
```

IsMemoryManagerSet

Declaration: function IsMemoryManagerSet: Boolean;

Description: IsMemoryManagerSet will return True if the memory manager has been set to another value than the system heap manager, it will return False otherwise.

Errors: None.

See also: SetMemoryManager ([182](#)), GetMemoryManager ([154](#))

Int

Declaration: Function Int (X : Real) : Real;

Description: Int returns the integer part of any Real X, as a Real.

Errors: None.

See also: Frac ([152](#)), Round ([179](#))

Listing: refex/ex34.pp

```
Program Example34;  
  
  { Program to demonstrate the Int function. }  
  
begin  
  Writeln (Int(123.456):0:1); { Prints 123.0 }  
  Writeln (Int(-123.456):0:1); { Prints -123.0 }  
end.
```

IOresult

Declaration: Function IOresult : Word;

Description: IOresult contains the result of any input/output call, when the {\$i-} compiler directive is active, disabling IO checking. When the flag is read, it is reset to zero. If IOresult is zero, the operation completed successfully. If non-zero, an error occurred. The following errors can occur:

DOS errors :

2 File not found.

- 3 Path not found.
- 4 Too many open files.
- 5 Access denied.
- 6 Invalid file handle.
- 12 Invalid file-access mode.
- 15 Invalid disk number.
- 16 Cannot remove current directory.
- 17 Cannot rename across volumes.

I/O errors :

- 100 Error when reading from disk.
- 101 Error when writing to disk.
- 102 File not assigned.
- 103 File not open.
- 104 File not opened for input.
- 105 File not opened for output.
- 106 Invalid number.

Fatal errors :

- 150 Disk is write protected.
- 151 Unknown device.
- 152 Drive not ready.
- 153 Unknown command.
- 154 CRC check failed.
- 155 Invalid drive specified..
- 156 Seek error on disk.
- 157 Invalid media type.
- 158 Sector not found.
- 159 Printer out of paper.
- 160 Error when writing to device.
- 161 Error when reading from device.
- 162 Hardware failure.

Errors: None.

See also: All I/O functions.

Listing: `refex/ex35.pp`

Program `Example35;`

{ Program to demonstrate the IOResult function. }

Var `F : text;`

begin

`Assign (f , paramstr (1));`

```
{ $i-}  
Reset ( f );  
{ $i+}  
If IOresult <> 0 then  
    writeln ( 'File ', paramstr(1), ' doesn't exist' )  
else  
    writeln ( 'File ', paramstr(1), ' exists' );  
end.
```

Length

Declaration: `Function Length (S : String) : Integer;`

Description: `Length` returns the length of the string `S`, which is limited to 255 for shortstrings. If the string `S` is empty, 0 is returned.

Note: The length of the string `S` is stored in `S[0]` for shortstrings only. The `Length` function should always be used on ansistrings and widestrings.

Errors: None.

See also: [Pos \(172\)](#)

Listing: `refex/ex36.pp`

Program `Example36;`

{ Program to demonstrate the Length function. }

```
Var S : String;  
    I : Integer;  
  
begin  
    S := '';  
    for i := 1 to 10 do  
        begin  
            S := S + '*';  
            Writeln ( Length(S):2, ' : ', s );  
        end;  
end.
```

Ln

Declaration: `Function Ln (X : Real) : Real;`

Description: `Ln` returns the natural logarithm of the Real parameter `X`. `X` must be positive.

Errors: An run-time error will occur when `X` is negative.

See also: [Exp \(147\)](#), [Power \(172\)](#)

Listing: `refex/ex37.pp`

Program `Example37;`

{ Program to demonstrate the Ln function. }

```
begin
  Writeln (Ln(1));      { Prints 0 }
  Writeln (Ln(Exp(1))); { Prints 1 }
end.
```

Lo

Declaration: Function Lo (O : Word or Longint) : Byte or Word;

Description: Lo returns the low byte of its argument if this is of type Integer or Word. It returns the low word of its argument if this is of type Longint or Cardinal.

Errors: None.

See also: Ord ([170](#)), Chr ([133](#)), Hi ([155](#))

Listing: refex/ex38.pp

Program Example38;

{ Program to demonstrate the Lo function. }

Var L : Longint;
 W : Word;

```
begin
  L:=(1 Shl 16) + (1 Shl 4); { $10010 }
  Writeln (Lo(L));          { Prints 16 }
  W:=(1 Shl 8) + (1 Shl 4); { $110 }
  Writeln (Lo(W));          { Prints 16 }
end.
```

LongJump

Declaration: Procedure LongJump (Var env : Jmp_Buf; Value : Longint);

Description: LongJump jumps to the address in the env jmp_buf, and restores the registers that were stored in it at the corresponding SetJump ([182](#)) call. In effect, program flow will continue at the SetJump call, which will return value instead of 0. If a value equal to zero is passed, it will be converted to 1 before passing it on. The call will not return, so it must be used with extreme care. This can be used for error recovery, for instance when a segmentation fault occurred.

Errors: None.

See also: SetJump ([182](#))

For an example, see SetJump ([182](#))

Low

Declaration: Function Low (Type identifier or variable reference) : Longint;

Description: The return value of Low depends on it's argument:

- 1.If the argument is an ordinal type, Low returns the lowest value in the range of the given ordinal type.
- 2.If the argument is an array type or an array type variable then Low returns the lowest possible value of it's index.

The return type is always the same type as the type of the argument

Errors: None.

See also: High ([155](#)), Ord ([170](#)), Pred ([172](#)), Succ ([188](#))

for an example, see High ([155](#)).

Lowercase

Declaration: `Function Lowercase (C : Char or String) : Char or String;`

Description: `Lowercase` returns the lowercase version of its argument C. If its argument is a string, then the complete string is converted to lowercase. The type of the returned value is the same as the type of the argument.

Errors: None.

See also: Uppcase ([189](#))

Listing: `refex/ex73.pp`

Program `Example73;`

{ Program to demonstrate the Lowercase function. }

Var `I : Longint;`

begin

For `i:=ord('A') to ord('Z')` **do**

write (`lowercase(chr(i))`);

Writeln;

Writeln (`Lowercase('ABCDEFGHIJKLMNOPQRSTUVWXYZ')`);

end.

Mark

Declaration: `Procedure Mark (Var P : Pointer);`

Description: This routine is here for compatibility with Turbo Pascal, but it is not implemented and currently does nothing.

Errors: None.

See also: Getmem ([153](#)), Freemem ([152](#)), New ([168](#)), Dispose ([142](#)), Maxavail ([166](#))

Maxavail

Declaration: `Function Maxavail : Longint;`

Description: `Maxavail` returns the size, in bytes, of the biggest free memory block in the heap.

Remark: The heap grows dynamically if more memory is needed than is available.

Errors: None.

See also: `Release` ([176](#)), `Memavail` ([166](#)), `Freemem` ([152](#)), `Getmem` ([153](#))

Listing: `refex/ex40.pp`

Program `Example40;`

{ Program to demonstrate the MaxAvail function. }

Var

`P : Pointer;`

`I : longint;`

begin

{ This will allocate memory until there is no more memory }

`I:=0;`

While `MaxAvail>=1000 do`

begin

`Inc (I);`

`GetMem (P,1000);`

end;

*{ Default 4MB heap is allocated , so 4000 blocks
should be allocated.*

*When compiled with the -Ch10000 switch , the program
will be able to allocate 10 block }*

WriteLn (`'Allocated ',i,' blocks of 1000 bytes'`);

end.

Memavail

Declaration: `Function Memavail : Longint;`

Description: `Memavail` returns the size, in bytes, of the free heap memory.

Remark: The heap grows dynamically if more memory is needed than is available. The heap size is not equal to the size of the memory available to the operating system, it is internal to the programs created by Free Pascal.

Errors: None.

See also: `Maxavail` ([166](#)), `Freemem` ([152](#)), `Getmem` ([153](#))

Listing: `refex/ex41.pp`

Program `Example41;`

{ Program to demonstrate the MemAvail function. }

Var

`P, PP : Pointer;`

```
begin
  GetMem (P,100);
  GetMem (PP,10000);
  FreeMem (P,100);
  { Due to the heap fragmentation introduced
    By the previous calls , the maximum amount of memory
    isn't equal to the maximum block size available . }
  Writeln ('Total heap available (Bytes) : ',MemAvail);
  Writeln ('Largest block available (Bytes) : ',MaxAvail);
end.
```

Mkdir

Declaration: Procedure Mkdir (const S : string);

Description: Mkdir creates a new directory S.

Errors: Depending on the state of the {SI} switch, a runtime error can be generated if there is an error. In the {SI-} state, use IOResult to check for errors.

See also: Chdir ([132](#)), Rmdir ([178](#))

For an example, see Rmdir ([178](#)).

Move

Declaration: Procedure Move (var Source, Dest; Count : Longint);

Description: Move moves Count bytes from Source to Dest.

Errors: If either Dest or Source is outside the accessible memory for the process, then a run-time error will be generated.

See also: Fillword ([151](#)), Fillchar ([150](#))

Listing: refex/ex42.pp

```
Program Example42;

{ Program to demonstrate the Move function . }

Var S1,S2 : String [30];

begin
  S1:= 'Hello World !';
  S2:= 'Bye , bye !';
  Move (S1,S2,Sizeof(S1));
  Writeln (S2);
end.
```

MoveChar0

Declaration: `procedure MoveChar0(var Src, Dest; Count: longint);`

Description: `MoveChar0` moves `Count` bytes from `Src` to `Dest`, and stops moving if a zero character is found.

Errors: No checking is done to see if `Count` stays within the memory allocated to the process.

See also: [Move \(167\)](#)

Listing: `refex/ex109.pp`

Program `Example109;`

{ Program to demonstrate the MoveChar0 function. }

Var

`Buf1, Buf2 : Array[1..80] of char;`
`I : longint;`

begin

`Randomize;`

`For I:=1 to 80 do`

`Buf1[I]:=chr(Random(16)+Ord('A'));`

`Writeln('Original buffer');`

`writeln(Buf1);`

`Buf1[Random(80)+1]:=#0;`

`MoveChar0(Buf1, Buf2, 80);`

`Writeln('Randomly zero-terminated Buffer');`

`Writeln(Buf2);`

`end.`

New

Declaration: `Procedure New (Var P : Pointer[, Constructor]);`

Description: `New` allocates a new instance of the type pointed to by `P`, and puts the address in `P`. If `P` is an object, then it is possible to specify the name of the constructor with which the instance will be created.

Errors: If not enough memory is available, `Nil` will be returned.

See also: [Dispose \(142\)](#), [Freemem \(152\)](#), [Getmem \(153\)](#), [Memavail \(166\)](#), [Maxavail \(166\)](#)

For an example, see [Dispose \(142\)](#).

Odd

Declaration: `Function Odd (X : Longint) : Boolean;`

Description: `Odd` returns `True` if `X` is odd, or `False` otherwise.

Errors: None.

See also: [Abs \(127\)](#), [Ord \(170\)](#)

Listing: `refex/ex43.pp`

```
Program Example43;  
  
{ Program to demonstrate the Odd function. }  
  
begin  
  If Odd(1) Then  
    WriteLn ( 'Everything OK with 1 ! ');  
  If Not Odd(2) Then  
    WriteLn ( 'Everything OK with 2 ! ');  
end.
```

OctStr

Declaration: `Function OctStr (Value : longint; cnt : byte) : String; Function OctStr (Value : int64; cnt : byte) : String;`

Description: OctStr returns a string with the octal representation of Value. The string has exactly cnt charaters.

Errors: None.

See also: Str ([187](#)), Val ([190](#)), BinStr ([130](#)), HexStr ([154](#))

Listing: refex/ex112.pp

```
Program example112;  
  
{ Program to demonstrate the OctStr function }  
  
Const Value = 45678;  
  
Var I : longint;  
  
begin  
  For I:=1 to 10 do  
    WriteLn ( OctStr(Value, I));  
  For I:=1 to 16 do  
    WriteLn ( OctStr(I, 3));  
end.
```

Ofs

Declaration: `Function Ofs (Var X) : Longint;`

Description: Ofs returns the offset of the address of a variable. This function is only supported for compatibility. In Free Pascal, it returns always the complete address of the variable, since Free Pascal is a 32 bit compiler.

Errors: None.

See also: DSeg ([143](#)), CSeg ([140](#)), Seg ([181](#)), Ptr ([173](#))

Listing: refex/ex44.pp

Program Example44;

{ Program to demonstrate the Ofs function. }

Var W : Pointer;

begin

W:=Pointer(**Ofs**(W)); *{ W contains its own offset. }*
end.

Ord

Declaration: Function Ord (X : Any ordinal type) : Longint;

Description: Ord returns the Ordinal value of a ordinal-type variable X.

Errors: None.

See also: Chr ([133](#)), Succ ([188](#)), Pred ([172](#)), High ([155](#)), Low ([164](#))

Listing: refex/ex45.pp

Program Example45;

{ Program to demonstrate the Ord,Pred,Succ functions. }

Type

TEnum = (Zero , One , Two , Three , Four);

Var

X : Longint;
Y : TEnum;

begin

X:=125;
Writeln (**Ord**(X)); *{ Prints 125 }*
X:=**Pred**(X);
Writeln (**Ord**(X)); *{ prints 124 }*
Y:= One;
Writeln (**Ord**(y)); *{ Prints 1 }*
Y:=**Succ**(Y);
Writeln (**Ord**(Y)); *{ Prints 2 }*
end.

Paramcount

Declaration: Function Paramcount : Longint;

Description: Paramcount returns the number of command-line arguments. If no arguments were given to the running program, 0 is returned.

Errors: None.

See also: Paramstr ([171](#))

Listing: refex/ex46.pp

Program Example46;

```
{ Program to demonstrate the ParamCount and ParamStr functions. }
Var
  I : Longint;

begin
  Writeln (paramstr(0), ' : Got ',ParamCount, ' command-line parameters: ');
  For i:=1 to ParamCount do
    Writeln (ParamStr (i));
end.
```

Paramstr

Declaration: Function Paramstr (L : Longint) : String;

Description: Paramstr returns the L-th command-line argument. L must be between 0 and Paramcount, these values included. The zeroth argument is the path and file name with which the program was started.

The command-line parameters will be truncated to a length of 255, even though the operating system may support bigger command-lines. The Objpas unit (used in objfpc or delphi mode) define versions of Paramstr which return the full-length command-line arguments.

When the complete command-line must be accessed, the argv pointer should be used to retrieve the real values of the command-line parameters.

Errors: None.

See also: Paramcount ([170](#))For an example, see Paramcount ([170](#)).**Pi**

Declaration: Function Pi : Real;

Description: Pi returns the value of Pi (3.1415926535897932385).

Errors: None.

See also: Cos ([140](#)), Sin ([184](#))**Listing:** refex/ex47.pp

Program Example47;

```
{ Program to demonstrate the Pi function. }

begin
  Writeln (Pi);           {3.1415926}
  Writeln (Sin(Pi));
end.
```

Pos

Declaration: `Function Pos (Const Substr : String; Const S : String) : Integer;`

Description: `Pos` returns the index of `Substr` in `S`, if `S` contains `Substr`. In case `Substr` isn't found, 0 is returned. The search is case-sensitive.

Errors: None

See also: `Length` ([163](#)), `Copy` ([139](#)), `Delete` ([141](#)), `Insert` ([160](#))

Listing: `refex/ex48.pp`

Program `Example48;`

{ Program to demonstrate the Pos function. }

Var

`S : String;`

begin

`S := 'The first space in this sentence is at position : ';`

`WriteLn (S, pos(' ', S));`

`S := 'The last letter of the alphabet doesn't appear in this sentence ';`

`If (Pos ('Z', S) = 0) and (Pos ('z', S) = 0) then`

`WriteLn (S);`

end.

Power

Declaration: `Function Power (base, expon : Real) : Real;`

Description: `Power` returns the value of `base` to the power `expon`. `Base` and `expon` can be of type `Longint`, in which case the result will also be a `Longint`.

The function actually returns `Exp (expon * Ln (base))`

Errors: None.

See also: `Exp` ([147](#)), `Ln` ([163](#))

Listing: `refex/ex78.pp`

Program `Example78;`

{ Program to demonstrate the Power function. }

begin

`WriteLn (Power(exp(1.0), 1.0):8:2); { Should print 2.72 }`

end.

Pred

Declaration: `Function Pred (X : Any ordinal type) : Same type;`

Description: `Pred` returns the element that precedes the element that was passed to it. If it is applied to the first value of the ordinal type, and the program was compiled with range checking on (`{ $R+ }`), then a run-time error will be generated.

Errors: Run-time error 201 is generated when the result is out of range.

See also: Ord ([170](#)), Pred ([172](#)), High ([155](#)), Low ([164](#))

for an example, see Ord ([170](#))

Ptr

Declaration: `Function Ptr (Sel, Off : Longint) : Pointer;`

Description: `Ptr` returns a pointer, pointing to the address specified by segment `Sel` and offset `Off`.

Remark:

1. In the 32-bit flat-memory model supported by Free Pascal, this function is obsolete.
2. The returned address is simply the offset.

Errors: None.

See also: Addr ([127](#))

Listing: `refex/ex59.pp`

Program `Example59;`

{ Program to demonstrate the Ptr function. }

Var `P : ^String;`
`S : String;`

begin

`S := 'Hello , World ! ';`
`P := Ptr (Seg (S), Longint (Ofs (S)));`
{ P now points to S ! }
`Writeln (P^);`

end.

Random

Declaration: `Function Random [(L : Longint)] : Longint or Real;`

Description: `Random` returns a random number larger or equal to 0 and strictly less than `L`. If the argument `L` is omitted, a Real number between 0 and 1 is returned. (0 included, 1 excluded)

Errors: None.

See also: Randomize ([174](#))

Listing: `refex/ex49.pp`

Program `Example49;`

{ Program to demonstrate the Random and Randomize functions. }

Var `I, Count, guess : Longint;`
`R : Real;`

```
begin
  Randomize; { This way we generate a new sequence every time
               the program is run}
  Count:=0;
  For i:=1 to 1000 do
    If Random>0.5 then inc(Count);
  Writeln ( 'Generated ',Count, ' numbers > 0.5 ');
  Writeln ( 'out of 1000 generated numbers.' );
  count:=0;
  For i:=1 to 5 do
    begin
      write ( 'Guess a number between 1 and 5 : ');
      readln(Guess);
      If Guess=Random(5)+1 then inc(count);
    end;
  Writeln ( 'You guessed ',Count, ' out of 5 correct.' );
end.
```

Randomize

Declaration: `Procedure Randomize ;`

Description: `Randomize` initializes the random number generator of Free Pascal, by giving a value to `Randseed`, calculated with the system clock.

Errors: None.

See also: [Random \(173\)](#)

For an example, see [Random \(173\)](#).

Read

Declaration: `Procedure Read ([Var F : Any file type], V1 [, V2, ... , Vn]);`

Description: `Read` reads one or more values from a file `F`, and stores the result in `V1`, `V2`, etc.; If no file `F` is specified, then standard input is read. If `F` is of type `Text`, then the variables `V1`, `V2` etc. must be of type `Char`, `Integer`, `Real`, `String` or `PChar`. If `F` is a typed file, then each of the variables must be of the type specified in the declaration of `F`. Untyped files are not allowed as an argument.

Errors: If no data is available, a run-time error is generated. This behavior can be controlled with the `{ $i }` compiler switch.

See also: [Readln \(175\)](#), [Blockread \(131\)](#), [Write \(191\)](#), [Blockwrite \(131\)](#)

Listing: `refex/ex50.pp`

Program `Example50;`

{ Program to demonstrate the Read(Ln) function. }

```
Var S : String;
    C : Char;
    F : File of char;
```

```
begin
```

```
Assign (F, 'ex50.pp');
Reset (F);
C:= 'A';
WriteLn ('The characters before the first space in ex50.pp are : ');
While not Eof(f) and (C<>' ') do
  Begin
    Read (F,C);
    Write (C);
  end;
WriteLn;
Close (F);
WriteLn ('Type some words. An empty line ends the program. ');
repeat
  ReadLn (S);
until S='';
end.
```

ReadLn

Declaration: Procedure ReadLn [Var F : Text], V1 [, V2, ... , Vn]);

Description: Read reads one or more values from a file F, and stores the result in V1, V2, etc. After that it goes to the next line in the file. The end of the line is marked by the `LineEnding` character sequence (which is platform dependent). The end-of-line marker is not considered part of the line and is ignored.

If no file F is specified, then standard input is read. The variables V1, V2 etc. must be of type Char, Integer, Real, String or PChar.

Errors: If no data is available, a run-time error is generated. This behavior can be controlled with the `{ $i }` compiler switch.

See also: Read ([174](#)), Blockread ([131](#)), Write ([191](#)), Blockwrite ([131](#))

For an example, see Read ([174](#)).

Real2Double

Declaration: Function Real2Double(r : real48) : double;

Description: The `Real2Double` function converts a Turbo Pascal style real (6 bytes long) to a native Free Pascal double type. It can be used e.g. to read old binary TP files with FPC and convert them to Free Pascal binary files.

Note that the assignment operator has been overloaded so a `Real48` type can be assigned directly to a double or extended.

Errors: None.

See also:

Listing: refex/ex110.pp

program Example110;

{ Program to demonstrate the Real2Double function. }


```
Var
  i : integer;
  R : Real48;
  D : Double;
  E : Extended;
  F : File of Real48;

begin
  Assign(F, 'reals.dat');
  Reset(f);
  For i:=1 to 10 do
    begin
      Read(F,R);
      D:=Real2Double(R);
      WriteLn('Real ',i,' : ',D);
      D:=R;
      WriteLn('Real (direct to double) ',i,' : ',D);
      E:=R;
      WriteLn('Real (direct to Extended) ',i,' : ',E);
    end;
  Close(f);
end.
```

Release

Declaration: `Procedure Release (Var P : pointer);`

Description: This routine is here for compatibility with Turbo Pascal, but it is not implemented and currently does nothing.

Errors: None.

See also: Mark ([165](#)), Memavail ([166](#)), Maxavail ([166](#)), Getmem ([153](#)), Freemem ([152](#)) New ([168](#)), Dispose ([142](#))

Rename

Declaration: `Procedure Rename (Var F : Any Filetype; Const S : String);`

Description: Rename changes the name of the assigned file F to S. F must be assigned, but not opened.

Errors: Depending on the state of the `{SI}` switch, a runtime error can be generated if there is an error. In the `{SI-}` state, use `IOResult` to check for errors.

See also: Erase ([144](#))

Listing: `refex/ex77.pp`

Program Example77;

```
{ Program to demonstrate the Rename function. }
Var F : Text;

begin
  Assign (F, paramstr(1));
  Rename (F, paramstr(2));
end.
```

Reset

Declaration: `Procedure Reset (Var F : Any File Type[; L : Longint]);`

Description: `Reset` opens a file `F` for reading. `F` can be any file type. If `F` is a text file, or refers to standard I/O (e.g. `”) then it is opened read-only, otherwise it is opened using the mode specified in filemode.`

If `F` is an untyped file, the record size can be specified in the optional parameter `L`. A default value of 128 is used.

File sharing is not taken into account when calling `Reset`.

Errors: Depending on the state of the `{SI}` switch, a runtime error can be generated if there is an error. In the `{SI-}` state, use `IOResult` to check for errors.

See also: `Rewrite` (177), `Assign` (129), `Close` (133), `Append` (128)

Listing: `refex/ex51.pp`

Program `Example51;`

{ Program to demonstrate the Reset function. }

Function `FileExists (Name : String) : boolean;`

Var `F : File;`

begin

{SI-}

`Assign (F,Name);`

`Reset (F);`

{SI+}

`FileExists := (IOResult=0) and (Name<>'');`

`Close (f);`

end;

begin

`If FileExists (Paramstr(1)) then`

`WriteLn ('File found')`

`else`

`WriteLn ('File NOT found');`

end.

Rewrite

Declaration: `Procedure Rewrite (Var F : Any File Type[; L : Longint]);`

Description: `Rewrite` opens a file `F` for writing. `F` can be any file type. If `F` is an untyped or typed file, then it is opened for reading and writing. If `F` is an untyped file, the record size can be specified in the optional parameter `L`. Default a value of 128 is used. if `Rewrite` finds a file with the same name as `F`, this file is truncated to length 0. If it doesn't find such a file, a new file is created.

Contrary to Turbo Pascal, Free Pascal opens the file with mode `fmoutput`. If it should be opened in `fminout` mode, an extra call to `Reset` (177) is needed.

File sharing is not taken into account when calling `Rewrite`.

Errors: Depending on the state of the `{SI}` switch, a runtime error can be generated if there is an error. In the `{SI-}` state, use `IOResult` to check for errors.

See also: [Reset \(177\)](#), [Assign \(129\)](#), [Close \(133\)](#), [Flush \(151\)](#), [Append \(128\)](#)

Listing: `refex/ex52.pp`

Program `Example52`;

{ Program to demonstrate the Rewrite function. }

Var `F : File;`
 `I : longint;`

begin

`Assign (F, 'Test.tmp');`
 { Create the file. Recordsize is 4 }
 `Rewrite (F, Sizeof(I));`
 For `I:=1 to 10 do`
 `BlockWrite (F,I,1);`
 `close (f);`
 { F contains now a binary representation of
 10 longints going from 1 to 10 }

end.

Rmdir

Declaration: `Procedure Rmdir (const S : string);`

Description: `Rmdir` removes the directory `S`.

Errors:

Errors: Depending on the state of the `{ $I }` switch, a runtime error can be generated if there is an error. In the `{ $I- }` state, use `IOResult` to check for errors.

See also: [Chdir \(132\)](#), [Mkdir \(167\)](#)

Listing: `refex/ex53.pp`

Program `Example53`;

{ Program to demonstrate the Mkdir and Rmdir functions. }

Const `D : String[8] = 'TEST.DIR';`

Var `S : String;`

begin

`WriteLn ('Making directory ',D);`
 `Mkdir (D);`
 `WriteLn ('Changing directory to ',D);`
 `ChDir (D);`
 `GetDir (0,S);`
 `WriteLn ('Current Directory is : ',S);`
 `WRiteln ('Going back');`
 `ChDir ('..');`
 `WriteLn ('Removing directory ',D);`
 `Rmdir (D);`

end.

Round

Declaration: `Function Round (X : Real) : Longint;`

Description: `Round` rounds `X` to the closest integer, which may be bigger or smaller than `X`.

Errors: None.

See also: `Frac` ([152](#)), `Int` ([161](#)), `Trunc` ([188](#))

Listing: `refex/ex54.pp`

Program `Example54;`

{ Program to demonstrate the Round function. }

begin

`WriteLn (Round(1234.56)); { Prints 1235 }`

`WriteLn (Round(-1234.56)); { Prints -1235 }`

`WriteLn (Round(12.3456)); { Prints 12 }`

`WriteLn (Round(-12.3456)); { Prints -12 }`

end.

Runerror

Declaration: `Procedure Runerror (ErrorCode : Word);`

Description: `Runerror` stops the execution of the program, and generates a run-time error `ErrorCode`.

Errors: None.

See also: `Exit` ([146](#)), `Halt` ([154](#))

Listing: `refex/ex55.pp`

Program `Example55;`

{ Program to demonstrate the RunError function. }

begin

{ The program will stop end emit a run-error 106 }

`RunError (106);`

end.

Seek

Declaration: `Procedure Seek (Var F; Count : Longint);`

Description: `Seek` sets the file-pointer for file `F` to record `Nr. Count`. The first record in a file has `Count=0`. `F` can be any file type, except `Text`. If `F` is an untyped file, with no record size specified in `Reset` ([177](#)) or `Rewrite` ([177](#)), 128 is assumed.

Errors: Depending on the state of the `{ $I }` switch, a runtime error can be generated if there is an error. In the `{ $I- }` state, use `IOResult` to check for errors.

See also: `Eof` ([143](#)), `SeekEof` ([180](#)), `SeekEoln` ([181](#))

Listing: refex/ex56.pp

```
Program Example56;

{ Program to demonstrate the Seek function. }

Var
  F : File;
  I, J : longint;

begin
  { Create a file and fill it with data }
  Assign (F, 'test.tmp');
  Rewrite(F); { Create file }
  Close(f);
  FileMode:=2;
  ReSet (F, Sizeof(i)); { Opened read/write }
  For I:=0 to 10 do
    BlockWrite (F, I, 1);
  { Go Back to the beginning of the file }
  Seek(F, 0);
  For I:=0 to 10 do
    begin
      BlockRead (F, J, 1);
      If J<>I then
        WriteLn ('Error: expected ', i, ', got ', j);
      end;
    Close (f);
  end.
```

SeekEof

Declaration: `Function SeekEof [(Var F : text)] : Boolean;`

Description: `SeekEof` returns `True` if the file-pointer is at the end of the file. It ignores all whitespace. Calling this function has the effect that the file-position is advanced until the first non-whitespace character or the end-of-file marker is reached. If the end-of-file marker is reached, `True` is returned. Otherwise, `False` is returned. If the parameter `F` is omitted, standard `Input` is assumed.

Errors: A run-time error is generated if the file `F` isn't opened.

See also: `Eof` ([143](#)), `SeekEoln` ([181](#)), `Seek` ([179](#))

Listing: refex/ex57.pp

```
Program Example57;

{ Program to demonstrate the SeekEof function. }
Var C : Char;

begin
  { this will print all characters from standard input except
    Whitespace characters. }
  While Not SeekEof do
    begin
      Read (C);
      Write (C);
    end;
```

end.

SeekEoln

Declaration: `Function SeekEoln [(Var F : text)] : Boolean;`

Description: `SeekEoln` returns `True` if the file-pointer is at the end of the current line. It ignores all whitespace. Calling this function has the effect that the file-position is advanced until the first non-whitespace character or the end-of-line marker is reached. If the end-of-line marker is reached, `True` is returned. Otherwise, `False` is returned. The end-of-line marker is defined as `#10`, the `LineFeed` character. If the parameter `F` is omitted, standard `Input` is assumed.

Errors: A run-time error is generated if the file `F` isn't opened.

See also: `Eof` ([143](#)), `SeekEof` ([180](#)), `Seek` ([179](#))

Listing: `refex/ex58.pp`

Program `Example58;`

```
{ Program to demonstrate the SeekEoln function. }
Var
  C : Char;

begin
  { This will read the first line of standard output and print
    all characters except whitespace. }
  While not SeekEoln do
    Begin
      Read (c);
      Write (c);
    end;
end.
```

Seg

Declaration: `Function Seg (Var X) : Longint;`

Description: `Seg` returns the segment of the address of a variable. This function is only supported for compatibility. In Free Pascal, it returns always 0, since Free Pascal is a 32 bit compiler, segments have no meaning.

Errors: None.

See also: `DSeg` ([143](#)), `CSeg` ([140](#)), `Ofs` ([169](#)), `Ptr` ([173](#))

Listing: `refex/ex60.pp`

Program `Example60;`

```
{ Program to demonstrate the Seg function. }
Var
  W : Word;

begin
  W:=Seg(W); { W contains its own Segment }
end.
```

SetMemoryManager

Declaration: `procedure SetMemoryManager(const MemMgr: TMemoryManager);`

Description: `SetMemoryManager` sets the current memory manager record to `MemMgr`.

Errors: None.

See also: `GetMemoryManager` ([154](#)), `IsMemoryManagerSet` ([161](#))

For an example, see [Programmers guide](#).

SetJump

Declaration: `Function SetJump (Var Env : Jump_Buf) : Longint;`

Description: `SetJump` fills `env` with the necessary data for a jump back to the point where it was called. It returns zero if called in this way. If the function returns nonzero, then it means that a call to `LongJump` ([164](#)) with `env` as an argument was made somewhere in the program.

Errors: None.

See also: `LongJump` ([164](#))

Listing: `refex/ex79.pp`

```
program example79;

{ Program to demonstrate the setjmp, longjmp functions }

procedure dojmp(var env : jmp_buf; value : longint);

begin
    value:=2;
    WriteLn ( 'Going to jump ! ');
    { This will return to the setjmp call ,
      and return value instead of 0 }
    longjmp(env,value);
end;

var env : jmp_buf;

begin
    if setjmp(env)=0 then
        begin
            writeln ( 'Passed first time.' );
            dojmp(env,2);
        end
    else
        writeln ( 'Passed second time.' );
    end.
```

SetLength

Declaration: `Procedure SetLength(var S : String; Len : Longint);`

Description: `SetLength` sets the length of the string `S` to `Len`. `S` can be an ansistring, a short string or a widestring. For `ShortStrings`, `Len` can maximally be 255. For `AnsiStrings` it can have any value. For `AnsiString` strings, `SetLength` *must* be used to set the length of the string.

Errors: None.

See also: `Length` ([163](#))

Listing: `refex/ex85.pp`

Program `Example85`;

{ Program to demonstrate the SetLength function. }

Var `S : String`;

begin

FillChar (`S[1]`, 100, #32);

Setlength (`S`, 100);

Writeln (' ', `S`, ' ');

end.

SetString

Declaration: `Procedure SetString`(`var S : String`; `Buf : PChar`; `Len : Longint`);

Description: `SetString` sets the length of the string `S` to `Len` and if `Buf` is non-nil, copies `Len` characters from `Buf` into `S`. `S` can be an ansistring, a short string or a widestring. For `ShortStrings`, `Len` can maximally be 255.

Errors: None.

See also: `SetLength` ([182](#))

SetTextBuf

Declaration: `Procedure SetTextBuf` (`Var f : Text`; `Var Buf[; Size : Word]`);

Description: `SetTextBuf` assigns an I/O buffer to a text file. The new buffer is located at `Buf` and is `Size` bytes long. If `Size` is omitted, then `SizeOf (Buf)` is assumed. The standard buffer of any text file is 128 bytes long. For heavy I/O operations this may prove too slow. The `SetTextBuf` procedure allows to set a bigger buffer for the IO of the application, thus reducing the number of system calls, and thus reducing the load on the system resources. The maximum size of the newly assigned buffer is 65355 bytes.

Remark:

- Never assign a new buffer to an opened file. A new buffer can be assigned immediately after a call to `Rewrite` ([177](#)), `Reset` ([177](#)) or `Append`, but not after the file was read from/written to. This may cause loss of data. If a new buffer must be assigned after read/write operations have been performed, the file should be flushed first. This will ensure that the current buffer is emptied.
- Take care that the assigned buffer is always valid. If a local variable is assigned as a buffer, then after the program exits the local program block, the buffer will no longer be valid, and stack problems may occur.

Errors: No checking on Size is done.

See also: Assign ([129](#)), Reset ([177](#)), Rewrite ([177](#)), Append ([128](#))

Listing: refex/ex61.pp

Program Example61;

{ Program to demonstrate the SetTextBuf function. }

Var

Fin , Fout : Text;
Ch : Char;
Bufin , Bufout : **Array**[1..10000] of byte;

begin

Assign (Fin , paramstr(1));
Reset (Fin);
Assign (Fout , paramstr(2));
Rewrite (Fout);
{ This is harmless before IO has begun }
{ Try this program again on a big file ,
after commenting out the following 2
lines and recompiling it. }

SetTextBuf (Fin , Bufin);
SetTextBuf (Fout , Bufout);

While not eof(Fin) **do**

begin

Read (Fin , ch);
write (Fout , ch);
end;

Close (Fin);
Close (Fout);

end.

Sin

Declaration: `Function Sin (X : Real) : Real;`

Description: Sin returns the sine of its argument X, where X is an angle in radians.

If the absolute value of the argument is larger than 2^{63} , then the result is undefined.

Errors: None.

See also: Cos ([140](#)), Pi ([171](#)), Exp ([147](#)), Ln ([163](#))

Listing: refex/ex62.pp

Program Example62;

{ Program to demonstrate the Sin function. }

begin

Writeln (Sin(Pi):0:1); *{ Prints 0.0 }*
Writeln (Sin(Pi/2):0:1); *{ Prints 1.0 }*

end.

SizeOf

Declaration: `Function SizeOf (X : Any Type) : Longint;`

Description: `SizeOf` returns the size, in bytes, of any variable or type-identifier.

Remark: This isn't really a RTL function. Its result is calculated at compile-time, and hard-coded in the executable.

Errors: None.

See also: `Addr` ([127](#))

Listing: `refex/ex63.pp`

Program `Example63;`

```
{ Program to demonstrate the SizeOf function. }
Var
  I : Longint;
  S : String [10];

begin
  WriteLn (SizeOf(I)); { Prints 4 }
  WriteLn (SizeOf(S)); { Prints 11 }
end.
```

Sptr

Declaration: `Function Sptr : Pointer;`

Description: `Sptr` returns the current stack pointer.

Errors: None.

See also: `SSeg` ([186](#))

Listing: `refex/ex64.pp`

Program `Example64;`

```
{ Program to demonstrate the SPtr function. }
Var
  P : Longint;

begin
  P := Sptr; { P Contains now the current stack position. }
end.
```

Sqr

Declaration: `Function Sqr (X : Real) : Real;`

Description: `Sqr` returns the square of its argument X.

Errors: None.

See also: `Sqrt` ([186](#)), `Ln` ([163](#)), `Exp` ([147](#))

Listing: refex/ex65.pp

Program Example65;

```
{ Program to demonstrate the Sqr function. }
Var i : Integer;

begin
  For i:=1 to 10 do
    writeln (Sqr(i):3);
  end.
```

Sqrt

Declaration: Function Sqrt (X : Real) : Real;

Description: Sqrt returns the square root of its argument X, which must be positive.

Errors: If X is negative, then a run-time error is generated.

See also: Sqr ([185](#)), Ln ([163](#)), Exp ([147](#))**Listing:** refex/ex66.pp

Program Example66;

```
{ Program to demonstrate the Sqrt function. }

begin
  Writeln (Sqrt(4):0:3); { Prints 2.000 }
  Writeln (Sqrt(2):0:3); { Prints 1.414 }
end.
```

SSeg

Declaration: Function SSeg : Longint;

Description: SSeg returns the Stack Segment. This function is only supported for compatibility reasons, as Sptr returns the correct contents of the stackpointer.

Errors: None.

See also: Sptr ([185](#))**Listing:** refex/ex67.pp

Program Example67;

```
{ Program to demonstrate the SSeg function. }
Var W : Longint;

begin
  W:=SSeg;
end.
```

Str

Declaration: `Procedure Str (Var X[:NumPlaces[:Decimals]]; Var S : String);`

Description: `Str` returns a string which represents the value of `X`. `X` can be any numerical type. The optional `NumPlaces` and `Decimals` specifiers control the formatting of the string.

Errors: None.

See also: [Val \(190\)](#)

Listing: `refex/ex68.pp`

Program `Example68;`

{ Program to demonstrate the Str function. }

Var `S : String;`

Function `IntToStr (I : Longint) : String;`

Var `S : String;`

begin

`Str (I,S);`

`IntToStr:=S;`

end;

begin

`S:= '*'+IntToStr(-233)+ '*';`

`WriteLn (S);`

end.

StringOfChar

Declaration: `Function StringOfChar(c : char;l : Integer) : String;`

Description: `StringOfChar` creates a new `String` of length `l` and fills it with the character `c`.

It is equivalent to the following calls:

`SetLength(StringOfChar,l);`

`FillChar(Pointer(StringOfChar)^,Length(StringOfChar),c);`

Errors: None.

See also: [SetLength \(182\)](#)

Listing: `refex/ex97.pp`

Program `Example97;`

{ \$H+ }

{ Program to demonstrate the StringOfChar function. }

Var `S : String;`

begin

```
S:=StringOfChar(' ',40)+'Aligned at column 41.';  
WriteLn(s);  
end.
```

Succ

Declaration: `Function Succ (X : Any ordinal type) : Same type;`

Description: `Succ` returns the element that succeeds the element that was passed to it. If it is applied to the last value of the ordinal type, and the program was compiled with range checking on (`{ $R+ }`), then a run-time error will be generated.

Errors: Run-time error 201 is generated when the result is out of range.

See also: `Ord` ([170](#)), `Pred` ([172](#)), `High` ([155](#)), `Low` ([164](#))

for an example, see `Ord` ([170](#)).

Swap

Declaration: `Function Swap (X) : Type of X;`

Description: `Swap` swaps the high and low order bytes of `X` if `X` is of type `Word` or `Integer`, or swaps the high and low order words of `X` if `X` is of type `Longint` or `Cardinal`. The return type is the type of `X`

Errors: None.

See also: `Lo` ([164](#)), `Hi` ([155](#))

Listing: `refex/ex69.pp`

Program `Example69;`

```
{ Program to demonstrate the Swap function. }  
Var W : Word;  
    L : Longint;  
  
begin  
  W:=$1234;  
  W:=Swap(W);  
  if W<>$3412 then  
    writeln ('Error when swapping word !');  
  L:=$12345678;  
  L:=Swap(L);  
  if L<>$56781234 then  
    writeln ('Error when swapping Longint !');  
end.
```

Trunc

Declaration: `Function Trunc (X : Real) : Longint;`

Description: `Trunc` returns the integer part of `X`, which is always smaller than (or equal to) `X` in absolute value.

Errors: None.

See also: [Frac \(152\)](#), [Int \(161\)](#), [Round \(179\)](#)

Listing: `refex/ex70.pp`

```
Program Example70;  
  
{ Program to demonstrate the Trunc function. }  
  
begin  
  Writeln (Trunc(123.456)); { Prints 123 }  
  Writeln (Trunc(-123.456)); { Prints -123 }  
  Writeln (Trunc(12.3456)); { Prints 12 }  
  Writeln (Trunc(-12.3456)); { Prints -12 }  
end.
```

Truncate

Declaration: `Procedure Truncate (Var F : file);`

Description: `Truncate` truncates the (opened) file `F` at the current file position.

Errors: Depending on the state of the `{ $I }` switch, a runtime error can be generated if there is an error. In the `{ $I- }` state, use `IOResult` to check for errors.

See also: [Append \(128\)](#), [Filepos \(148\)](#), [Seek \(179\)](#)

Listing: `refex/ex71.pp`

```
Program Example71;  
  
{ Program to demonstrate the Truncate function. }  
  
Var F : File of longint;  
    I, L : Longint;  
  
begin  
  Assign (F, 'test.tmp');  
  Rewrite (F);  
  For I:=1 to 10 Do  
    Write (F, I);  
  Writeln ( 'Filesize before Truncate : ', FileSize(F));  
  Close (f);  
  Reset (F);  
  Repeat  
    Read (F, I);  
  Until i=5;  
  Truncate (F);  
  Writeln ( 'Filesize after Truncate : ', FileSize(F));  
  Close (f);  
end.
```

Uppcase

Declaration: `Function Uppcase (C : Char or string) : Char or String;`

Description: `Uppcase` returns the uppercase version of its argument `C`. If its argument is a string, then the complete string is converted to uppercase. The type of the returned value is the same as the type of the argument.

Errors: None.

See also: `Lowercase` ([165](#))

Listing: `refex/ex72.pp`

Program `Example72`;

```
{ Program to demonstrate the Uppcase function. }

Var I : Longint;

begin
  For i:=ord('a') to ord('z') do
    write (upcase(chr(i)));
  Writeln;
  { This doesn't work in TP, but it does in Free Pascal }
  Writeln (Uppcase('abcdefghijklmnopqrstuvwxyz'));
end.
```

Val

Declaration: `Procedure Val (const S : string;var V;var Code : word);`

Description: `Val` converts the value represented in the string `S` to a numerical value, and stores this value in the variable `V`, which can be of type `Longint`, `Real` and `Byte`. If the conversion isn't successful, then the parameter `Code` contains the index of the character in `S` which prevented the conversion. The string `S` is allowed to contain spaces in the beginning.

The string `S` can contain a number in decimal, hexadecimal, binary or octal format, as described in the language reference.

Errors: If the conversion doesn't succeed, the value of `Code` indicates the position where the conversion went wrong.

See also: `Str` ([187](#))

Listing: `refex/ex74.pp`

Program `Example74`;

```
{ Program to demonstrate the Val function. }
Var I, Code : Integer;

begin
  Val (ParamStr (1), I, Code);
  If Code<>0 then
    Writeln ('Error at position ',code,' : ',Paramstr(1)[Code])
  else
    Writeln ('Value : ',I);
end.
```

Write

Declaration: `Procedure Write ([Var F : Any filetype;] V1 [; V2; ... , Vn]);`

Description: `Write` writes the contents of the variables `V1`, `V2` etc. to the file `F`. `F` can be a typed file, or a `Text` file. If `F` is a typed file, then the variables `V1`, `V2` etc. must be of the same type as the type in the declaration of `F`. Untyped files are not allowed. If the parameter `F` is omitted, standard output is assumed. If `F` is of type `Text`, then the necessary conversions are done such that the output of the variables is in human-readable format. This conversion is done for all numerical types. Strings are printed exactly as they are in memory, as well as `PChar` types. The format of the numerical conversions can be influenced through the following modifiers: `OutputVariable : NumChars [: Decimals]` This will print the value of `OutputVariable` with a minimum of `NumChars` characters, from which `Decimals` are reserved for the decimals. If the number cannot be represented with `NumChars` characters, `NumChars` will be increased, until the representation fits. If the representation requires less than `NumChars` characters then the output is filled up with spaces, to the left of the generated string, thus resulting in a right-aligned representation. If no formatting is specified, then the number is written using its natural length, with nothing in front of it if it's positive, and a minus sign if it's negative. Real numbers are, by default, written in scientific notation.

Errors: If an error occurs, a run-time error is generated. This behavior can be controlled with the `{ $i }` switch.

See also: `WriteLn` (191), `Read` (174), `ReadLn` (175), `Blockwrite` (131)

WriteLn

Declaration: `Procedure WriteLn ([Var F : Text;] [V1 [; V2; ... , Vn]]);`

Description: `WriteLn` does the same as `Write` (191) for text files, and emits a Carriage Return - LineFeed character pair after that. If the parameter `F` is omitted, standard output is assumed. If no variables are specified, a Carriage Return - LineFeed character pair is emitted, resulting in a new line in the file `F`.

Remark: Under LINUX and UNIX, the Carriage Return character is omitted, as customary in Unix environments.

Errors: If an error occurs, a run-time error is generated. This behavior can be controlled with the `{ $i }` switch.

See also: `Write` (191), `Read` (174), `ReadLn` (175), `Blockwrite` (131)

Listing: `refex/ex75.pp`

Program `Example75;`

```
{ Program to demonstrate the Write(In) function. }

Var
  F : File of Longint;
  L : Longint;

begin
  Write ('This is on the first line ! '); { No CR/LF pair! }
  Writeln ('And this too...');
  Writeln ('But this is already on the second line...');
  Assign (f, 'test.tmp');
  Rewrite (f);
  For L:=1 to 10 do
    write (F,L); { No writeln allowed here ! }
```



```
    Close ( f );  
end.
```

Chapter 15

The OBJPAS unit

The `objpas` unit is meant for compatibility with Object Pascal as implemented by Delphi. The unit is loaded automatically by the Free Pascal compiler whenever the `Delphi` or `objfpc` mode is entered, either through the command line switches `-Sd` or `-Sh` or with the `{ $MODE DELPHI }` or `{ $MODE OBJFPC }` directives.

It redefines some basic pascal types, introduces some functions for compatibility with Delphi's system unit, and introduces some methods for the management of the resource string tables.

15.1 Types

The `objpas` unit redefines two integer types, for compatibility with Delphi:

```
type
  smallint = system.integer;
  integer  = system.longint;
```

The resource string tables can be managed with a callback function which the user must provide: `TResourceIterator`.

```
Type
  TResourceIterator =
    Function (Name, Value : AnsiString; Hash : Longint): AnsiString;
```

15.2 Functions and Procedures

AssignFile

Declaration: `Procedure AssignFile(Var f: FileType; Name: Character type);`

Description: `AssignFile` is completely equivalent to the system unit's `Assign` (129) function: It assigns `Name` to a function of any type (`FileType` can be `Text` or a typed or untyped `File` variable). `Name` can be a string, a single character or a `PChar`.

It is most likely introduced to avoid confusion between the regular `Assign` (129) function and the `Assign` method of `TPersistent` in the Delphi VCL.

Errors: None.

See also: [CloseFile \(194\)](#), [Assign \(129\)](#), [Reset \(177\)](#), [Rewrite \(177\)](#), [Append \(128\)](#)

Listing: [refex/ex88.pp](#)

```
Program Example88;  
  
  { Program to demonstrate the AssignFile and CloseFile functions. }  
  
  {$MODE Delphi}  
  
  Var F : text;  
  
  begin  
    AssignFile(F, 'textfile.tmp');  
    Rewrite(F);  
    WriteLn (F, 'This is a silly example of AssignFile and CloseFile. ');  
    CloseFile(F);  
  end.
```

CloseFile

Declaration: `Procedure CloseFile(Var F: FileType);`

Description: `CloseFile` flushes and closes a file `F` of any file type. `F` can be `Text` or a typed or untyped `File` variable. After a call to `CloseFile`, any attempt to write to the file `F` will result in an error.

It is most likely introduced to avoid confusion between the regular [Close \(133\)](#) function and the `Close` method of `TForm` in the Delphi VCL.

Errors: None.

See also: [Close \(133\)](#), [AssignFile \(193\)](#), [Reset \(177\)](#), [Rewrite \(177\)](#), [Append \(128\)](#)

for an example, see [AssignFile \(193\)](#).

Freemem

Declaration: `Procedure FreeMem(Var p:pointer[;Size:Longint]);`

Description: `FreeMem` releases the memory reserved by a call to [GetMem \(195\)](#). The (optional) `Size` parameter is ignored, since the object pascal version of `GetMem` stores the amount of memory that was requested.

Be sure not to release memory that was not obtained with the `Getmem` call in `Objpas`. Normally, this should not happen, since `objpas` changes the default memory manager to it's own memory manager.

Errors: None.

See also: [Freemem \(152\)](#), [GetMem \(195\)](#), [Getmem \(153\)](#)

Listing: [refex/ex89.pp](#)

```
Program Example89;  
  
  { Program to demonstrate the FreeMem function. }  
  {$Mode Delphi}
```

```
Var P : Pointer;  
  
begin  
  WriteLn ( 'Memory before : ',Memavail);  
  GetMem(P,10000);  
  FreeMem(P);  
  WriteLn ( 'Memory after : ',Memavail);  
end.
```

Getmem

Declaration: `Procedure Getmem(Var P:pointer;Size:Longint);`

Description: `GetMem` reserves `Size` bytes of memory on the heap and returns a pointer to it in `P`. `Size` is stored at offset -4 of the result, and is used to release the memory again. `P` can be a typed or untyped pointer.

Be sure to release this memory with the `FreeMem` (194) call defined in the `objpas` unit.

Errors: In case no more memory is available, and no more memory could be obtained from the system a run-time error is triggered.

See also: `FreeMem` (194), `Getmem` (153).

For an example, see `FreeMem` (194).

GetStringCurrentValue

Declaration: `Function GetStringCurrentValue(TableIndex,StringIndex : Longint)
: AnsiString;`

Description: `GetStringCurrentValue` returns the current value of the resourcestring in table `TableIndex` with index `StringIndex`.

The current value depends on the system of internationalization that was used, and which language is selected when the program is executed.

Errors: If either `TableIndex` or `StringIndex` are out of range, then an empty string is returned.

See also: `SetResourceStrings` (200), `GetStringDefaultValue` (196), `GetStringHash` (196), `GetStringName` (197), `ResourceStringTableCount` (199), `ResourceStringCount` (199)

Listing: `refex/ex90.pp`

Program `Example90;`

```
{ Program to demonstrate the GetStringCurrentValue function. }  
{ $Mode Delphi }
```

`ResourceString`

```
  First = 'First string';  
  Second = 'Second String';
```

```
Var I,J : Longint;
```

```
begin
  { Print current values of all resourcestrings }
  For I:=0 to ResourceStringTableCount-1 do
    For J:=0 to ResourceStringCount(i)-1 do
      Writeln (I, ', ', J, ' : ', GetResourceStringCurrentValue(I, J));
end.
```

GetResourceStringDefaultValue

Declaration: Function GetResourceStringDefaultValue(TableIndex, StringIndex : Longint)
: AnsiString

Description: GetResourceStringDefaultValue returns the default value of the resourcestring in table TableIndex with index StringIndex.

The default value is the value of the string that appears in the source code of the programmer, and is compiled into the program.

Errors: If either TableIndex or StringIndex are out of range, then a empty string is returned.

Errors:

See also: SetResourceStrings ([200](#)), GetResourceStringCurrentValue ([195](#)), GetResourceStringHash ([196](#)), GetResourceStringName ([197](#)), ResourceStringTableCount ([199](#)), ResourceStringCount ([199](#))

Listing: refex/ex91.pp

Program Example91;

```
{ Program to demonstrate the GetResourceStringDefaultValue function. }
{$Mode Delphi}
```

ResourceString

```
First = 'First string';
Second = 'Second String';
```

Var I, J : Longint;

```
begin
  { Print default values of all resourcestrings }
  For I:=0 to ResourceStringTableCount-1 do
    For J:=0 to ResourceStringCount(i)-1 do
      Writeln (I, ', ', J, ' : ', GetResourceStringDefaultValue(I, J));
end.
```

GetResourceStringHash

Declaration: Function GetResourceStringHash(TableIndex, StringIndex : Longint) : Longint;

Description: GetResourceStringHash returns the hash value associated with the resource string in table TableIndex, with index StringIndex.

The hash value is calculated from the default value of the resource string in a manner that gives the same result as the GNU gettext mechanism. It is stored in the resourcestring tables, so retrieval is faster than actually calculating the hash for each string.

Errors: If either `TableIndex` or `StringIndex` is zero, 0 is returned.

See also: `Hash` (197) `SetResourceStrings` (200), `GetResourceStringDefaultValue` (196), `GetResourceStringHash` (196), `GetResourceStringName` (197), `ResourceStringTableCount` (199), `ResourceStringCount` (199)

For an example, see `Hash` (197).

GetResourceStringName

Declaration: `Function GetResourceStringName(TableIndex,StringIndex : Longint) : AnsiString;`

Description: `GetResourceStringName` returns the name of the resourcestring in table `TableIndex` with index `StringIndex`. The name of the string is always the unit name in which the string was declared, followed by a period and the name of the constant, all in lowercase.

If a unit `MyUnit` declares a resourcestring `MyTitle` then the name returned will be `myunit.mytitle`. A resourcestring in the program file will have the name of the program prepended.

The name returned by this function is also the name that is stored in the resourcestring file generated by the compiler.

Strictly speaking, this information isn't necessary for the functioning of the program, it is provided only as a means to easier translation of strings.

Errors: If either `TableIndex` or `StringIndex` is zero, an empty string is returned.

See also: `SetResourceStrings` (200), `GetResourceStringDefaultValue` (196), `GetResourceStringHash` (196), `GetResourceStringName` (197), `ResourceStringTableCount` (199), `ResourceStringCount` (199)

Listing: refex/ex92.pp

Program Example92;

```
{ Program to demonstrate the GetResourceStringName function. }
{$Mode Delphi}
```

```
ResourceString
```

```
    First = 'First string';
    Second = 'Second String';
```

```
Var I,J : Longint;
```

```
begin
```

```
    { Print names of all resourcestrings }
```

```
    For I:=0 to ResourceStringTableCount-1 do
```

```
        For J:=0 to ResourceStringCount(I)-1 do
```

```
            WriteLn (I, ', ', J, ' : ', GetResourceStringName(I,J));
```

```
end.
```

Hash

Declaration: `Function Hash(S : AnsiString) : longint;`

Description: Hash calculates the hash value of the string S in a manner that is compatible with the GNU gettext hash value for the string. It is the same value that is stored in the Resource string tables, and which can be retrieved with the `GetResourceStringHash` (196) function call.

Errors: None. In case the calculated hash value should be 0, the returned result will be -1.

See also: `GetResourceStringHash` (196),

Listing: `refex/ex93.pp`

Program `Example93;`

```
{ Program to demonstrate the Hash function . }
{$Mode Delphi}

ResourceString

    First  = 'First string';
    Second = 'Second String';

Var I,J : Longint;

begin
    For I:=0 to ResourceStringTableCount-1 do
        For J:=0 to ResourceStringCount(i)-1 do
            If Hash(GetResourceStringDefaultValue(I,J))
                <>GetResourceStringHash(I,J) then
                WriteLn ('Hash mismatch at ',I,', ',J)
            else
                WriteLn ('Hash (',I,', ',J,') matches. ');
end.
```

Paramstr

Declaration: `Function ParamStr(Param : Integer) : Ansistring;`

Description: `ParamStr` returns the Param-th command-line parameter as an `AnsiString`. The system unit `Paramstr` (171) function limits the result to 255 characters.

The zeroeth command-line parameter contains the path of the executable, except on LINUX, where it is the command as typed on the command-line.

Errors: In case Param is an invalid value, an empty string is returned.

See also: `Paramstr` (171)

For an example, see `Paramstr` (171).

ReAllocMem

Declaration: `function ReAllocMem(var p:pointer;Size:Longint):pointer;`

Description: `ReAllocMem` resizes the memory pointed to by P so it has size Size. The value of P may change during this operation. The contents of the memory pointed to by P (if any) will be copied to the new location, but may be truncated if the newly allocated memory block is smaller in size. If a larger block is allocated, only the used memory is initialized, extra memory will not be zeroed out.

Note that P may be nil, in that case the behaviour of `ReAllocMem` is equivalent to `Getmem` (153).

Errors: If no memory is available then a run-time error will occur.

See also: [Getmem \(153\)](#), [Freemem \(152\)](#)

ResetResourceTables

Declaration: `Procedure ResetResourceTables;`

Description: `ResetResourceTables` resets all resource strings to their default (i.e. as in the source code) values.

Normally, this should never be called from a user's program. It is called in the initialization code of the objpas unit. However, if the resourcetables get messed up for some reason, this procedure will fix them again.

Errors: None.

See also: [SetResourceStrings \(200\)](#), [GetResourceStringDefaultValue \(196\)](#), [GetResourceStringHash \(196\)](#), [GetResourceStringName \(197\)](#), [ResourceStringTableCount \(199\)](#), [ResourceStringCount \(199\)](#)

ResourceStringCount

Declaration: `Function ResourceStringCount(TableIndex : longint) : longint;`

Description: `ResourceStringCount` returns the number of resource strings in the table with index `TableIndex`. The strings in a particular table are numbered from 0 to `ResourceStringCount-1`, i.e. they're zero based.

Errors: If an invalid `TableIndex` is given, -1 is returned.

See also: [SetResourceStrings \(200\)](#), [GetResourceStringCurrentValue \(195\)](#), [GetResourceStringDefaultValue \(196\)](#), [GetResourceStringHash \(196\)](#), [GetResourceStringName \(197\)](#), [ResourceStringTableCount \(199\)](#),

For an example, see [GetResourceStringDefaultValue \(196\)](#)

ResourceStringTableCount

Declaration: `Function ResourceStringTableCount : Longint;`

Description: `ResourceStringTableCount` returns the number of resource string tables; this may be zero if no resource strings are used in a program.

The tables are numbered from 0 to `ResourceStringTableCount-1`, i.e. they're zero based.

Errors:

See also: [SetResourceStrings \(200\)](#), [GetResourceStringDefaultValue \(196\)](#), [GetResourceStringHash \(196\)](#), [GetResourceStringName \(197\)](#), [ResourceStringCount \(199\)](#)

For an example, see [GetResourceStringDefaultValue \(196\)](#)

SetResourceStrings

Declaration: TResourceIterator = Function (Name, Value : AnsiString; Hash : Longint): AnsiString;
Procedure SetResourceStrings (SetFunction : TResourceIterator);

Description: SetResourceStrings calls SetFunction for all resource strings in the resource string tables and sets the resource string's current value to the value returned by SetFunction.

The Name, Value and Hash parameters passed to the iterator function are the values stored in the tables.

Errors: None.

See also: GetResourceStringCurrentValue (195), GetResourceStringDefaultValue (196), GetResourceStringHash (196), GetResourceStringName (197), ResourceStringTableCount (199), ResourceStringCount (199)

Listing: refex/ex95.pp

Program Example95;

```
{ Program to demonstrate the SetResourceStrings function. }  
{$Mode objfpc}
```

ResourceString

```
First = 'First string';  
Second = 'Second String';
```

```
Var I, J : Longint;  
S : AnsiString;
```

```
Function Translate (Name, Value : AnsiString; Hash : longint): AnsiString;
```

```
begin  
  Writeln ('Translate (', Name, ') => ', Value);  
  Write ('->');  
  Readln (Result);  
end;
```

```
begin  
  SetResourceStrings (@Translate);  
  Writeln ('Translated strings : ');  
  For I:=0 to ResourceStringTableCount-1 do  
    For J:=0 to ResourceStringCount(I)-1 do  
      begin  
        Writeln (GetResourceStringDefaultValue(I, J));  
        Writeln ('Translates to : ');  
        Writeln (GetResourceStringCurrentValue(I, J));  
      end;  
end.
```

SetResourceStringValue

Declaration: Function SetResourceStringValue(TableIndex, StringIndex : longint; Value : Ansistring) : Boolean;

Description: `SetResourceStringValue` assigns `Value` to the resource string in table `TableIndex` with index `StringIndex`.

Errors:

See also: `SetResourceStrings` (200), `GetResourceStringCurrentValue` (195), `GetResourceStringDefaultValue` (196), `GetResourceStringHash` (196), `GetResourceStringName` (197), `ResourceStringTableCount` (199), `ResourceStringCount` (199)

Listing: `refex/ex94.pp`

Program `Example94`;

```
{ Program to demonstrate the SetResourceStringValue function. }
{$Mode Delphi}

ResourceString

    First = 'First string';
    Second = 'Second String';

Var I,J : Longint;
    S : AnsiString;

begin
    { Print current values of all resourcestrings }
    For I:=0 to ResourceStringTableCount-1 do
        For J:=0 to ResourceStringCount(i)-1 do
            begin
                Writeln ( 'Translate => ',GetResourceStringDefaultValue(I,J));
                Write   ( '→');
                Readln(S);
                SetResourceStringValue(I,J,S);
            end;
        Writeln ( 'Translated strings : ');
        For I:=0 to ResourceStringTableCount-1 do
            For J:=0 to ResourceStringCount(i)-1 do
                begin
                    Writeln ( GetResourceStringDefaultValue(I,J));
                    Writeln ( 'Translates to : ');
                    Writeln ( GetResourceStringCurrentValue(I,J));
                end;
            end;
        end.
```

Index

Abs, [127](#)
Addr, [127](#)
Append, [128](#)
Arctan, [128](#)
Assert, [129](#)
Assign, [129](#)
Assigned, [130](#)
AssignFile, [193](#)

BinStr, [130](#)
Blockread, [131](#)
Blockwrite, [131](#)
Break, [132](#)

Chdir, [132](#)
Chr, [133](#)
Close, [133](#)
CloseFile, [194](#)
CompareByte, [134](#)
CompareChar, [135](#)
CompareDWord, [136](#)
CompareWord, [137](#)
Concat, [138](#)
Continue, [139](#)
Copy, [139](#)
Cos, [140](#)
CSeg, [140](#)

Dec, [141](#)
Delete, [141](#)
Dispose, [142](#)
DSeg, [143](#)

Eof, [143](#)
Eoln, [144](#)
Erase, [144](#)
Exclude, [145](#)
Exit, [146](#)
Exp, [147](#)

Filepos, [148](#)
Filesize, [148](#)
FillByte, [149](#)
Fillchar, [150](#)
FillDWord, [150](#)
Fillword, [151](#)

Flush, [151](#)
Frac, [152](#)
Freemem, [152](#), [194](#)

Getdir, [153](#)
Getmem, [153](#), [195](#)
GetMemoryManager, [154](#)
GetStringCurrentValue, [195](#)
GetStringDefaultValue, [196](#)
GetStringHash, [196](#)
GetStringName, [197](#)

Halt, [154](#)
Hash, [197](#)
HexStr, [154](#)
Hi, [155](#)
High, [155](#)

Inc, [156](#)
Include, [157](#)
IndexByte, [157](#)
IndexChar, [158](#)
IndexDWord, [159](#)
IndexWord, [160](#)
Insert, [160](#)
Int, [161](#)
IOresult, [161](#)
IsMemoryManagerSet, [161](#)

Length, [163](#)
Ln, [163](#)
Lo, [164](#)
LongJump, [164](#)
Low, [164](#)
Lowercase, [165](#)

Mark, [165](#)
Maxavail, [166](#)
Memavail, [166](#)
Mkdir, [167](#)
Move, [167](#)
MoveChar0, [168](#)

New, [168](#)

OctStr, [169](#)

Odd, [168](#)
Ofs, [169](#)
Ord, [170](#)

Paramcount, [170](#)
Paramstr, [171](#), [198](#)
Pi, [171](#)
Pos, [172](#)
Power, [172](#)
Pred, [172](#)
Ptr, [173](#)

Random, [173](#)
Randomize, [174](#)
Read, [174](#)
Readln, [175](#)
Real2Double, [175](#)
ReAllocMem, [198](#)
Release, [176](#)
Rename, [176](#)
Reset, [177](#)
ResetResourceTables, [199](#)
ResourceStringCount, [199](#)
ResourceStringTableCount, [199](#)
Rewrite, [177](#)
Rmdir, [178](#)
Round, [179](#)
Runerror, [179](#)

Seek, [179](#)
SeekEof, [180](#)
SeekEoln, [181](#)
Seg, [181](#)
SetJump, [182](#)
SetLength, [182](#)
SetMemoryManager, [182](#)
SetResourceStrings, [200](#)
SetResourceStringValue, [200](#)
SetString, [183](#)
SetTextBuf, [183](#)
Sin, [184](#)
SizeOf, [185](#)
Sptr, [185](#)
Sqr, [185](#)
Sqrt, [186](#)
SSeg, [186](#)
Str, [187](#)
StringOfChar, [187](#)
Succ, [188](#)
Swap, [188](#)

Trunc, [188](#)
Truncate, [189](#)

Uppcase, [189](#)

Val, [190](#)

Write, [191](#)
WriteLn, [191](#)